


# PROGRAMMI DI MATEMATICA E STATISTICA In **BASIC**



Hervé Haut



GRUPPO  
EDITORIALE  
JACKSON

EDIZIONE  
ITALIANA



# **PROGRAMMI DI MATEMATICA E STATISTICA in BASIC**

di  
**Hervé Haut**



GRUPPO  
EDITORIALE  
JACKSON  
Via Rosellini, 12  
20124 Milano

*Hervé Haut è ingegnere civile e Dottore in Scienze Fisiche.*

*Egli è attualmente assistente della cattedra di radioprotezione dell'Università Cattolica "Louvain La Neuve" in Belgio.*



- © Copyright per l'edizione originale Editions du P.S.I. - 1981
- © Copyright per l'edizione originale italiana Gruppo Editoriale Jackson s.r.l. - 1982

Il Gruppo Editoriale Jackson ringrazia per il prezioso lavoro svolto nella stesura dell'edizione italiana la signora Francesca di Fiore e l'Ing. Roberto Pancaldi.

Tutti i diritti sono riservati. Stampato in Italia. Nessuna parte di questo libro può essere riprodotta, memorizzata in sistemi di archivio, o trasmessa in qualsiasi forma o mezzo, elettronico, meccanico, fotocopia, registrazione o altri senza la preventiva autorizzazione scritta dell'editore.

**Prima edizione dicembre 1982**

Stampato in Italia da:

S.p.A. Alberto Matarelli - Milano - Stabilimento Grafico

Fotocomposizione CorpoNove s.n.c. - Bergamo - tel. (035) 22.33.63-22.33.65

# SOMMARIO

## A — PROGRAMMI MATEMATICI

1 — Interpolazione polinomiale di Lagrange . . . . .	1
2 — Valore numerico di un polinomio e delle sue derivate . . . . .	9
3 — Derivazione numerica di una funzione discreta (interpolazione di Newton) . . . . .	19
4 — Radici di equazioni di secondo, terzo e quarto grado . . . . .	31
5 — Integrazione numerica: . . . . .	49
— regola di Simpson (2n intervalli)	
— regola di Bode (10 punti)	
— regola di Gauss (12 punti)	
— regola di Chebyshev (9 punti)	
— estremo superiore infinito	
6 — Equazioni differenziali di primo e secondo ordine . . . . .	65
7 — Grafico di una funzione . . . . .	77
8 — Ricerca degli zeri di una funzione . . . . .	87
9 — Risoluzione dell'equazione matriciale $AX = B$ (inversione di una matrice) . . . . .	95
10 — Autovalori ed autovettori di una matrice reale simmetrica . . . . .	115

## B — PROGRAMMI STATISTICI

11 — Calcolo delle funzioni statistiche per una o due variabili . . . . .	129
12 — Calcolo delle medie e dei momenti di una variabile stocastica . . . . .	139
13 — Distribuzioni statistiche: . . . . .	147
— distribuzione binomiale	
— distribuzione di Poisson	
— distribuzione normale	
— distribuzione normale a due variabili	
— distribuzione in chi-quadro	
— distribuzione t di Student	
14 — Regressione ad un parametro: . . . . .	173
— regressione lineare semplice	
— adattamento ad una curva del tipo $y = ax^b$	
— adattamento ad una curva esponenziale	
— adattamento ad una curva logaritmica	
15 — Regressione lineare multipla . . . . .	189
16 — Adattamento ad un polinomio . . . . .	205

<b>BIBLIOGRAFIA</b> . . . . .	217
-------------------------------	-----



# PRESENTAZIONE

*I programmi presentati in quest'opera sono stati scelti in modo da portare il lettore a formarsi una logica di base per la risoluzione di problemi di tipo matematico o statistico.*

*Per la codifica di tali programmi si è utilizzato il Basic Applesoft implementato sul sistema Apple II, ma, dato che le particolarità di questa versione del Basic sono state utilizzate essenzialmente nella gestione dell'Input-Output e non nelle routines di calcolo, risulta molto semplice adattare questi stessi programmi ad un'altra versione del linguaggio Basic.*

*Per consentire un approccio graduale alle tecniche dell'analisi numerica, ci si è sforzati di presentare i problemi in ordine di difficoltà crescente. Fanno eccezione i primi due programmi di statistica (numeri 11 e 12) che per la semplicità di sviluppo potrebbero essere trattati anche prima di tutti gli altri.*

*Ad ogni programma si è preposta una esposizione schematica del metodo numerico e delle tecniche di programmazione utilizzati. Il lettore che trovasse insufficiente questa esposizione potrà comunque consultare la bibliografia consigliata a cui l'autore ha fatto specifico riferimento.*

*Completa ogni capitolo, oltre ovviamente al listato del programma, il diagramma a blocchi relativo all'algoritmo utilizzato (stilato secondo i criteri di normalizzazione A-FNOR — ISO) e un esempio pratico in cui, tra l'altro, vengono specificati il tempo e la quantità di memoria utilizzati.*

*Nell'intento di permettere al lettore di inserire ciascuno dei programmi presentati in un disegno più vasto, si sono strutturati i programmi stessi nel modo seguente:*

- un programma generale che assicura esclusivamente la gestione corretta su video dell'input-output;*
- una routine che rappresenta la risoluzione del problema proposto e che può essere a sua volta suddivisa in più subroutines.*

*Per ogni procedura vengono dettagliate le regole per l'uso con particolare riguardo*

*a: dati necessari (in input), risultati forniti, variabili, vettori e matrici utilizzati, funzioni e sottoprogrammi richiamati.*

*Il dimensionamento dei vettori e delle matrici viene fatto nel programma principale in modo da evitare l'errore "redim'd array" se uno di questi viene utilizzato in più procedure, oppure se viene più volte richiamata la procedura che contiene un'istruzione di dimensionamento.*

*Consigliamo al lettore, per approfondire la comprensione dei programmi, di provare ogni programma proposto con un semplice esempio e di seguirne l'andamento passo passo, in modo da afferrare meglio il meccanismo degli algoritmi utilizzati.*



## PROGRAMMA NUMERO 1

# INTERPOLAZIONE POLINOMIALE DI LAGRANGE

Si supponga di conoscere i valori di una funzione  $y = f(x)$  in  $n + 1$  punti  $[y_0 = f(x_0), y_1 = f(x_1), \dots, y_n = f(x_n)]$  e di voler calcolare il valore di  $y$  corrispondente ad un generico valore di  $x$ .

L'interpolazione polinomiale di Lagrange risolve questo problema permettendo di costruire una funzione polinomiale  $p(x)$  (di grado minore o uguale a  $n$ ) passante per gli  $n+1$  punti del piano  $(x, y)$ .

Il valore cercato di  $y$  si può allora facilmente ricavare dal polinomio interpolatore  $p(x)$   $[y = p(x)]$ .

Il valore cercato di  $y$  si può allora facilmente ricavare dal polinomio interpolatore  $p(x)$   $[y = p(x)]$ .

## 1 — METODO NUMERICO

Date  $n+1$  coppie di valori  $(x_i, y_i)$ , si definiscono "funzioni elementari di Lagrange" relative a tali punti le funzioni  $L_i(x)$

$$(1) \quad L_i(x) = \prod_{\substack{j=0 \\ j \neq i}}^n \frac{x - x_j}{x_i - x_j} \quad (i = 0, 1, \dots, n)$$

Si può verificare facilmente che queste funzioni soddisfano alla seguente proprietà:

$$L_i(x_k) = \delta_{ik} \quad (k = 0, 1, \dots, n) \quad \text{ove} \quad \delta_{ik} = \begin{cases} 1 & \text{se } i = k \\ 0 & \text{se } i \neq k \end{cases}$$

Il polinomio

$$(2) \quad p(x) = \sum_{i=0}^n y_i L_i(x)$$

soddisfa le condizioni  $y_i = p(x_i)$  in quanto risulta:

$$p(x_k) = \sum_{i=0}^n y_i L_i(x_k) = \sum_{i=0}^n y_i \delta_{ik} = y_k \quad (k = 0, 1, \dots, n)$$


e pertanto l'espressione (2) rappresenta il polinomio interpolatore di Lagrange. Questo polinomio permette di interpolare in un punto qualsiasi la funzione definita dalle  $n+1$  coppie  $x_i, y_i$ .


*Riferimenti bibliografici:* A2, B1, B2, C2, H1, L3.

## 2 — TECNICA DI PROGRAMMAZIONE

Le  $N+1$  coppie di valori  $(X_i, Y_i)$  vengono memorizzate nei due vettori  $X(I)$  e  $Y(I)$  ( $I = 0, 1, \dots, N$ ). L'espressione (2) è allora calcolata mediante un ciclo ( $I = 0, 1, \dots, N$ ) in cui ogni termine della (2) è a sua volta calcolato, per  $X = X_I$ , mediante un ciclo ( $J = 0, 1, \dots, I-1, I+1, \dots, N$ ) applicando la formula (1).

### 3 — DIAGRAMMA A BLOCCHI





## 4 — PROGRAMMA

```

1  REM          INTERPOLAZIONE POLINOMIALE
2  REM          DI LAGRANGE
3  REM
4  REM          AUTORE:H.HAUT
5  REM
6  REM
7  REM
8  REM  DESCRIZIONE:
9  REM      QUESTO PROGRAMMA CONSENTE IL CALCOLO DEL
10 REM      VALORE DEL POLINOMIO INTERPOLATORE
11 REM      DI LAGRANGE PASSANTE PER UN NUMERO ARBITRARIO
12 REM      DI PUNTI (X,Y)
13 REM
14 REM      *****
15 REM
16 REM
100 REM
110 REM      REGOLE D'USO
120 REM
130 HOME
140 PRINT : PRINT "          INTERPOLAZIONE POLINOMIALE": PRINT "          DI
LAGRANGE"
150 PRINT : PRINT "          AUTORE: H.HAUT"
160 VTAB (10)
170 PRINT " 1.INTRODURRE IL NUMERO DI COPPIE": PRINT "          CHE DEFINISCON
O IL POLINOMIO": PRINT "          INTERPOLATORE": PRINT "          (N+1 COPPIE DE
FINISCONO UN POLINOMIO          DI GRADO N)"
180 PRINT
190 PRINT " 2.INTRODURRE SUCCESSIVAMENTE LE COPPIE          X,Y": PRINT
200 PRINT " 3.DEFINIRE LA VARIABILE X PER CUI": PRINT "          SI VUOLE CALC
OLARE IL VALORE": PRINT "          DI INTERPOLAZIONE PER Y"
210 VTAB (23): HTAB (5): PRINT "PREMERE UN TASTO PER CONTINUARE ": GET A$

220 HOME
230 REM
240 REM      GESTIONE DEI DATI IN INPUT
250 REM
260 INPUT "IL NUMERO DI COPPIE=":N1:N = N1 - 1
270 REM
280 REM      ISTRUZIONI DI DIMENSIONAMENTO
290 REM
300 DIM X(N),Y(N)
310 FOR I = 0 TO N: REM      INSERIMENTO DATI
320 PRINT "COPPIA ";I + 1;": "
330 INPUT "          X=":X(I)
340 INPUT "          Y=":Y(I)
350 NEXT I
360 PRINT : INPUT "INTERPOLAZIONE PER X=":XI
370 REM
380 REM      CHIAMATA DEL SOTTOPROGRAMMA
390 REM
400 GOSUB 1000
410 REM
420 REM      GESTIONE DEI RISULTATI
430 REM
440 PRINT "VALORE INTERPOLATO Y=":YI
450 PRINT : INPUT "VOLETE L'INTERPOLAZIONE PER UN ALTRO          VALORE (S O N)
? ":A$
460 IF A$ = "S" THEN GOTO 360
470 END

1000 REM *****
1010 REM      SOTTOPROGRAMMA PER IL CALCOLO DELL'INTERPOLAZIONE
1020 REM      POLINOMIALE DI LAGRANGE
1030 REM
1040 REM

```

```

1050 REM
1060 REM      1.DATI NECESSARI
1070 REM      * N=GRADO DEL POLINOMIO=NUMERO DI COPPIE -1!!
1080 REM      * X(I) E Y(I) (I=0,1,...,N) =VETTORI
1090 REM      CONTENENTI LE N+1 COPPIE (X,Y)
1100 REM      * XI=VALORE DI X PER CUI L'INTER-
1110 REM      POLAZIONE DEVE ESSERE CALCOLATA
1120 REM      2.RISULTATI FORNITI:
1130 REM      * YI=RISULTATO DELL'INTERPOLAZIONE
1140 REM      3.VARIABILI UTILIZZATE:
1150 REM      DE,I,J,N,NU,XI,YI
1160 REM      4.VETTORI UTILIZZATI:
1170 REM      X(N),Y(N)
1180 REM      *****
1190 YI = 0
1200 FOR I = 0 TO N
1210 NU = Y(I): REM      INIZIALIZZAZIONE DEL NUMERATORE
1220 DE = 1: REM      E DEL DENOMINATORE
1225 REM
1230 REM      CALCOLO DI UN TERMINE DELLA SOMMA
1240 FOR J = 0 TO N
1250 IF J = I THEN GOTO 1280
1260 NU = NU * (XI - X(J))
1270 DE = DE * (X(I) - X(J))
1280 NEXT J
1285 REM
1290 REM      CALCOLO DELLA SOMMATORIA
1300 YI = YI + NU / DE
1310 NEXT I
1320 RETURN

```

## 5 — ESEMPIO PRATICO

Si cerchi il valore di interpolazione per  $X = 3.5$  ed  $X = 4.2$  di una funzione definita dalle cinque coppie:

$X,Y = (0,1), (1,-1), (2,-13), (4,5), (5,131)$

RUN

INTERPOLAZIONE POLINOMIALE  
DI LAGRANGE

AUTORE: H.HAUT

1.INTRODURRE IL NUMERO DI COPPIE  
CHE DEFINISCONO IL POLINOMIO  
INTERPOLATORE  
(N+1 COPPIE DEFINISCONO UN POLINOMIO  
DI GRADO N)

2.INTRODURRE SUCCESSIVAMENTE LE COPPIE  
X,Y

3.DEFINIRE LA VARIABILE X PER CUI  
SI VUOLE CALCOLARE IL VALORE  
DI INTERPOLAZIONE PER Y  
PREMERE UN TASTO PER CONTINUARE

TI NUMERO DI COPPIE=5  
COPPIA 1:

X=0

Y=1

```
COPPIA 2:  
  X=1  
  Y=-1  
COPPIA 3:  
  X=2  
  Y=-13  
COPPIA 4:  
  X=4  
  Y=5  
COPPIA 5:  
  X=5  
  Y=131
```

```
INTERPOLAZIONE PER X=3.5  
VALORE INTERPOLATO Y=-16.9375
```

```
VOLETE L'INTERPOLAZIONE PER UN ALTRO  
VALORE (S O N) ? S
```

```
INTERPOLAZIONE PER X=4.2  
VALORE INTERPOLATO Y=20.0175999
```

```
VOLETE L'INTERPOLAZIONE PER UN ALTRO  
VALORE (S O N) ? N
```

tempo d'esecuzione: 1"

memoria richiesta: 2484 bytes (senza REM : 768)





## PROGRAMMA NUMERO 2

# VALORE NUMERICO DI UN POLINOMIO E DELLE SUE DERIVATE

Si vuole calcolare il valore assunto da un polinomio dato (di grado  $n$ ) e dalle sue derivate in un generico punto  $x_0$ .

## 1 — METODO NUMERICO

### 1.1 — Calcolo del valore del polinomio per $x = x_0$

Sia dato il polinomio:

$$(1) \quad p(x) = \sum_{i=0}^n a_i x^{n-i} = a_0 x^n + a_1 x^{n-1} + \dots + a_n$$

Per calcolare il valore assunto da tale polinomio per  $x = x_0$  si consideri l'espressione (1) (con  $x = x_0$ ) scritta nella seguente forma equivalente:

$$p(x_0) = (\dots (( (a_0 x_0 + a_1) x_0 + a_2) x_0 + a_3) x_0 + \dots) x_0 + a_n$$

da cui si ricava facilmente l'algoritmo di calcolo, noto sotto il nome di algoritmo di Horner - Ruffini:

definiti per il polinomio (1) e per il valore  $x_0$  i numeri di Horner nel modo seguente:

$$k_0 = a_0$$

$$k_1 = k_0 x_0 + a_1$$

$$k_2 = k_1 x_0 + a_2$$

.....

$$k_i = k_{i-1} x_0 + a_i$$

.....

$$k_n = k_{n-1} x_0 + a_n$$

si avrà

$$p(x_0) =: k_n$$

## 1.2 — Calcolo delle derivate

### 1.2.1 — DERIVATA PRIMA

La derivata di  $p(x)$  sarà un polinomio di grado  $n - 1$ .

Definito un polinomio  $p_1(x)$  tale che:

$$(3) \quad p(x) = p(x_0) + (x - x_0) p_1(x)$$

si ponga

$$(4) \quad p_1(x) = \sum_{i=0}^{n-1} b_i x^{n-1-i}$$

Derivando entrambi i termini della (3) e ponendo  $x = x_0$  si ottiene:

$$p'(x_0) = p_1(x_0)$$

Dunque il valore della derivata di  $p(x)$  in  $x_0$  coincide con il valore assunto dal polinomio  $p_1(x)$  nello stesso punto.

Utilizzando la (3) e la (4) si può allora scrivere:

$$(5) \quad \sum_{i=0}^n a_i x^{n-i} = p(x_0) + (x - x_0) \sum_{i=0}^{n-1} b_i x^{n-1-i}$$

e, tramite un confronto tra i due membri esprimere i coefficienti  $b_i$  in funzione dei coefficienti  $a_i$ .

Si può facilmente verificare che i coefficienti del polinomio  $p_1(x)$  sono uguali ai primi  $n$  numeri di Horner di  $p(x_0)$ :  $b_i = k_i$  ( $i = 0, 1, \dots, n - 1$ ).

Per calcolare il valore del polinomio  $p_1(x) = p'(x)$  nel punto  $x_0$  si può applicare ora l'algoritmo di Horner.

Definiti i numeri di Horner per  $p_1(x_0)$  come:

$$\begin{aligned} q_n &= k_0 \\ (6) \quad q_1 &= q_0 x_0 + k_1 \\ &\dots\dots\dots \\ q_{n-1} &= q_{n-2} x_0 + k_{n-1} \end{aligned}$$

si ottiene:

$$p'(x_0) = q_{n-1}$$

### 1.2.2 — DERIVATE DI ORDINE SUPERIORE

Generalizzando il ragionamento del paragrafo precedente e definiti i polinomi

$$p_m(x) = p_m(x_0) + (x - x_0) p_{m+1}(x_0), (m = 0, 1, \dots, n - 1)$$

si può dimostrare che:

- $p(x) = p_0(x)$
- il valore della derivata di ordine  $i$  di  $p(x)$  nel punto  $x = x_0$  è dato da:

$$(7) \quad p^{(i)}(x_0) = i! p_i(x_0).$$

- i coefficienti del polinomio  $p_i(x)$  sono i numeri di Horner relativi al polinomio  $p_{i-1}(x_0)$ .

Queste considerazioni permettono di sviluppare dettagliatamente l'algoritmo richiesto.

*Riferimenti bibliografici:* C1, D1, L3.

## 2 — TECNICA DI PROGRAMMAZIONE

I coefficienti  $a_i$  sono memorizzati in un vettore  $A(I)$ ,  $i = 0, 1, \dots, N$ .

### 2.1 — Calcolo di $p(x_0)$

Si segue scrupolosamente l'algoritmo (2), inizializzando  $P = A(0)$  ed eseguendo ciclicamente  $N$  volte l'espressione  $P = P * X_0 + A(I)$ .

All'uscita dal ciclo la variabile  $P$  conterrà il valore assunto dal polinomio  $p(x)$  per  $x = x_0$ .

### 2.2 — Calcolo di $p(x_0)$ e di $p'(x_0)$

Si inizializzano le variabili  $P1$  e  $P$  rispettivamente con i valori 0 e  $A(0)$ .

Si calcolano ciclicamente per  $N$  volte i valori delle espressioni

$$P1 = P1 * X_0 + P \text{ e } P = P * X_0 + A(I)$$

seguendo gli algoritmi (3) e (6).

All'uscita dal ciclo si avrà:

$$P = p(x_0) \text{ e } P1 = p'(x_0)$$

### 2.3 — Calcolo di $p(x)$ e delle sue derivate per $x = x_0$

Si utilizza un vettore  $P(l)$  ( $l = 0, 1, \dots, N$ ) in cui l'elemento  $i$ -esimo è destinato a contenere il valore della derivata  $i$ -esima di  $p(x)$  calcolata per  $x = x_0$ . [Si ponga  $P(0) = p_0(x_0) = p(x_0)$ ].

Inizializzato il vettore  $P$  con i valori  $P(l) = A(N-1)$  si procede nel calcolo mediante  $N$  passaggi successivi ( $l = 1, \dots, N$ ) secondo lo schema seguente:


Nel passaggio  $l$ -esimo si calcola ciclicamente per  $N - l - 1$  volte l'espressione


$$P(J) = P(J) + X_0 * P(J+1) \quad (J = N-1, N-2, \dots, l-1)$$

Dopo ogni passaggio, l'ultima componente del vettore  $P$  calcolata [vale a dire  $P(l-1)$ ] viene moltiplicata per il fattoriale di  $(l-1)$ , conformemente a quanto espresso nella formula (7).

$l = 1$		$l = 2$		$\dots$		$l = N - 1$	$l = N$
$P(N-1)$	$P(N-1) + X_0 * P(N)$	$P(N-1) = P(N-1) + X_0 * P(N)$	$P(N-2) = P(N-2) + X_0 * P(N-1)$	$\dots$	$\dots$	$P(N-1) = P(N-1) + X_0 * P(N)$	$P(N-1) = P(N-1) + X_0 * P(N)$
$P(N-2)$	$P(N-2) + X_0 * P(N-1)$	$P(N-2) = P(N-2) + X_0 * P(N-1)$	$P(N-3) = P(N-3) + X_0 * P(N-2)$	$\dots$	$\dots$	$P(N-2) = P(N-2) + X_0 * P(N-1)$	
			$P(1) = P(1) + X_0 * P(2)$	$\dots$	$\dots$		
$P(0)$	$P(0) + X_0 * P(1)$						

### 3 — DIAGRAMMA A BLOCCHI





## 4 - PROGRAMMA

```

1  REM      CALCOLO DEL VALORE
2  REM      NUMERICO DI UN POLINOMIO
3  REM      E DELLE SUE DERIVATE
4  REM
5  REM      AUTORE:H.HAUT
6  REM
7  REM
8  REM      DESCRIZIONE:DOPO AVER INTRODOTTO
9  REM      I COEFFICIENTI DEL POLINOMIO
10 REM      E IL PUNTO PER CUI SI VUOL ESEGUIRE
11 REM      IL CALCOLO, IL PROGRAMMA CALCOLA
12 REM      IL VALORE DEL POLINOMIO NEL PUNTO
13 REM      OPPURE TALE VALORE E QUELLO
14 REM      DELLA DERIVATA PRIMA,OPPURE
15 REM      IL VALORE DEL POLINOMIO E DI
16 REM      TUTTE LE SUE DERIVATE,A SCELTA
17 REM
18 REM      *****
19 REM
20 REM
100 REM
110 REM
120 REM
130 HOME
140 PRINT : PRINT "  CALCOLO DEL VALORE NUMERICO DI UN          POLINOMIO E
    DELLE DERIVATE "
150 PRINT : PRINT "                                AUTORE:H.HAUT":
160 PRINT "  REGOLE D'USO:" : PRINT
170 PRINT "      1.INTRODURRE DI SEGUITO IL GRADO          DEL POLINOMIO,I
    COEFFICIENTI E      IL PUNTO DI CALCOLO": PRINT
180 PRINT "      2.SCEGLIERE IL TIPO DI CALCOLO:
    1=VALORE DEL POLINOMIO
    =VALORE DEL POLINOMIO E DELLA      DERIVATA PRIMA"
190 PRINT "      3=VALORE DEL POLINOMIO E      DI TUTTE L
    E DERIVATE"
200 PRINT : PRINT "SIMBOLOGIA:P0(X)=VALORE DEL POLINOMIO          P1(
X),P2(X)...PN(X)=VALORI      DELLE DERIVATE "
210 PRINT : PRINT "PREMERE UN TASTO PER CONTINUARE": GET A%
220 HOME
230 REM
240 REM      GESTIONE DEI DATI IN INPUT
250 REM
260 INPUT "GRADO DEL POLINOMIO ";N: PRINT
270 REM
280 REM      ISTRUZIONI DI DIMENSIONAMENTO
290 REM
300 DIM A(N),P(N)
310 PRINT : FOR I = 0 TO N: HTAB (5): PRINT "COEF. DI X^";N   TICU = PEEK
(37): VTAB (CV): HTAB (20): INPUT "=";A(I): NEXT I
320 PRINT : INPUT "CALCOLO NEL PUNTO X0=";X0
330 PRINT : INPUT "TIPO DI CALCOLO SCELTO ";TE
340 REM
350 REM      CHIAMATA DEL SOTTOPROGRAMMA
360 REM
370 GOSUB 1000
380 REM
390 REM      GESTIONE DEI RISULTATI
400 REM
410 HOME
420 IF TE = 1 THEN PRINT "P0(";X0;")=";P: GOTO 480
430 IF TE = 2 THEN PRINT "P0(";X0;")=";P: PRINT "P1(";X0;")=";P1: GOTO 4
80
440 FOR I = 0 TO N: PRINT "P";I;"(";X0;")=";P(I):, NEXT I: GOTO 480
450 REM
460 REM      POSSIBILITA' DI CALCOLO IN UN ALTRO PUNTO

```

```

470 REM
480 PRINT "VOLETE IL CALCOLO PER UN ALTRO PUNTO? "; INPUT "
(S O N)"; A$
490 IF A$ = "S" THEN GOTO 320
500 END
1000 REM *****
1010 REM SOTTOPROGRAMMA PER IL CALCOLO DEL
1020 REM VALORE DEL POLINOMIO IN X0
1030 REM  $P(X) = A(0) X^N + A(1) X^{(N-1)} + \dots + A(N-1) X + A(N)$ 
1040 REM E DI TUTTE LE SUE DERIVATE
1050 REM
1060 REM REGOLE D'USO:
1070 REM 1. DATI NECESSARI:
1080 REM * N=GRADO DEL POLINOMIO
1090 REM * A(I)=VETTORE DEI COEFFICIENTI
1100 REM SECONDO IL MODELLO DELLA RIGA 1030
1110 REM * X0=PUNTO DI CALCOLO
1120 REM * TE=1,2,3 :
1130 REM 1=CALCOLO DI P(X0)
1140 REM 2=CALCOLO DI P(X0) E P'(X0)
1150 REM 3=CALCOLO DI P(X0) E DI TUTTE LE DERIVATE
1160 REM 2. RISULTATI FORNITI:
1170 REM * SE TE=1, P=P(X0)
1180 REM * SE TE=2, P=P(X0) E P1=SUA DERIVATA PRIMA
1190 REM * SE TE=3, P(0)=P(X0); P(1), ..., P(N)=LE DERIVATE
1200 REM SUCCESSIVE IN X=X0
1210 REM
1220 REM 3. VARIABILI UTILIZZATE:
1230 REM I, J, K, L, N, P, P1, X0
1240 REM 4. VETTORI UTILIZZATI:
1250 REM A(N), P(N)
1260 REM *****
1270 REM SCELTA DEL TIPO DI CALCOLO
1280 ON TE GOTO 1300, 1360, 1430
1290 REM TE=1
1300 P = A(0)
1310 FOR I = 1 TO N
1320 P = P * X0 + A(I)
1330 NEXT I
1340 RETURN
1350 REM TE=2
1360 P1 = 0; P = A(0)
1370 FOR I = 1 TO N
1380 P1 = P1 * X0 + P
1390 P = P * X0 + A(I)
1400 NEXT I
1410 RETURN
1420 REM TE=3
1430 FOR I = 0 TO N: P(I) = A(N - I): NEXT I
1440 L = 1: REM INIZIALIZZAZIONE DEL FATTORIALE
1450 FOR I = 1 TO N
1460 FOR J = I TO N
1470 K = N + I - 1 - J
1480 P(K) = P(K) + X0 * P(K + 1)
1490 NEXT J
1500 P(K) = P(K) * L
1510 L = L * I
1520 NEXT I
1530 P(N) = P(N) * L
1540 RETURN

```



## 5 — ESEMPIO PRATICO

Si vuole calcolare il valore del polinomio di quinto grado:

$x^5 + 4x^2 + x - 2$  e delle sue derivate per  $x = 2$ .

RUN

CALCOLO DEL VALORE NUMERICO DI UN  
POLINOMIO E DELLE DERIVATE

AUTORE:H.HAUT

REGOLE D'USO:

1.INTRODURRE DI SEGUITO IL GRADO  
DEL POLINOMIO,I COEFFICIENTI E  
IL PUNTO DI CALCOLO

2.SCEGLIERE IL TIPO DI CALCOLO:

1=VALORE DEL POLINOMIO  
2=VALORE DEL POLINOMIO E DELLA  
DERIVATA PRIMA  
3=VALORE DEL POLINOMIO E  
DI TUTTE LE DERIVATE

SIMBOLOGIA:P0(X)=VALORE DEL POLINOMIO  
P1(X),P2(X),...PN(X)=VALORI  
DELLE DERIVATE

PREMERE UN TASTO PER CONTINUARE  
GRADO DEL POLINOMIO 5

COEF. DI X^5	=1
COEF. DI X^4	=0
COEF. DI X^3	=0
COEF. DI X^2	=4
COEF. DI X^1	=1
COEF. DI X^0	=-2

CALCOLO NEL PUNTO X0=2

TIPO DI CALCOLO SCELTO 3	
P0(2)=48	P1(2)=97
P2(2)=168	P3(2)=240
P4(2)=240	P5(2)=120

VOLETE IL CALCOLO PER UN ALTRO PUNTO?  
(S O N): N

tempo d'esecuzione: 1"

memoria richiesta: 3424 bytes (senza REM : 1131)



## PROGRAMMA NUMERO 3

# DERIVAZIONE NUMERICA DI UNA FUNZIONE DISCRETA

## INTERPOLAZIONE DI NEWTON

Come l'interpolazione polinomiale di Lagrange, l'interpolazione di Newton permette di calcolare il valore assunto da una funzione  $f(x)$ , definita da  $n + 1$  coppie di valori  $[x_i, y_i = f(x_i)]$ , in un punto qualsiasi.

Questo metodo di interpolazione è senza dubbio più efficace del precedente in quanto permette di calcolare direttamente sia il valore assunto dalla funzione sia quello assunto dalle sue prime  $n$  derivate in un punto a piacere.

### 1 — METODO NUMERICO

#### 1.1 — Polinomio interpolatore di Newton

Il polinomio interpolatore di Newton, passante per gli  $n + 1$  punti  $[x_i, y_i = f(x_i)]$  ( $i = 0, 1, \dots, n$ ) è il polinomio (di grado minore o uguale a  $n$ ):

$$P_0(x) = D_n + D_{n-1}(x - x_n) + D_{n-2}(x - x_n)(x - x_{n-1}) + \dots \\ + D_0(x - x_n)(x - x_{n-1}) \dots (x - x_0)$$

ove i  $D_i$  rappresentano le “differenze divise” definite nel modo seguente:

$$D_n = [x_n] = f(x_n)$$

$$D_{n-1} = [x_n, x_{n-1}] = \frac{[x_n] - [x_{n-1}]}{x_n - x_{n-1}}$$

$$(1) \quad D_{n-2} \dots [x_n, x_{n-1}, x_{n-2}] = \frac{[x_n, x_{n-1}] - [x_{n-1}, x_{n-2}]}{x_n - x_{n-2}}$$

$$D_0 [x_n, x_{n-1}, \dots, x_0] = \frac{[x_n, \dots, x_1] - [x_{n-1}, \dots, x_0]}{x_n - x_0}$$

Il valore del polinomio  $P_0(x)$  per  $x = t$  può essere allora calcolato con l'algoritmo di Horner (cfr programma numero 2) nel modo seguente:

$$(2) \quad \begin{aligned} a_0 &= D_0 \\ a_1 &= a_0(t - x_1) + D_1 \end{aligned}$$

$$P_0(t) = a_n = a_{n-1}(t - x_n) + D_n$$

ove  $a_0, \dots, a_n$  rappresentano i coefficienti di Horner di  $P_0(x)$ . In effetti il polinomio  $P_0(x)$  può anche essere scritto nella forma:

$$P_0(x) = [\dots [ [D_0(x - x_1) + D_1] (x - x_2) + D_2] (x - x_3) + \dots] (x - x_n) + D_n$$

## 1.2 — Derivazioni successive

Definiti i polinomi  $P_k(x)$  di grado  $n-k$  tramite la formula di ricorrenza:

$$P_k(x) = (x - t) P_{k+1}(x) + P_k(t) \quad (k = 0, \dots, n-1)$$

si può dimostrare che:

a) la derivata di ordine  $k$  di  $P_0(x)$  calcolata in  $x = t$  è data da:

$$p^{(k)}(t) = K! P_k(t)$$

b) le differenze divise di  $P_k(x)$  sono i numeri di Horner di  $P_{k-1}(x)$ .

*Riferimenti bibliografici:* B1, B2, C2, C3, H1, L3, S2.

## 2 — TECNICA DI PROGRAMMAZIONE

### 2.1 — Calcolo dei $D_i$

I valori iniziali  $x_i$  e  $y_i$  sono, come al solito, memorizzati nei vettori  $X(I)$  ed  $Y(I)$ .

Il calcolo dei  $D_i$  si effettua con l'aiuto di un vettore  $D(J)$  che viene inizializzato ponendo  $D(J) = Y(J)$ .

Si procede poi in  $n$  passaggi distinti ( $I = 1, 2, \dots, N$ ). Ad ogni passaggio si effettua per  $N-I+1$  volte l'istruzione:

$$D(J) = \frac{D(J+1) - D(J)}{X(J+1) - X(J)} \quad (J = 1, \dots, N-1)$$


si ottengono così le  $N + 1$  differenze divise come illustrato nello schema seguente


	inizializ- zazione	$I = 1$	$I = 2$	...	$I = N - 1$	$I = N$
$D(0) =$	$[x_0]$	$[x_1, x_0]$	$[x_2, x_1, x_0]$	...	$[x_{n-1}, \dots, x_0]$	$[x_n, \dots, x_0]$
$D(1) =$	$[x_1]$	$[x_2, x_1]$	$[x_3, x_2, x_1]$	...	$[x_n, \dots, x_1]$	
$D(2) =$	$[x_2]$	$[x_3, x_2]$	$[x_4, x_3, x_2]$			
	.	.	.	.		
	.	.	.	.		
$D(N-2) =$	$[x_{n-2}]$	$[x_{n-1}, x_{n-2}]$	$[x_n, x_{n-1}, x_{n-2}]$			
$D(N-1) =$	$[x_{n-1}]$	$[x_n, x_{n-1}]$				
$D(N) =$	$[x_n]$					



## 2.2 — Calcolo delle derivate

· Per il calcolo delle derivate si segue un procedimento analogo a quello illustrato nel capitolo precedente (cfr paragrafo 2.3).


### 3 — DIAGRAMMA A BLOCCHI














#### 4 — PROGRAMMA

```

1  REM      DERIVAZIONE NUMERICA DI UNA FUNZIONE DISCRETA
2  REM      INTERPOLAZIONE DI NEWTON.
3  REM
4  REM      AUTORE:H.HAUT
5  REM
6  REM
7  REM      DESCRIZIONE:
8  REM      PARTENDO DA N+1 COPPIE (X,Y=F(X)),
9  REM      IL PROGRAMMA PERMETTE DI CALCOLARE IN
10 REM      UN PUNTO X QUALSIASI IL VALORE DI
11 REM      UNA FUNZIONE F E DI TUTTE LE SUE DERIVATE
12 REM      CON L'INTERPOLAZIONE POLINOMIALE DI NEWTON
13 REM      *****
14 REM
15 REM
100 REM
110 REM
120 REM
130 HOME
140 PRINT TAB( 6);"DERIVAZIONE NUMERICA"
150 PRINT TAB( 6);"DI UNA FUNZIONE DISCRETA:"
160 PRINT TAB( 6);"INTERPOLAZIONE DI NEWTON"
170 PRINT : PRINT TAB( 20);"AUTORE:H.HAUT": PRINT : PRINT
180 PRINT "REGOLE D'USO:"

```

```

190 PRINT "      1.INSERIRE IL NUMERO DI COPPIE          (X(I),Y(I)=F(X(I
   )) ).
200 PRINT "      2.INSERIRE LE COPPIE (X,Y) L'UNA      DOPO L'ALTRA "
210 PRINT "      3.DEFINIRE IL PUNTO X PER CUI          SI VUOLE L'INTER
   POLAZIONE"
220 PRINT "      4.SCEGLIERE IL TIPO DI CALCOLO:"
230 PRINT "          1=CALCOLO DI F(X)"
240 PRINT "          2=CALCOLO DI F(X) E F'(X)"
250 PRINT "          3=CALCOLO DI F(X) E DELLE SUE      DERIVATE"
260 UTAB (23): PRINT "PREMERE UN TASTO PER CONTINUARE ": GET A$
270 HOME
280 REM
290 REM      GESTIONE DEI DATI IN INPUT
300 REM
310 INPUT "IL NUMERO DI COPPIE (X,Y)=";N1: PRINT
320 N = N1 - 1: REM      N=GRADO DEL POLINOMIO
330 REM
340 REM      ISTRUZIONI DI DIMENSIONAMENTO
350 REM
360 DIM X(N),Y(N),F(N),D(N)
370 REM
380 FOR I = 0 TO N
390 PRINT "COPIA ";I + 1;" : "
400 INPUT "          X=";X(I)
410 INPUT "          Y=";Y(I)
420 NEXT I
430 PRINT : INPUT "CALCOLO NEL PUNTO X=";X
440 PRINT : INPUT "SCELTA DEL TIPO DI CALCOLO: ";CH
450 REM
460 REM      CHIAMATA DEL SOTTOPROGRAMMA
470 REM
480 GOSUB 1000
490 REM
500 REM      GESTIONE DEI RISULTATI
510 REM
520 ON CH GOTO 530,560,590
530 REM      TIPO=1
540 PRINT "F(";X;")=";P: PRINT
550 GOTO 630
560 REM      TIPO=2
570 PRINT "F(";X;")=";P: PRINT "F'(";X;")=";P1: PRINT
580 GOTO 630
590 REM      TIPO=3
600 PRINT "LE DERIVATE SUCCESSIVE DI F IN X SONO:"
610 FOR I = 0 TO N - 1 STEP 2: PRINT "F";I; TAB( 4);"=";P(I); TAB( 21);"F
   ";I + 1; TAB( 24);"=";P(I + 1): NEXT I
620 IF N = 2 * INT (N / 2) THEN PRINT "F";N; TAB( 4);"=";P(N)
630 REM      POSSIBILITA' D'UN ALTRO CALCOLO
640 INPUT "SI VUOLE UN'ALTRA INTERPOLAZIONE          (S O N)?";A$
650 IF A$ = "S" THEN GOTO 430
660 END
1000 REM      *****
1010 REM      SOTTOPROGRAMMA PER IL CALCOLO DELL'INTERPOLAZIONE
1020 REM      POLINOMIALE DI NEWTON.
1030 REM
1040 REM
1050 REM
1060 REM      1.DATI NECESSARI
1070 REM      * N=GRADO DEL POLINOMIO D'INTERPOLAZIONE
1080 REM      (=IL NUMERO DI COPPIE (X,Y) - 1 !!)
1090 REM      * X(I) E Y(I) = VETTORI CONTENENTI
1100 REM      , LE N+1 COPPIE (X,Y=F(X)) .
1110 REM      (I=0,1,...,N)
1120 REM      * X=PUNTO PER CUI SI VUOLE L'INTERPOLAZIONE
1130 REM      * CH =TIPO DI CALCOLO SCELTO :
1140 REM          1=CALCOLO DI F(X)
1150 REM          2=CALCOLO DI F(X) E F'(X)

```

```

1160 REM          3=CALCOLO DI F(X) E DI TUTTE LE SUE DERIVATE
1170 REM
1180 REM          2.RISULTATI FORNITI:
1190 REM          SE CH=1, P=F(X)
1200 REM          SE CH=2: P=F(X),P1=F'(X)
1220 REM          SE CH=3,P(0)=F(X)
1230 REM          P(I) (I=1,...,N) = LE DERIVATE
1240 REM          SUCCESSIVE DI F IN X
1250 REM          3.VARIABILI UTILIZZATE:
1260 REM          F,I,J,K,L,N,N1,X
1270 REM          4.VETTORI UTILIZZATI:
1280 REM          D(N),P(N),X(N),Y(N)
1290 REM          *****
1300 REM          INIZIALIZZAZIONE DELLE DIFFERENZE DIVISE
1310 FOR I = 0 TO N
1320 D(I) = Y(I)
1330 NEXT I
1340 REM          CALCOLO DELLE DIFFERENZE DIVISE
1350 FOR I = 1 TO N
1360 FOR J = 0 TO N - I
1370 D(J) = (D(J + 1) - D(J)) / (X(J + I) - X(J))
1380 NEXT J
1390 NEXT I
1400 REM          SCELTA DEL TIPO DI CALCOLO
1410 ON CH GOTO 1420,1480,1560
1420 REM          CH=1
1430 P = D(0)
1440 FOR I = 1 TO N
1450 P = P * (X - X(I)) + D(I)
1460 NEXT I
1470 RETURN
1480 REM          CH=2
1490 P = D(0)
1500 P1 = 0
1510 FOR I = 1 TO N
1520 P1 = P1 * (X - X(I)) + P
1530 P = P * (X - X(I)) + D(I)
1540 NEXT I
1550 RETURN
1560 REM          CH=3
1570 REM          INIZIALIZZAZIONE DEI P(I)
1580 FOR I = 0 TO N
1590 P(I) = D(N - I)
1600 NEXT I
1610 F = 1: REM          INIZIALIZZAZIONE DEL FATTORIALE
1620 N1 = N - 1
1630 REM          CALCOLO DELLE DERIVATE SUCCESSIVE
1640 FOR I = 1 TO N
1650 L = N1 + I
1660 FOR J = I TO N
1670 K = L - J
1680 P(K) = P(K) + P(K + 1) * (X - X(J))
1690 NEXT J
1700 P(K) = P(K) * F: REM          MOLTIPLICAZIONE PER IL FATTORIALE DI (I-1)
1710 F = F * I: REM          CALCOLO DEL FATTORIALE (F=I!)
1720 NEXT I
1730 P(N) = P(N) * F
1740 RETURN

```

## 5 — ESEMPIO PRATICO

Data una funzione definita tramite le 6 coppie di valori (x,y):

(0,1), (1,-1), (2,-19), (3,-35), (4,113), (5,851)

si vuole calcolare il valore della funzione e delle sue derivate per  $x = 6$ .

```
RUN
  DERIVAZIONE NUMERICA
  DI UNA FUNZIONE DISCRETA:
  INTERPOLAZIONE DI NEWTON

      AUTORE:H.HAUT

REGOLE D'USO:
  1.INSERIRE IL NUMERO DI COPPIE
    (X(I),Y(I)=F(X(I))) .
  2.INSERIRE LE COPPIE (X,Y) L'UNA
    DOPO L'ALTRA
  3.DEFINIRE IL PUNTO X PER CUI
    SI VUOLE L'INTERPOLAZIONE
  4.SCEGLIERE IL TIPO DI CALCOLO:
    1=CALCOLO DI F(X)
    2=CALCOLO DI F(X) E F'(X)
    3=CALCOLO DI F(X) E DELLE SUE
      DERIVATE
```

PREMERE UN TASTO PER CONTINUARE  
IL NUMERO DI COPPIE (X,Y)=6

```
COPPIA 1 :
      X=0
      Y=1
COPPIA 2 :
      X=1
      Y=-1
COPPIA 3 :
      X=2
      Y=-19
COPPIA 4 :
      X=3
      Y=-35
COPPIA 5 :
      X=4
      Y=113
COPPIA 6 :
      X=5
      Y=851
```

CALCOLO NEL PUNTO X=6

```
SCelta DEL TIPO DI CALCOLO: 3
LE DERIVATE SUCCESSIVE DI F IN X SONO:
F0 =2989          F1 =3228
F2 =2662          F3 =1596
F4 =624           F5 =120
SI VUOLE UN'ALTRA INTERPOLAZIONE
(S O N)? :N
```

tempo d'esecuzione: 1''1

memoria richiesta: 4170 bytes (senza REM : 1433)



## PROGRAMMA NUMERO 4

# RADICI DI EQUAZIONI DI SECONDO, TERZO E QUARTO GRADO

## 1 — METODI NUMERICI

### 1.1 — Equazione di secondo grado

Le soluzioni dell'equazione  $x^2 + bx + c = 0$ , posto  $q = \sqrt{b^2 - 4c}$ , sono:

a) se  $q \geq 0$  due radici reali 
$$x_{1,2} = -\frac{b}{2} \pm \frac{\sqrt{q}}{2} \quad (1)$$

b) se  $q < 0$  due radici complesse 
$$x_{1,2} = -\frac{b}{2} \pm i \frac{\sqrt{-q}}{2}$$

### 1.2 — Equazione di terzo grado

Sia data l'equazione  $x^3 + a_2x^2 + a_1x + a_0 = 0$  (2) e sia  $\lambda$  il massimo dei valori assoluti dei coefficienti  $a_i$  ( $\lambda = \max |a_i|$ ,  $i = 0, 1, 2$ ).

Si può dimostrare che:

- a) se  $a_0 > 0$  esiste una radice reale negativa nell'intervallo  $[-(1 + \lambda), 0]$
- b) se  $a_0 < 0$  esiste una radice reale positiva nell'intervallo  $[0, (1 + \lambda)]$ .

Si procede al calcolo di questa soluzione reale  $x_1$  utilizzando il metodo detto della

bisezione dell'intervallo (cfr programma numero 8); in seguito dividendo le (2) per  $(x - x_1)$  si trova che le altre due soluzioni sono le radici dell'equazione quadratica:

$$x^2 + (a_2 + x_1) x + [a_1 + (a_2 + x_1) x_1] = 0$$

che si risolve seguendo le regole (1).

### 1.3 — Equazione di quarto grado

Si voglia risolvere l'equazione:

$$(3) \quad x^4 + ax^3 + bx^2 + cx + d = 0$$

Ponendo  $x = y - a/4$  l'equazione (3) si scompone nelle due equazioni di secondo grado in  $y$ :

$$(4) \quad (y^2 + 2ky + 1) (y^2 - 2ky + m) = 0$$

ove:

$$\left. \begin{array}{l} \text{a) } l + m - 4k^2 = q \\ 2k(m - l) = r \\ lm = s \end{array} \right\} (5) \quad \begin{array}{l} q = b - 3a^2/8 \\ r = c - (ab/2) + a^3/8 \\ s = d - (ac/4) + (a^2b/16) - (3a^4/256) \end{array}$$

b)  $k^2 = z$  è una soluzione reale dell'equazione cubica

$$(6) \quad z^3 + \alpha z^2 + \beta z + \gamma = 0$$

con  $\alpha = q/2$

$$(7) \quad \begin{array}{l} \beta = (q^2 - 4s)/16 \\ \gamma = -r^2/64 \end{array}$$

Il metodo consiste dunque nel risolvere l'equazione (6) con il sistema esposto nel paragrafo 1.2, nel calcolare poi i valori di  $l$  e  $m$  risolvendo il sistema (5) e nel risolvere infine le due equazioni (4) in modo da trovare le quattro radici cercate.

*Riferimenti bibliografici:* B1, B3, D1, J1, W1.



## 2 – TECNICA DI PROGRAMMAZIONE

Si definiscono un vettore  $R(4)$ , destinato a contenere le soluzioni dell'equazione ed una matrice  $A(4,4)$  contenente i coefficienti dell'equazione di grado  $N$ :

$$\sum_{l=0}^N A(N, l)X^l = 0 \quad (N = 2, 3, 4)$$

Il sottoprogramma 1000 affronta il problema nella sua generalità, mentre i sottoprogrammi 2000 e 3000 risolvono rispettivamente le equazioni "normalizzate":

$$x^3 + A(3,2) x^2 + A(3,1) x + A(3,0) = 0 \text{ e } x^2 + A(2,1) x + A(2,0) = 0$$

Considerando risolto il problema relativo alla "ricerca di uno zero" (esposto nel capitolo 8), il programma si snoda in modo piuttosto lineare.

Vogliamo solo far risaltare che il sistema (5) viene risolto con tecniche diverse a seconda che sia uguale o diverso da zero, e precisamente:

a) se  $\gamma \neq 0$ , essendo anche  $K^2 \neq 0$ , si possono utilizzare le formule:

$$l + m = q + 4k^2$$

(8)

$$l - m = r/2k$$

b) se  $\gamma = 0$ , ne consegue che  $k^2 = 0$  è una soluzione della (6) e pertanto:

1. se  $q^2 - 4s \geq 0$  si ha:

$$\begin{aligned} l + m &= q \\ lm &= s \end{aligned}$$


che equivale a:


$$(9) \quad \begin{aligned} l + m &= q \\ l - m &= \sqrt{q^2 - 4s} \end{aligned}$$


2. se  $q^2 - 4s < 0$  non si possono utilizzare le (9).


In questo caso però l'equazione (6) si riduce a:  $z^2 + \alpha z + \beta = 0$  e ammette due soluzioni reali di segno opposto (essendo  $\beta < 0$ ). Basta allora considerare per  $k^2$  la soluzione positiva e calcolare di conseguenza, tramite le (8), i valori di  $l$  ed  $m$ .


### 3 — DIAGRAMMA A BLOCCHI














## 4 — PROGRAMMA

```

1  REM          RISOLUZIONE DI EQUAZIONI
2  REM          DI SECONDO,TERZO,QUARTO GRADO
3  REM
4  REM          AUTORE:H.HAUT
5  REM
6  REM
7  REM          DESCRIZIONE:
8  REM          IL PROGRAMMA PERMETTE DI CALCOLARE TUTTE
9  REM          LE SOLUZIONI (EVENTUALMENTE COMPLESSE)
10 REM          DI EQUAZIONI ALGEBRICHE
11 REM          DI SECONDO,TERZO E QUARTO GRADO
12 REM          *****
13 REM
14 REM
100 REM
110 REM          REGOLE D'USO
120 REM
130 HOME
140 UTAB (3): PRINT TAB( 7);"RISOLUZIONE DI EQUAZIONI"
150 PRINT TAB( 7);"DI SECONDO,TERZO O QUARTO GRADO"
160 PRINT : PRINT TAB( 20);"AUTORE:H.HAUT"
170 UTAB (7): PRINT "REGOLE D'USO: "; PRINT
180 PRINT "      1.INTRODURRE IL GRADO DELL'EQUAZIONE"
190 PRINT "      2.INTRODURRE I COEFFICIENTI          PARTENDO DA QUEL
    LO DELLA PIU' ALTA POTENZA DI X": PRINT
200 PRINT "      3.LE SOLUZIONI SONO DATE SOTTO LA FORMA:
    (PARTE REALE) + I (PARTE IMMAGINARIA)"
210 UTAB (22): PRINT "PREMERE UN TASTO PER CONTINUARE": GET A$
220 REM
230 HOME
240 REM
250 REM          ISTRUZIONI DI DIMENSIONAMENTO
260 REM
270 DIM A(4,4),R(4)
280 REM
290 REM          GESTIONE DEI DATI IN INPUT
300 REM
310 INPUT "QUALE E' IL GRADO DELL'EQUAZIONE? ";N: PRINT
320 FOR I = N TO 0 STEP - 1
330 HTAB (5): PRINT "COEFFICIENTE DI X^";I;
350 INPUT "=";A(N,I)
360 NEXT I
370 PRINT : PRINT
380 REM
390 REM          CHIAMATA DEL SOTTOPROGRAMMA
400 REM
410 GOSUB 1000
420 REM
430 REM          GESTIONE DEI RISULTATI
440 REM
450 ON N - 1 GOTO 470,520,580
460 REM
470 REM          EQUAZIONE DI SECONDO GRADO
480 REM
490 IF IM = 0 THEN PRINT "RADICE 1=";R(1): PRINT "RADICE 2=";R(2): END
500 PRINT "RADICE 1=";R(1);" +I ";IM: PRINT "RADICE 2=";R(2);" -I ";IM: END
510 REM
520 REM          EQUAZIONE DI TERZO GRADO
530 REM
540 PRINT "RADICE 1=";R(3)
550 IF IM = 0 THEN PRINT "RADICE 2=";R(1): PRINT "RADICE 3=";R(2): END

```



```

560 PRINT "RADICE 2=";R(1);" +I ";IM; PRINT "RADICE 3=";R(2);" -I "
;IM: END
570 REM
580 REM EQUAZIONE DI QUARTO GRADO
590 REM
600 IF IM = 0 THEN PRINT "RADICE 1=";R(1): PRINT "RADICE 2=";R(2): GOTO
620
610 PRINT "RADICE 1=";R(1);" +I ";IM; PRINT "RADICE 2=";R(2);" -I "
;IM
620 IF II = 0 THEN PRINT "RADICE 3=";R(3): PRINT "RADICE 4=";R(4): END
630 PRINT "RADICE 3=";R(3);" +I ";II: PRINT "RADICE 4=";R(4);" -I "
;II
640 END

1000 REM *****
1010 REM SOTTOPROGRAMMA PER IL CALCOLO DELLE
1020 REM RADICI DI UN'EQUAZIONE DI SECONDO,TERZO
1025 REM O QUARTO GRADO
1030 REM
1040 REM REGOLE D'USO PER IL SOTTOPROGRAMMA
1050 REM
1060 REM 1.DATI NECESSARI:
1070 REM * N=GRADO DELL'EQUAZIONE
1080 REM * A(N,I)=COEFFICIENTI DI X^I NELLA
1090 REM EQUAZIONE: (I=0,...,N)
1100 REM 2.RISULTATI FORNITI:
1110 REM LE RADICI SONO DATE NELLE FORME:
1120 REM *SE N=2: R(1)+I IM
1130 REM R(2)-I IM
1140 REM *SE N=3: R(1)+I IM
1150 REM R(2)-I IM
1160 REM R(3)
1170 REM *SE N=4: R(1)+I IM
1180 REM R(2)-I IM
1190 REM R(3)+I IM
1200 REM R(4)-I IM
1210 REM 3.VARIABILI UTILIZZATE:
1220 REM A,AM,AY,B,C,D,ER,I,II,IM,K,L,M
1230 REM Q,Q1,Q2,R,S,SW,TE,TT,XA,XB,XC,Y1,Y2,W
1240 REM 4.ETTORE E MATRICE UTILIZZATI:
1250 REM A(4,4),R(4)
1260 REM 5.FUNZIONE UTILIZZATA:
1270 REM F(X)
1280 REM 6.SOTTOPROGRAMMI RICHIAMATI:
1290 REM 2000,3000
1300 REM
1310 REM OSSERVAZIONE IMPORTANTE:
1320 REM A(N,N) DEVE ESSERE DIVERSO DA ZERO
1330 REM (L'EQUAZIONE DEVE ESSERE VERAMENTE DI GRADO N)
1340 REM *****
1350 REM NORMALIZZAZIONE DEI COEFFICIENTI
1360 FOR I = 0 TO N - 1
1370 A(N,I) = A(N,I) / A(N,N)
1380 NEXT I
1390 REM SALTO RELATIVO A N
1400 ON N - 1 GOTO 1410,1440,1480
1410 REM EQUAZIONE DI SECONDO GRADO
1420 GOSUB 3000
1430 RETURN
1440 REM EQUAZIONE DI TERZO GRADO
1450 GOSUB 2000
1460 RETURN
1470 REM
1480 REM EQUAZIONE DI QUARTO GRADO
1490 REM
1500 A = A(4,3):B = A(4,2):C = A(4,1):D = A(4,0)
1510 REM

```

```

1520 REM      CALCOLO DI Q,R,S (CFR (4) NEL TESTO)
1530 REM
1540 Q = R - (3 * A * A / 8)
1550 R = C - (A * B / 2) + (A * A * A / 8)
1560 S = D - (A * C / 4) + (A * A * B / 16) - (3 * A * A * A * A / 256)
1570 REM
1580 REM      DEFINIZIONE DEI COEFFICIENTI DELL'EQUAZIONE
1600 REM      CURICA (5) DEL TESTO
1610 A(3,2) = Q / 2
1620 A(3,1) = (Q * Q - 4 * S) / 16
1630 A(3,0) = - (R * R / 64)
1640 REM
1650 REM      CALCOLO DI UNA RADICE REALE DELL'EQUAZIONE
1660 REM
1670 IF (R < > 0) OR (A(3,1) > = 0) THEN GOTO 1760
1680 REM
1690 REM      CASO PARTICOLARE IN CUI L'EQUAZIONE (5) E' DI SECONDO GRADO
1700 REM
1710 A(2,1) = A(3,2); A(2,0) = A(3,1)
1720 GOSUB 3000: REM      RISOLUZIONE DI (5)
1730 R(3) = R(1): REM      SI PRENDE LA RADICE POSITIVA
1740 GOTO 1810
1750 REM
1760 REM      CHIAMATA DEL SOTTOPROGRAMMA 3000 CON SW=-1 PER CALCOLARE UNA
1770 REM      RADICE
1780 SW = - 1
1790 GOSUB 2000
1800 REM
1810 K = SQR (R(3)): REM      RADICE REALE DELL'EQUAZIONE (5)
1820 IF K = 0 THEN R = SQR (Q * Q - 4 * S): GOTO 1880: REM      CALCOLO DI
      L-M SE K=0
1830 Q = Q + (4 * R(3)): REM      L+M
1840 R = R / (2 * K): REM      L-M
1850 REM
1860 REM      CALCOLO DI L ED M
1870 REM
1880 L = (Q - R) / 2; M = (Q + R) / 2
1890 REM
1900 REM      RISOLUZIONE DELLE 2 EQUAZ. DI 2NDO GRADO (CFR (3) NEL TESTO)
1910 REM
1920 A(2,1) = 2 * K; A(2,0) = L
1930 GOSUB 3160: REM      PRIMA EQUAZIONE
1940 R(3) = R(1) - (A(4,3) / 4); R(4) = R(2) - (A(4,3) / 4); II = IM: REM
      TRASFERIMENTO DELLA SOLUZIONE IN R(3), R(4), II
1950 A(2,1) = - 2 * K; A(2,0) = M
1960 GOSUB 3160: REM      SECONDA EQUAZIONE
1970 R(2) = R(2) - (A(4,3) / 4); R(1) = R(1) - (A(4,3) / 4)
1980 RETURN
2000 REM      *****
2010 REM      SOTTOPROGRAMMA PER IL CALCOLO DELLE SOLUZIONI DELL'EQUAZIONE

2020 REM      X^3 + A(3,2)*X^2 + A(3,1)*X + A(3,0) = 0
2030 REM
2040 REM      REGOLE PER L'USO DEL SOTTOPROGRAMMA:
2050 REM      1.DATI NECESSARI:
2060 REM      * A(3,2), A(3,1) ED A(3,0)=COEFFICIENTI
2070 REM      DELL'EQUAZIONE
2080 REM      * SW=-1 SE SI VUOLE CALCOLARE SOLO UNA
2090 REM      RADICE REALE
2100 REM      ALTRIMENTI SW QUALSIASI
2110 REM      2.RISULTATI FORNITI:
2120 REM      *SE SW=-1, R(3)=RADICE REALE
2130 REM      *ALTRIMENTI LE RADICI DELL'EQUAZIONE SONO
2140 REM      R(1) +I IM
2150 REM      R(2) -I IM
2160 REM      R(3)
2170 REM      3.VARIABILI UTILIZZATE:

```

```

2180 REM      A,AM,B,ER,I,IM,Q1,Q2,SW,TF,TT,W,
2190 REM      XA,XR,XC,Y1,Y2
2200 REM      4. VETTORE F MATRICE UTILIZZATI:
2210 REM      A(4,4),R(4)
2220 REM      5. FUNZIONE UTILIZZATA:
2230 REM      F(X)
2240 REM      6. SOTTOPROGRAMMA CHIAMATO:
2250 REM      3000
2260 REM      *****
2270 IF A(3,0) = 0 THEN XC = 0: GOTO 2560: REM      IN QUESTO CASO C'E' UNA
RADICE NULLA
2280 REM
2290 REM      RICERCA DEL COEFFICIENTE MAGGIORE IN VALORE ASSOLUTO
2300 REM
2310 AM = ABS (A(3,0))
2320 FOR I = 1 TO 3
2330 TT = ABS (A(3,I))
2340 IF (AM < TT) THEN AM = TT
2350 NEXT I
2360 REM
2370 REM      DEFINIZIONE DELL'INTERVALLO IN CUI C'E' UNA RADICE REALE
2380 REM
2390 IF A(3,0) > 0 THEN A = - AM - 1: B = 0: GOTO 2440
2400 A = 0: B = AM + 1
2410 REM
2420 REM      RICERCA DELLA RADICE REALE XC NELL'INTERVALLO A,B:
2425 REM      METODO DELLA BISEZIONE
2430 REM
2440 DEF FN F(X) = X * X * X + A(3,2) * X * X + A(3,1) * X + A(3,0)
2450 REM      DEFINIZIONE DELL'INTERVALLO IN CUI C'E' LA RADICE: XA,XB
2460 XA = A: XB = B: Y1 = FN F(A): Y2 = FN F(B)
2470 ER = 1E - 6: REM      PRECISIONE VOLUTA
2480 IF Y1 = 0 THEN XC = A: GOTO 2560: REM      RADICE TROVATA
2490 IF Y2 = 0 THEN XC = B: GOTO 2560: REM      RADICE TROVATA
2500 XC = (XA + XB) / 2: TE = FN F(XC): REM      XC=PUNTO MEDIO DI XA,XB
2505 IF TE = 0 THEN GOTO 2560
2510 IF (XB - XA) < ER THEN GOTO 2560:
2511 REM      SE L'AMPIEZZA DELL'INTERVALLO E'
2515 REM      MINORE DELLA PRECISIONE RICHIESTA, XC E' LA RADICE CERCATA
2516 REM
2520 REM      DEFINIZIONE DEL NUOVO INTERVALLO E PROSECUZIONE DELLA RICERCA
2530 IF (FN F(XA) * TE) > 0 THEN XA = XC: GOTO 2500
2540 XB = XC: GOTO 2500
2550 REM
2560 R(3) = XC: REM      R(3)=RADICE REALE
2570 REM
2580 IF SW = - 1 THEN RETURN: REM      SI CALCOLA SOLO LA RADICE REALE
2590 REM
2600 REM      CALCOLO DELLE RADICI DELL'EQUAZIONE QUADRATICA (2) DEL TESTO
2610 REM
2620 REM      DEFINIZIONE DEI COEFFICIENTI
2630 A(2,1) = A(3,2) + XC
2640 A(2,0) = A(3,1) + A(3,2) * XC + XC * XC
2650 GOSUB 3000: REM      CALCOLO DELLE RADICI
2660 RETURN
3000 REM      *****
3010 REM      SOTTOPROGRAMMA PER IL CALCOLO DELLE RADICI DELL'EQUAZIONE:
3020 REM       $X^2 + A(2,1)X + A(2,0) = 0$ 
3030 REM
3040 REM      REGOLE PER L'USO DEL SOTTOPROGRAMMA
3050 REM      1. DATI NECESSARI
3060 REM      * A(2,1) ED A(2,0) = COEFFICIENTI DELL'EQUAZIONE
3080 REM      2. RISULTATI FORNITI
3090 REM      LE RADICI SONO: R(1)+I IM
3100 REM      R(2)-I IM
3110 REM      3. VARIABILI UTILIZZATE
3120 REM      IM,Q1,Q2,W

```

```

3130 REM      4. VETTORE E MATRICE UTILIZZATI
3140 REM      A(4,4), R(4)
3150 REM      *****
3160 W = A(2,1) * A(2,1) - 4 * A(2,0): REM W=DISCRIMINANTE DELL'EQUAZIONE

3170 Q1 = - A(2,1) / 2
3180 REM      CONTROLLO SUL SEGNO DI W
3190 ON ( SGN (W) + 2) GOTO 3210,3280,3340
3200 REM
3210 REM      W<0
3220 REM
3230 Q2 = SQR ( - W) / 2
3240 IM = Q2: REM      PARTE IMMAGINARIA
3250 R(1) = Q1: R(2) = Q1: REM      PARTI REALI
3260 RETURN
3270 REM
3280 REM      W=0
3290 REM
3300 R(1) = Q1: R(2) = Q1: REM      RADICE REALE DOPPIA
3310 IM = 0
3320 RETURN
3330 REM
3340 REM      W>0
3350 REM
3360 Q2 = SQR (W) / 2
3370 IM = 0: REM      DUE RADICI REALI
3380 R(1) = Q1 + Q2
3390 R(2) = Q1 - Q2
3400 RETURN

```

## 5 — ESEMPIO PRATICO

Calcolare le soluzioni dell'equazione  $x^4 + 4x^2 - 3x + 3 = 0$

```

RUN
RISOLUZIONE DI EQUAZIONI
DI SECONDO, TERZO O QUARTO GRADO

```

AUTORE: H. HAUT

REGOLE D'USO:

1. INTRODURRE IL GRADO DELL'EQUAZIONE

2. INTRODURRE I COEFFICIENTI  
PARTENDO DA QUELLO DELLA PIU'  
ALTA POTENZA DI X

3. LE SOLUZIONI SONO DATE SOTTO  
LA FORMA:  
(PARTE REALE) + I (PARTE IMMA-  
GINARIA)

PREMERE UN TASTO PER CONTINUARE  
QUALE E' IL GRADO DELL'EQUAZIONE? 4

```

COEFFICIENTE DI X^4=1
COEFFICIENTE DI X^3=0
COEFFICIENTE DI X^2=4
COEFFICIENTE DI X^1=3
COEFFICIENTE DI X^0=-3

```

```

RADICE 1=.271822931 +I 2.19841689
RADICE 2=.271822931 -I 2.19841689
RADICE 3=.555982246
RADICE 4=-1.09962811

```

7

RUN  
RISOLUZIONE DI EQUAZIONI  
DI SECONDO, TERZO O QUARTO GRADO

AUTORE:H.HAUT

REGOLE D'USO:

1. INTRODURRE IL GRADO DELL'EQUAZIONE

2. INTRODURRE I COEFFICIENTI  
PARTENDO DA QUELLO DELLA PIU'  
ALTA POTENZA DI X

3. LE SOLUZIONI SONO DATE SOTTO  
LA FORMA:  
(PARTE REALE) + I (PARTE IMMA-  
GINARIA)

PREMERE UN TASTO PER CONTINUARE  
QUALE E' IL GRADO DELL'EQUAZIONE? 4

COEFFICIENTE DI  $X^4=1$   
COEFFICIENTE DI  $X^3=0$   
COEFFICIENTE DI  $X^2=4$   
COEFFICIENTE DI  $X^1=-3$   
COEFFICIENTE DI  $X^0=3$

RADICE 1= .449702657 +I .731070323  
RADICE 2= .449702657 -I .731070323  
RADICE 3= -.449702657 +I 1.96723185  
RADICE 4= -.449702657 -I 1.96723185

tempo d'esecuzione: 3"5

memoria richiesta: 8286 bytes (senza REM : 2373)



## PROGRAMMA NUMERO 5

# INTEGRAZIONE NUMERICA

Il programma realizzato permette di scegliere tra cinque diversi metodi numerici di integrazione. I primi quattro permettono di integrare una funzione definita in un intervallo finito mentre il quinto permette di effettuare l'integrazione su un dominio il cui estremo superiore sia infinito.

## 1 — METODI NUMERICI

### 1.1 — Regola di Simpson con 2n intervalli

$$\int_a^b f(x) \, dx = \frac{h}{3} [f_0 + 4(f_1 + f_3 + \dots + f_{2n-1}) + 2(f_2 + f_4 + \dots + f_{2n-2}) + f_{2n}]$$

ove  $h = \frac{b-a}{2n}$  e  $f_i = f(x_i)$ ,  $x_i = a + ih$

### 1.2 — Regola di Bode (10 punti)

$$\int_a^b f(x) \, dx = \frac{5h}{299376} [16067 (f_0 + f_{10}) + 106300 (f_1 + f_9) - 48525 (f_2 + f_8) + 272400 (f_3 + f_7) - 260550 (f_4 + f_6) + 213684 (f_5 + f_5)]$$

ove  $h = \frac{b-a}{10}$  e  $f_i = f(x_i)$ ,  $x_i = a + ih$

### 1.3 — Regola di Gauss (12 punti)

$$\int_a^b f(x) \, dx = \frac{b-a}{2} \sum_{i=0}^{11} w_i f(y_i)$$

$$\text{ove } y_i = \left( \frac{b-a}{2} \right) x_i + \left( \frac{b+a}{2} \right)$$

con

$x_0 = -x_6 = .12523 \ 34085$	$w_0 = w_6 = .24914 \ 70458$
$x_1 = -x_7 = .36783 \ 14989$	$w_1 = w_7 = .23349 \ 25365$
$x_2 = -x_8 = .58731 \ 79542$	$w_2 = w_8 = .20316 \ 74267$
$x_3 = -x_9 = .76990 \ 26741$	$w_3 = w_9 = .16007 \ 83285$
$x_4 = -x_{10} = .90411 \ 72563$	$w_4 = w_{10} = .10693 \ 93259$
$x_5 = -x_{11} = .98156 \ 06342$	$w_5 = w_{11} = .04717 \ 53363$

### 1.4 — Regola di Chebyshev (9 punti)

$$\int_a^b f(x) \, dx = \frac{b-a}{9} \sum_{i=0}^8 f \left( \frac{b-a}{2} x_i + \frac{b+a}{2} \right)$$

ove

$x_0 = 0$
$x_1 = -x_5 = .16790 \ 61842$
$x_2 = -x_6 = .52876 \ 17831$
$x_3 = -x_7 = .60101 \ 86554$
$x_4 = -x_8 = .91158 \ 93077$

### 1.5 — Estremo superiore infinito

Si utilizza un caso particolare della regola di Gauss (12 punti) esposto nel paragrafo 1.3. Si ha:

$$\int_a^\infty f(x) \, dx = 2 \sum_{i=0}^{11} \frac{w_i}{(1+x_i)^2} f \left( \frac{2}{1+x_i} + a - 1 \right)$$

dove gli  $x_i$  ed i  $w_i$  sono definiti come nel paragrafo 1.3.

**Nota:** Conviene scegliere di volta in volta il metodo più conveniente. Il lettore potrà trovare a questo proposito numerose notizie nella letteratura consigliata. Una delle precauzioni da prendere consiste comunque nel verificare che la funzione integranda non presenti dei punti di discontinuità all'interno dell'intervallo di integrazione.

*Riferimenti bibliografici:* A2, B1, B2, C3, C4, H1, L1, L2, S3, S2, W1.





## 2 — TECNICA DI PROGRAMMAZIONE


Tutti i metodi sopra citati si prestano ad una programmazione lineare che non presenta eccessive difficoltà.


I valori  $x_i$ ,  $w_i$  e le ascisse di Chebyshev sono memorizzate nei vettori  $X(11)$  e  $W(11)$ ; A e B rappresentano gli estremi di integrazione.


### 3 — DIAGRAMMA A BLOCCHI



















## 4 — PROGRAMMA

```

1  REM          INTEGRAZIONE NUMERICA
2  REM
3  REM          AUTORE H.HAUT
4  REM
5  REM
6  REM
7  REM          DESCRIZIONE:
8  REM          IL PROGRAMMA PROPONE CINQUE METODI DIVERSI
9  REM          PER CALCOLARE L'INTEGRALE DI UNA FUNZIONE
10 REM          DEFINITA DALL'UTENTE. QUESTI METODI SONO:
11 REM          1.REGOLA DI SIMPSON
12 REM          2.REGOLA DI BODE (10 PUNTI)
13 REM          3.FORMULA DI GAUSS(12 PUNTI)
14 REM          4.FORMULA DI CHERYSHEV(9 PUNTI)
15 REM          5.FORMULA DI GAUSS(LIMITE SUPERIORE INFINITO)
16 REM
17 REM
18 REM          L'UTENTE DEVE DEFINIRE LA FUNZIONE F(X)
19 REM          ALLA LINEA 270
20 REM
21 REM
100 REM
110 REM          REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 9);"INTEGRAZIONE NUMERICA."
150 PRINT : PRINT TAB( 20);"AUTORE:H.HAUT": PRINT
160 PRINT "REGOLE D'USO:"
170 PRINT "      1.L'UTENTE DEVE DEFINIRE LA          FUNZIONE DA INT
      EGRARE ALLA LINEA          270 COME INDICATO:          27
      0 DEF FN F(X)=1/(1+X+X*X)
180 PRINT "      SE NON E' GIA' STATO FATTO          PREMIERE RESET,
      DEFINIRE LA          FUNZIONE E LANCIARE IL PROGRAMMA"
190 PRINT "      2.SCEGLIERE POI UNO DEI METODI          SEGUENTI:
          1=REGOLA DI SIMPSON          2=R
          EGOLA DI BODE          3=FORMULA DI GAUSS"
200 PRINT "      4=FORMULA DI CHERYSHEV          5=LIMITE SUPER
          IORE INFINITO"
210 PRINT "      3.DEFINIRE INFINE I LIMITI          D'INTEGRAZIONE"

220 UTAB (23): PRINT "PREMIERE UN TASTO PER CONTINUARE ": GET A$
230 HOME
240 REM
250 REM          DEFINIZIONE DELLA FUNZIONE
260 REM
270 DEF FN F(X) = 1 / (1 + X + X * X)
280 REM
290 REM          ISTRUZIONI DI DIMENSIONAMENTO
300 REM
310 DIM X(11),W(11),VC(5)
320 REM
330 REM          GESTIONE DEI DATI IN INPUT
340 REM
350 PRINT "SCEGLIERE IL TIPO DI CALCOLO :1=SIMPSON 2=BODE;3=GAUSS;4=CHERY
      SHEV;5=LIMITE SUP INF."
360 PRINT : INPUT "TIPO =" ;M
370 PRINT
380 REM
390 REM          PER LA REGOLA DI SIMPSON SI DEVE SCEGLIERE IL NUMERO DI
395 REM          INTERVALLI
400 REM
410 IF M = 1 THEN PRINT "CALCOLO CON 2N INTERVALLI ": INPUT " N=" ;N
420 INPUT "LIMITE INFERIORE =" ;A
430 REM
440 REM          PER IL LIMITE SUP.INFINITO, SI DEFINISCE SOLO IL LIMITE INF.

```

```

450 REM
460 IF M = 5 THEN GOTO 480
470 INPUT 'LIMITE SUPERIORE =';B
480 REM
490 REM      CHIAMATA DEI SOTTOPROGRAMMI SCELTI
500 REM
510 ON M GOSUB 1000,2000,3000,4000,5000
520 REM
530 REM      GESTIONE DEI RISULTATI
540 REM
550 PRINT : PRINT
560 PRINT 'VALORE DELL'INTEGRALE=';NT
570 END
1000 REM *****
1010 REM      SOTTOPROGRAMMA D'INTEGRAZIONE
1020 REM      REGOLA DI SIMPSON CON 2N INTERVALLI
1030 REM
1040 REM      REGOLE D'USO:
1050 REM          1.DATI NECESSARI
1060 REM              * N=NUMERO DEGLI INTERVALLI
1070 REM              * A=LIMITE INFERIORE
1080 REM              * B=LIMITE SUPERIORE
1090 REM              * F(X)=FUNZIONE DA INTEGRARE
1100 REM          2.RISULTATI FORNITI:
1110 REM              * NT=VALORE DELL'INTEGRALE
1120 REM          3.VARIABILI UTILIZZATE
1130 REM              A,B,H,I,N,N1,N2,NT,ST,X
1140 REM          4.FUNZIONE UTILIZZATA
1150 REM              F(X)
1160 REM *****
1170 NT = 0: REM      INIZIALIZZAZIONE DELL'INTEGRALE
1180 H = (B - A) / (2 * N): REM      DEFINIZIONE DEL PASSO
1190 N1 = 2 * N - 1: N2 = N1 - 1
1200 REM
1210 REM      SOMMATORIA SUI PUNTI DISPARI
1220 REM
1230 ST = 0: REM      AZZERAMENTO DELLA SOMMA PROVVISORIA
1240 FOR I = 1 TO N1 STEP 2
1250 X = A + (I * H)
1260 ST = ST + FN F(X)
1270 NEXT I
1280 NT = ST * 4
1290 REM
1300 REM      SOMMATORIA SUI PUNTI PARI
1310 REM
1320 ST = 0: REM      RIAZZERAMENTO DELLA SOMMA PROVVISORIA
1330 FOR I = 2 TO N2 STEP 2
1340 X = A + (I * H)
1350 ST = ST + FN F(X)
1360 NEXT I
1370 REM
1380 REM      CALCOLO DI NT
1390 REM
1400 NT = NT + 2 * ST + FN F(A) + FN F(B)
1410 NT = NT * H / 3
1420 RETURN
2000 REM *****
2010 REM      SOTTOPROGRAMMA D'INTEGRAZIONE
2020 REM      REGOLA DI RODE (10 PUNTI)
2030 REM
2040 REM      REGOLE D'USO
2050 REM          1.DATI NECESSARI:
2060 REM              *A=LIMITE INFERIORE
2070 REM              *B=LIMITE SUPERIORE
2080 REM              *F(X)=FUNZIONE DA INTEGRARE
2090 REM          2.RISULTATI FORNITI:
2100 REM              *NT=VALORE DELL'INTEGRALE

```

```

2110 REM      3.VARIABILI UTILIZZATE
2120 REM      A,B,H,I,J,NT,ST,X
2130 REM      4.VEETTORE UTILIZZATO
2140 REM      VC(5)
2150 REM      5.FUNZIONE UTILIZZATA
2160 REM      F(X)
2170 REM      *****
2180 NT = 0: REM      INIZIALIZZAZIONE DELL'INTEGRALE
2190 H = (B - A) / 10: REM      DEFINIZIONE DEL PASSO
2200 REM
2210 REM      DEFINIZIONE DEL VETTORE VC
2220 REM      VC(I)=COEFFICIENTE DI F(T)+F(10-I) DALLA FORMULA DEL TESTO
2230 REM
2240 VC(0) = .2683414836:VC(1) = 1.775359414:VC(2) = - .8104357063:VC(3) =
4.5494662883:VC(4) = - 4.351551227:VC(5) = 3.568823152
2250 REM
2260 REM      CALCOLO DELLA SOMMA
2270 REM
2280 FOR I = 0 TO 5
2290 J = 10 - I
2300 X = A + (I * H)
2310 ST = FN F(X) /
2320 X = A + (J * H)
2330 ST = ST + FN F(X): REM      ST=F(I)+F(10-I) (CFR TESTO)
2340 NT = NT + VC(I) * ST: REM      CALCOLO DI NT
2350 NEXT I
2360 NT = NT * H
2370 RETURN
3000 REM      *****
3010 REM      SOTTOPROGRAMMA D'INTEGRAZIONE
3020 REM      FORMULA DI GAUSS (12 PUNTI)
3030 REM
3040 REM      REGOLE D'USO:
3050 REM      1.DATI NECESSARI:
3060 REM      * A=LIMITE INFERIORE
3070 REM      * B=LIMITE SUPERIORE
3080 REM      * F(X)=FUNZIONE DA INTEGRARE
3090 REM      2.RISULTATI FORNITI:
3100 REM      * NT=VALORE DELL'INTEGRALE
3110 REM      3.VARIABILI UTILIZZATE:
3120 REM      A,B,I,NT,Y
3130 REM      4.VEETTORE UTILIZZATO:
3140 REM      X(11),W(11)
3150 REM      5.FUNZIONE UTILIZZATA
3160 REM      F(X)
3170 REM      6.SOTTOPROGRAMMA CHIAMATO:
3180 REM      6000
3190 REM      *****
3200 NT = 0: REM      INIZIALIZZAZIONE DELL'INTEGRALE
3210 REM
3220 REM      CHIAMATA DEL SOTTOPROGRAMMA 3000
3230 REM      CHE DEFINISCE I PUNTI PER CALCOLARE Y(I) (CFR TESTO) E
3235 REM      I PESI W(I)
3240 REM
3250 GOSUB 6000
3260 REM
3270 REM      CALCOLO DELLA SOMMA
3280 REM
3290 FOR I = 0 TO 11
3300 Y = (B - A) * X(I) / 2: Y = Y + (B + A) / 2: REM      Y=PUNTI DI CALCOLO
PER F
3310 NT = NT + W(I) * FN F(Y): REM      CALCOLO DELL'INTEGRALE
3320 NEXT I
3330 NT = NT * (B - A) / 2
3370 RETURN
4000 REM      *****
4010 REM      SOTTOPROGRAMMA D'INTEGRAZIONE

```

```

4020 REM      FORMULA DI CHERYSHEV (9 PUNTI)
4030 REM
4040 REM      REGOLE D'USO:
4050 REM      1.DATI NECESSARI:
4060 REM          * A=LIMITE INFERIORE
4070 REM          * B=LIMITE SUPERIORE
4080 REM          * F(X)=FUNZIONE DA INTEGRARE
4090 REM      2.RISULTATI FORNITI:
4100 REM          * NT=VALORE DELL'INTEGRALE
4110 REM      3.VARIABILI UTILIZZATE:
4120 REM          A,B,I,NT,Y
4130 REM      4.VETTORE UTILIZZATO:
4140 REM          X(B)
4150 REM      5.FUNZIONE UTILIZZATA:
4160 REM          F(X)
4170 REM      *****
4180 NT = 0: REM      INIZIALIZZAZIONE DELL'INTEGRALE
4190 REM
4200 REM      DEFINIZIONE DEGLI X(I) DEL TESTO
4210 REM
4220 X(0) = 0: X(1) = .1679061842: X(2) = .5287617831: X(3) = .6010186554: X(4)
    ) = .9115893077: X(5) = - X(1): X(6) = - X(2): X(7) = - X(3): X(8) = -
    X(4)
4230 REM
4240 REM      CALCOLO DELLA SOMMA
4250 REM
4260 FOR I = 0 TO 8
4270 Y = (B - A) * X(I) / 2: Y = Y + (B + A) / 2: REM      DEFINIZIONE DEI PUN-
    TI IN CUI CALCOLARE F
4280 NT = NT + FN F(Y): REM      CALCOLO DELL'INTEGRALE
4290 NEXT I
4300 NT = NT * (B - A) / 9
4310 RETURN
5000 REM      *****
5010 REM      SOTTOPROGRAMMA D'INTEGRAZIONE
5020 REM      CON LIMITE SUPERIORE INFINITO
5030 REM      FORMULA DI GAUSS (12 PUNTI)
5040 REM
5050 REM      REGOLE D'USO
5060 REM      1.DATI NECESSARI:
5070 REM          * A=LIMITE INFERIORE
5080 REM          * F(X)=FUNZIONE DA INTEGRARE
5090 REM      2.RISULTATI FORNITI:
5100 REM          * NT=VALORE DELL'INTEGRALE
5110 REM      3.VARIABILI UTILIZZATE:
5120 REM          A,I,NT,Y,YT
5130 REM      4.VETTORE UTILIZZATO:
5140 REM          X(I),W(I)
5150 REM      5.FUNZIONE UTILIZZATA:
5160 REM          F(X)
5170 REM      6.SOTTOPROGRAMMA CHIAMATO:
5180 REM          6000
5190 REM      *****
5200 NT = 0: REM      INIZIALIZZAZIONE DELL'INTEGRALE
5210 REM
5220 REM      CHIAMATA DEL SOTTOPROGRAMMA CHE DEFINISCE
5230 REM      GLI X(I) E W(I) (CFR TESTO)
5240 REM
5250 GOSUB 6000
5260 REM
5270 REM      CALCOLO DELLA SOMMA
5280 REM
5290 FOR I = 0 TO 11
5300 Y = (2 / (1 + X(I))) + A - 1: REM      PUNTI IN CUI CALCOLARE F
5310 YT = FN F(Y)
5320 YT = YT / ((1 + X(I)) ^ 2): REM      VALORE PROVVISORIO DI UN TERMINE
    DELLA SOMMA

```

```

5330 NT = NT + W(I) * YT: REM    CALCOLO DELL'INTEGRALE
5340 NEXT I
5350 NT = NT * 2
5360 RETURN
6000 REM *****
6010 REM    SOTTOPROGRAMMA DI INIZIALIZZAZIONE DEI VETTORI
6020 REM    X(I) E W(I) NECESSARI PER L'INTEGRAZIONE
6030 REM    SECONDO LA FORMULA DI GAUSS 12 PUNTI
6040 REM
6050 REM    SOLO GLI ELEMENTI
6060 REM    X(11) E W(11)
6070 REM    SONO UTILIZZATI
6080 REM *****
6090 REM
6100 REM    INIZIALIZZAZIONE DEGLI X(I)
6110 REM
6120 X(0) = .1252334085: X(1) = .3678314989: X(2) = .5873179542: X(3) = .7699
026741: X(4) = .9041172563: X(5) = .9815606342
6130 X(6) = - X(0): X(7) = - X(1): X(8) = - X(2): X(9) = - X(3): X(10) =
X(4): X(11) = - X(5)
6140 REM
6150 REM    INIZIALIZZAZIONE DEI W(I)
6160 REM
6170 W(0) = .2491470458: W(1) = .2334925365: W(2) = .2031674267: W(3) = .1600
783285: W(4) = .1069393259: W(5) = .0471753363
6180 W(6) = W(0): W(7) = W(1): W(8) = W(2): W(9) = W(3): W(10) = W(4): W(11) =
W(5)
6190 RETURN

```

## 5 — ESEMPIO PRATICO

Si voglia calcolare l'integrale tra 0 e 1 della funzione:

$$f(x) = 1/(1 + x + x^2)$$

Il risultato esatto di tale integrale è  $\pi/3 \sqrt{3} = 0.604599788$ .

I vari metodi proposti portano ai seguenti risultati:

1. Simpson (n = 10):	.604599557
2. Bode:	.604599787
3. Gauss:	.604599788
4. Chebyshev:	.604599734

L'esempio illustra la risoluzione con il metodo di Gauss.

RUN

INTEGRAZIONE NUMERICA.

AUTORE:H.HAUT

REGOLE D'USO:

1. L'UTENTE DEVE DEFINIRE LA FUNZIONE DA INTEGRARE ALLA LINEA 270 COME INDICATO:
- ```
270 DEF FN F(X)=1/(1+X+X*X)
```

SE NON E' GIA' STATO FATTO  
PREMERE RESET, DEFINIRE LA  
FUNZIONE E LANCIARE IL PROGRAMMA  
2. SCEGLIERE POI UNO DEI METODI  
SEGUENTI:  
1=REGOLA DI SIMPSON  
2=REGOLA DI RODE  
3=FORMULA DI GAUSS  
4=FORMULA DI CHEBYSHEV  
5=LIMITE SUPERIORE INFINITO  
3. DEFINIRE INFINE I LIMITI  
D'INTEGRAZIONE

PREMERE UN TASTO PER CONTINUARE  
SCEGLIERE IL TIPO DI CALCOLO :1=SIMPSON  
2=RODE;3=GAUSS;4=CHERYSHEV;5=LIMITE SUP  
INF.

TIPO =3

LIMITE INFERIORE =0  
LIMITE SUPERIORE =1

VALORE DELL'INTEGRALE=.604599788

tempo d'esecuzione: 1"

memoria richiesta: 8186 bytes (senza REM : 2526).

## PROGRAMMA NUMERO 6

# EQUAZIONI DIFFERENZIALI DEL PRIMO E SECONDO ORDINE

Questo programma permette di calcolare il valore assunto da una funzione  $y(x)$  in un generico punto (oppure di tabulare la funzione stessa) quando questa è definita tramite una equazione differenziale del primo o secondo ordine, con le condizioni iniziali in un punto  $x_0$ . Il metodo di risoluzione adottato è quello di Runge-Kutta (quarto ordine).

## 1 — METODI NUMERICI

### 1.1 — Equazione differenziale del primo ordine

Si voglia risolvere l'equazione:

$$(1) \quad y'(x) = \frac{dy}{dx} = f(x, y)$$

sotto la condizione iniziale  $y(x_0) = y_0$ , calcolando il valore di  $y$  in  $x = x_f$ .

Seguendo il metodo di Runge-Kutta, si divide l'intervallo  $x_f - x_0$  in segmenti di lunghezza  $h$  e si tabula la funzione  $y$  da  $x_0$  a  $x_f$ , con passo  $h$ , utilizzando il seguente algoritmo:

$$(2) \quad y_{n+1} = y_n + 1/8 (k_1 + 3k_2 + 3k_3 + k_4) + (\text{errore dell'ordine } h^5)$$

con la convenzione:

$$y_n = y(x_n) \quad \text{con } x_n = x_0 + nh$$

I  $k_i$  sono dati da:

$$\begin{aligned}
 (3) \quad k_1 &:= h f(x_n, y_n) \\
 k_2 &:= h f(x_n + h/3, y_n + k_1/3) \\
 k_3 &:= h f(x_n + 2h/3, y_n - k_1/3 + k_2) \\
 k_4 &:= h f(x_n + h, y_n + k_1 - k_2 + k_3).
 \end{aligned}$$

## 1.2 — Equazione differenziale del secondo ordine

Come nel caso precedente, la soluzione di:

$$(4) \quad y''(x) = \frac{d^2 y}{dx^2} = f(x, y, y')$$

sotto le condizioni iniziali:

$$y(x_0) = y_0 \text{ e } y'(x_0) = y'_0$$

è tabulata da  $x_0$  ad  $x_f$  con passo  $h$  nel modo seguente:

$$\begin{aligned}
 (5) \quad y_{n+1} &:= y_n + h [y'_n + 1/6 (l_1 + l_2 + l_3)] \\
 y'_{n+1} &:= y'_n + 1/6 (l_1 + 2l_2 + 2l_3 + l_4)
 \end{aligned}$$

ove l'errore di approssimazione è dell'ordine di  $h^5$ .

Si ha:

$$\begin{aligned}
 (6) \quad l_1 &= h f(x_n, y_n, y'_n) \\
 l_2 &= h f(x_n + h/2, y_n + h/2 y'_n, y'_n + l_1/2) \\
 l_3 &= h f(x_n + h/2, y_n + h/2 y'_n + h/4 l_1, y'_n + l_2/2) \\
 l_4 &= h f(x_n + h, y_n + h y'_n + h/2 l_2, y'_n + l_3)
 \end{aligned}$$

*Riferimenti bibliografici:* A2, B1, B2, C3, H1, L1, S2, S3.

## 2 — TECNICA DI PROGRAMMAZIONE

Il programma realizzato permette all'utente di definire le due equazioni differenziali  $y' = F1(x, y)$  e  $y'' = F2(x, y, y')$ , di scegliere le condizioni iniziali  $(x_0, y_0)$  e  $(x_0, y_0, y'_0)$  e di determinare il passo  $H$  da utilizzare nonché il valore  $x_f = XF$ .



Per calcolare il valore di  $y$  in  $x_f$  bisogna scegliere il passo  $H$  in modo tale che  $x_f = x_0 + nH$  (con  $n$  intero).

Per evitare di calcolare le funzioni  $F1$  ed  $F2$  con dei sottoprogrammi, si possono definire nel modo seguente:

1) DEF FN  $F1(X) = f(X, Y)$

il calcolo di  $C = F1(A, B)$  può allora essere svolto con le istruzioni:

$Y = B$   
 $C = FN F1(A)$

2) DEF FN  $F2(X) = f(x, y, z)$


in cui  $z$  rappresenta  $y'$  e il calcolo di  $D = F2(A, B, C)$  può essere realizzato nel modo seguente:


$Y = B$   
 $Z = C$   
 $D = FN F2(A)$


Le risoluzioni delle equazioni differenziali secondo gli algoritmi (2) e (5) sono realizzate tramite i sottoprogrammi 2000 e 3000.


Il programma infine permette all'utente di scegliere tra il calcolo dei soli valori finali  $(x_f, y_f, y'_f)$  e la stampa della tabella dei valori assunti da  $x, y, y'$  ad ogni passo. (Ovviamente quest'ultima richiesta richiederà più tempo per essere esaudita).

### 3 — DIAGRAMMA A BLOCCHI









## 4 — PROGRAMMA

```

1  REM      EQUAZIONI DIFFERENZIALI
2  REM      DEL PRIMO E SECONDO ORDINE
3  REM
4  REM      AUTORE:H.HAUT
5  REM
6  REM
7  REM      DESCRIZIONE:
8  REM      IL PROGRAMMA PERMETTE DI RISOLVERE LE
9  REM      SEGUENTI EQUAZIONI DIFFERENZIALI
10 REM      1.Y'(X)=F(X,Y)
11 REM      2.Y''(X)=F(X,Y,Y')
12 REM      CON IL METODO DI RUNGE-KUTTA
13 REM
14 REM      *****
15 REM
16 REM
100 REM
110 REM      REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 7);"EQUAZIONI DIFFERENZIALI"
150 PRINT TAB( 7);"DEL 1MO E 2DO ORDINE."
160 PRINT : PRINT TAB( 20);"AUTORE:H.HAUT"
170 PRINT
180 PRINT "1.L'EQUAZIONE DIFFERENZIALE DI 1MO      ORDINE Y'=F1(X,Y) DEVE
E ESSERE DEFINITA"
190 PRINT "  ALLA LINEA 320 SECONDO IL MODELLO:"
200 PRINT "    DEF F1(X)=X^2-Y^2 ,DOVE      F1(X)=Y'"
210 PRINT "2.L'EQUAZIONE DIFFERENZIALE DI
211 PRINT "  SECONDO ORDINE:"; PRINT "  Y''=F2(X,Y,Y')"
212 PRINT "  DEVE ESSERE DEFINITA ALLA LINEA"
220 PRINT "  360 SECONDO IL MODELLO:"
230 PRINT "    DEF F2(X)=X*Z-Y ,DOVE      F2(X)=Y'
      Z=Y'"
240 PRINT "3.INSERIRE POI L'ORDINE DELL'EQUAZIONE"
250 PRINT "4.DEFINIRE LE CONDIZIONI INIZIALI, IL      VALORE FINALE DI X E
DI IL PASSO VOLUTO"
260 VTab 22: PRINT "PREMERE UN TASTO PER CONTINUARE " : GET A$
270 HOME
280 REM
290 REM      DEFINIZIONI DELLE EQUAZIONI
300 REM
310 REM      PRIM'ORDINE
320 DEF FN F1(X) = 4 * X - 2 * Y
330 REM      F1(X)=Y'
340 REM
350 REM      SECONDO ORDINE
360 DEF FN F2(X) = - 4 * Y - 4 * Z
370 REM      F2(X)=Y''
380 REM      Z=Y'
390 REM
400 REM      GESTIONE DEI RISULTATI
410 REM
420 INPUT "ORDINE DELL'EQUAZ.= ";DE: PRINT : PRINT
430 ON DE GOTO 470,700: REM      SALTO A SECONDA DELL'ORDINE
440 REM
450 REM      PRIMO ORDINE
460 REM
470 PRINT "CONDIZIONI INIZIALI: "
480 INPUT "  X0=";X0
490 INPUT "  Y0=";Y0
500 PRINT : INPUT "CALCOLO NEL PUNTO X=";XF: PRINT
510 PRINT : PRINT "L'APPROSSIMAZIONE E' DI H^5!"
520 INPUT "  PASSO H=";H

```

```

530 REM
540 REM POSSIBILITA' DI TABULARE
550 REM
560 INPUT "VOLETE UNA TABELLA?";A$
570 IF A$ = 'S' THEN GOTO 950
580 REM
590 REM CHIAMATA DEL SOTTOPROGRAMMA
600 REM
610 GOSUB 2000
620 REM
630 REM GESTIONE DEI RISULTATI (PRIMO ORDINE)
640 REM
650 PRINT "RISULTATI:"
660 PRINT " X =";X
670 PRINT " Y =";Y0
680 PRINT " Y' =";YF
690 END
700 REM
710 REM SECONDO ORDINE
720 REM
730 PRINT "CONDIZIONI INIZIALI:"
740 INPUT " X0=";X0
750 INPUT " Y0=";Y0
760 INPUT " Y'0=";Z0
770 PRINT : INPUT "CALCOLO NEL PUNTO X=";XF
780 PRINT : PRINT "L'APPROSSIMAZIONE E' DI H^5!"
790 INPUT " PASSO H=";H
800 PRINT
810 INPUT "VOLETE UNA TABELLA?";A$: REM POSSIBILITA' DI TABULARE
820 IF A$ = 'S' THEN GOTO 950
830 REM
840 REM CHIAMATA DEL SOTTOPROGRAMMA
850 REM
860 GOSUB 3000
870 REM
880 REM GESTIONE DEI RISULTATI (SECONDO ORDINE)
890 REM
900 PRINT "RISULTATI:"
910 PRINT " X=";X
920 PRINT " Y=";Y0
930 PRINT " Y'=";Z0
940 END
950 REM
960 REM TABULATO
970 REM
980 HOME
990 PRINT " X "; TAB( 8); " Y "; TAB( 22); " Y' "
1000 PRINT "----"; TAB( 8); "----"; TAB( 22); "----"
1010 XI = X0
1020 N = (XF - XI) / H: REM NUMERO DI PUNTI INTERMEDI DA CALCOLARE
1030 REM
1040 REM EVENTUALE CORREZIONE DI N PER ARROTONDAMENTO
1050 IF INT (N) < > N THEN N = N + 1
1060 REM
1070 FOR I = 0 TO N - 1
1080 X0 = XI + I * H
1090 XF = XI + (I + 1) * H
1100 REM CALCOLO DI UN PUNTO INTERMEDIO
1110 ON DE GOSUB 2000,3000
1120 REM STAMPA DEL TABULATO
1130 IF DE = 2 THEN YF = Z0
1140 PRINT XF;" "; TAB( 8);Y0;" "; TAB( 22);YF
1150 REM PROSECUZIONE DELLA TABELLA
1160 NEXT I
1170 END
2000 REM *****
2010 REM SOTTOPROGRAMMA PER RISOLVERE L'EQUAZIONE DIFFERENZIALE:

```

```

2020 REM      Y'=F1(X,Y)
2030 REM      CON IL METODO DI RUNGE-KUTTA
2040 REM
2050 REM      REGOLE PER L'USO DEL SOTTOPROGRAMMA
2060 REM
2070 REM      1.DATI NECESSARI:
2080 REM          * F1(X,Y) DEFINITA DA FN F1(X)
2090 REM          SECONDO IL MODELLO:
2100 REM              DEF FN F1(X)=4*X-2*Y
2110 REM              (IL CALCOLO DI C=F1(A,B) SI EFFETTUA
2120 REM                  CON Y=B;C=FN F1(A) )
2130 REM          * X0=VALORE INIZIALE DI X
2140 REM          * Y0=VALORE INIZIALE DI Y
2150 REM          * H=PASSO
2160 REM              (L'INTERVALLO XF-XI DEVE CONTENERE
2170 REM                  UN NUMERO INTERO DI PASSI)
2180 REM          * XF=VALORE FINALE DEL CALCOLO
2190 REM      2.RISULTATI FORNITI:
2200 REM          * X=VALORE FINALE DI X
2210 REM          * Y0=VALORE DI Y IN X
2220 REM          * YP=VALORE DI Y' IN X
2230 REM      3.VARIABILI UTILIZZATE:
2240 REM          H,K1,K2,K3,K4,X0,X,XF,Y0,Y,YD,YP
2250 REM      4.FUNZIONE UTILIZZATA:
2260 REM          F1(X)
2270 REM      *****
2280 Y = Y0: REM      INIZIALIZZAZIONE DI Y
2290 FOR X = X0 TO XF - H / 10 STEP H
2300 YD = FN F1(X): REM      CALCOLO DI Y'=F(X,Y0)
2310 REM      CALCOLO DI K1
2320 K1 = H * YD
2330 REM      CALCOLO DI K2
2340 Y = Y0 + K1 / 3
2350 K2 = H * FN F1(X + H / 3)
2360 REM      CALCOLO DI K3
2370 Y = Y0 - K1 / 3 + K2
2380 K3 = H * FN F1(X + 2 * H / 3)
2390 REM      CALCOLO DI K4
2400 Y = Y0 + K1 - K2 + K3
2410 K4 = H * FN F1(X + H)
2420 REM      CALCOLO DI Y(N+1) (CFR TESTO)
2430 Y0 = Y0 + (K1 + 3 * K2 + 3 * K3 + K4) / 8
2440 Y = Y0: REM      CALCOLO DEL VALORE DI Y
2450 YP = FN F1(X + H)
2460 NEXT X
2470 REM      ALL'USCITA DAL CICLO,X=XF,Y0=YP E YP=Y'F
2480 RETURN
3000 REM      *****
3010 REM      SOTTOPROGRAMMA PER LA RISOLUZIONE DELL'EQUAZIONE DIFFERENZIALE
3020 REM          Y''=F2(X,Y,Y')
3030 REM      CON IL METODO DI RUNGE-KUTTA
3040 REM
3050 REM      REGOLE PER L'USO DEL SOTTOPROGRAMMA
3060 REM
3070 REM      1.DATI NECESSARI:
3080 REM          * F2(X,Y,Y') DEFINITA DA FN F2(X)
3090 REM          SECONDO IL MODELLO:
3100 REM              DEF FN F2(X)=X*XZ-Y
3110 REM              CON Z=Y'
3120 REM              (IL CALCOLO DI D=F2(A,B,C) SI EFFETTUA CON:
3130 REM                  Y=B;Z=C;D=FN F2(A) )
3140 REM          * X0=VALORE INIZIALE DI X
3150 REM          * Y0=VALORE INIZIALE DI Y
3160 REM          * Z0=VALORE INIZIALE DI Y'
3170 REM          * H=PASSO
3180 REM              (L'INTERVALLO XF,XI DEVE CONTENERE
3190 REM                  UN NUMERO INTERO DI PASSI)

```



```

3200 REM      * XF=VALORE FINALE DA CALCOLARE
3210 REM      2.RISULTATI FORNITI:
3220 REM      * X=VALORE FINALE DI X
3230 REM      * Y0=VALORE DI Y IN X
3240 REM      * Z0=VALORE DI Y' IN X
3250 REM      3.VARIABILI UTILIZZATE:
3260 REM      H,I1,I2,I3,I4,X0,X,XF,Y0,Y,Z0,Z
3270 REM      4.FUNZIONE UTILIZZATA:
3280 REM      F2(X)
3290 REM      *****
3300 FOR X = X0 TO XF - H / 10 STEP H
3310 REM      CALCOLO DI I1
3320 Y = Y0
3330 Z = Z0
3340 I1 = H * FN F2(X)
3350 REM      CALCOLO DI I2
3360 Y = Y0 + H * Z0 / 2
3370 Z = Z0 + I1 / 2
3380 I2 = H * FN F2(X + H / 2)
3390 REM      CALCOLO DI I3
3400 Z = Z0 + I2 / 2
3410 Y = Y + H * I1 / 4
3420 I3 = H * FN F2(X + H / 2)
3430 REM      CALCOLO DI I4
3440 Y = Y0 + H * Z0 + H * I2 / 2
3450 Z = Z0 + I3
3460 I4 = H * FN F2(X + H)
3470 REM      CALCOLO DEL VALORE DI Y0
3480 Y0 = Y0 + H * (Z0 + (I1 + I2 + I3) / 4)
3490 REM      CALCOLO DEL VALORE DI Z0
3500 Z0 = Z0 + (I1 + 2 * I2 + 2 * I3 + I4) / 6
3510 NEXT X
3520 REM
3530 REM      ALL'USCITA DAL CICLO,X=XF,Y0=YF E Z0=Y'F
3540 RETURN

```

## 5 — ESEMPIO PRATICO

L'esempio mostra la tabulazione tra 0 e 1 delle funzioni  $y(x)$  e  $y'(x)$ , definite tramite l'equazione differenziale del primo ordine  $y' = 4x - 2y$  e soddisfacenti alle condizioni iniziali  $x = 0$  e  $y = -2$ , realizzata con passo .5.

La soluzione esatta,

$y = 2x - 1 - e^{-2x}$ , dà per  $x = 1$  il valore  $y = .864664716$  e il valore  $y' = 2.27067056$ .

```

RUN
EQUAZIONI DIFFERENZIALI
DEL 1MO E 2DO ORDINE.

```

AUTORE:H.HAUT

```

1.L'EQUAZIONE DIFFERENZIALE DI 1MO
ORDINE Y'=F1(X,Y) DEVE ESSERE DEFINITA
ALLA LINEA 320 SECONDO IL MODELLO:
DEF F1(X)=X^2-Y^2 ,DOVE
F1(X)=Y'

```

```

2.L'EQUAZIONE DIFFERENZIALE DI
SECONDO ORDINE:
Y''=F2(X,Y,Y'')
DEVE ESSERE DEFINITA ALLA LINEA
360 SECONDO IL MODELLO:
DEF F2(X)=X*Z-Y ,DOVE
F2(X)=Y''
Z=Y'
3.INSERIRE POI L'ORDINE DELL'EQUAZIONE
4.DEFINIRE LE CONDIZIONI INIZIALI, IL
VALORE FINALE DI X ED IL PASSO VOLUTO
PREMERE UN TASTO PER CONTINUARE
ORDINE DELL'EQUAZ.= 1

```

```

CONDIZIONI INIZIALI:
X0=0
Y0=-2

```

CALCOLO NEL PUNTO X=1

```

L'APPROSSIMAZIONE E' DI H^5!
PASSO H=.05

```

VOLETE UNA TABELLA?S

| X   | Y           | Y'         |
|-----|-------------|------------|
| --- | ---         | ----       |
| .05 | -1.8048375  | 3.809675   |
| .1  | -1.6187309  | 3.6374618  |
| .15 | -1.44081842 | 3.48163685 |
| .2  | -1.27032029 | 3.34064058 |
| .25 | -1.10653094 | 3.21306187 |
| .3  | -.948811935 | 3.09762387 |
| .35 | -.796585619 | 2.99317124 |
| .4  | -.64932929  | 2.89865858 |
| .45 | -.506569991 | 2.81313998 |
| .5  | -.367879775 | 2.73575955 |
| .55 | -.232871415 | 2.66574283 |
| .6  | -.101194539 | 2.60238908 |
| .65 | .027467886  | 2.54506423 |
| .7  | .153402723  | 2.49319455 |
| .75 | .276869537  | 2.44626093 |
| .8  | .398103189  | 2.40379362 |
| .85 | .517316195  | 2.36536761 |
| .9  | .634700842  | 2.33059832 |
| .95 | .750431124  | 2.29913775 |
| 1   | .864664472  | 2.27067106 |

tempo d'esecuzione: 4"

memoria richiesta: 6061 bytes (senza REM : 1762).

## PROGRAMMA NUMERO 7

# GRAFICO DI UNA FUNZIONE

Questo programma utilizza le funzioni video-grafiche del sistema APPLE II PLUS per rappresentare una funzione definita dall'utente in un intervallo a piacere. Le scale sono realizzate in modo da sfruttare al massimo i  $270 \times 150$  punti dello schermo.

### 1 — TECNICA DI PROGRAMMAZIONE

Dopo aver definito la funzione  $f(x)$  che si vuole rappresentare e gli estremi dell'intervallo di definizione  $XI$  e  $XF$ , si utilizza il sottoprogramma 1000 per calcolare il valore assunto dalla funzione stessa in 270 punti compresi tra  $XI$  e  $XF$ . Tali valori vengono memorizzati in un vettore  $Y(I)$  e tra essi si sceglie il valore massimo  $YF$  e il valore minimo  $YI$ . Si sceglie infine la scala delle ordinate in modo da sfruttare in toto i 150 punti verticali dello schermo.

La corrispondenza tra i valori  $X, Y$  calcolati ed i valori  $XP, YP$  che danno le coordinate grafiche è realizzata dalle seguenti formule:


$$XP = 270 + (X - XF) \frac{270}{(XF - XI)}$$


$$YP = 150 - (Y - YI) \frac{150}{(YF - YI)}$$


Il grafico della funzione viene così eseguito all'interno di un quadrato graduato, contenente anche gli assi cartesiani nel caso in cui  $XI < 0 < XF$  oppure  $YI < 0 < YF$ .


Viene inoltre lasciata all'utente la possibilità di utilizzare anche un secondo schermo supplementare.

2 — DIAGRAMMA A BLOCCHI









### 3 — PROGRAMMA

```

1  REM          GRAFICO DI UNA FUNZIONE F(X)
2  REM
3  REM          AUTORE:H.HAUT
4  REM
5  REM
6  REM          DESCRIZIONE
7  REM          IL PROGRAMMA PERMETTE DI DISEGNARE IL GRAFICO
8  REM          DI UNA FUNZIONE F(X) DEFINITA DALL'UTENTE
9  REM          IN UN INTERVALLO A PIACERE
10 REM
11 REM *****
100 REM
110 REM  REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 7);"GRAFICO DI UNA FUNZIONE"
150 PRINT : PRINT TAB( 20);"AUTORE:H.HAUT"
160 PRINT
170 PRINT "1.IL PROGRAMMA REALIZZA IL GRAFICO DI          UNA FUNZIONE F(X) IN
    UN QUADRATO DI          (270*150) PUNTI."
180 PRINT
190 PRINT " LE SCALE,GLI ASSI E IL PASSO SONO          CALCOLATI DAL PROGRA
    MMA"
200 PRINT
210 PRINT "2.LA FUNZIONE F(X) DEVE ESSERE DEFINITA          DALL'UTENTE ALLA LI
    NEA 340 SECONDO          IL MODELLO:"
220 PRINT "          340 DEF F(X)=SIN(X)*COS(X)"
230 PRINT " SE NON E' STATO ANCORA FATTO PREMERE          IL TASTO RESET, DEFI
    NIRE LA FUNZIONE          E BATTERE RUN"
240 PRINT
250 PRINT "3.INTRODURRE INFINE I VALORI INIZIALE E          FINALE DELLA VARIABI
    LE X CHE DELIMITA-          NO L'INTERVALLO DI DEFINIZIONE"
260 UTAB 23: PRINT "PREMERE UN TASTO PER CONTINUARE": GET A$
270 REM
280 REM  ISTRUZIONI DI DIMENSIONAMENTO
290 REM
300 DIM Y(270)
310 REM
320 REM  DEFINIZIONE DELLA FUNZIONE
330 REM
340 DEF FN F(X) = (X - 1) * (X + 1) * (X - 2) * SIN (X)
350 REM
360 REM  GESTIONE DEI DATI IN INPUT.
370 REM
380 HOME : INPUT "X INIZIALE=";XI
390 INPUT "X FINALE =" ;XF
400 REM
410 REM  CHIAMATA DEL SOTTOPROGRAMMA
420 REM
430 GOSUB 1000
440 REM
450 REM  GRIGLIA OPZIONALE
460 REM
470 INPUT "GRIGLIA (S O N)";A$
480 UTAB 1
490 IF A$ = "N" THEN END
500 REM
510 REM  TRACCIA DELLA GRIGLIA
520 REM
530 FOR I = 1 TO 9
540 XP = 27 * I:YP = 159 - 15 * I
550 HPLOT 0,YP TO 270,YP
560 HPLOT XP,9 TO XP,159.
570 NEXT I

```



```

580 FND
1000 REM *****
1010 REM SOTTOPROGRAMMA PER IL GRAFICO DI UNA FUNZIONE
1020 REM
1030 REM REGOLE PER L'USO DEL SOTTOPROGRAMMA
1040 REM
1050 REM 1.DATI NECESSARI:
1060 REM * XI=VALORE INIZIALE DI X
1070 REM * XF=VALORE FINALE DI X
1080 REM * F(X)=FUNZIONE DI CUI SI VUOLE IL GRAFICO
1090 REM DA XI A XF
1100 REM 2.RISULTATI FORNITI:
1110 REM GRAFICO DI F(X) SU UNA GRIGIA DI
1120 REM 270*150 PUNTI
1130 REM 3.VARIANTI UTILIZZATE:
1140 REM FX,XF,XI,XL,YP,YF,YI,YL,YP
1150 REM 4.VEETTORE UTILIZZATO:
1160 REM Y(270)
1170 REM 5.FUNZIONE UTILIZZATA:
1180 REM F(X)
1190 REM *****
1200 XI = (XF - XI) / 270: REM FATTORE DI SCALA PER X
1210 REM
1220 REM CALCOLO DI Y MASSIMO (YF) E MINIMO (YI)
1230 REM
1240 YI = 1E38:YF = - 1E38: REM INIZIALIZZAZIONE
1250 FOR I = 0 TO 270
1260 X = XI + I * XI
1270 FX = FN F(X)
1280 Y(I) = FX
1290 IF FX > YF THEN YF = FX
1300 IF FX < YI THEN YI = FX
1310 NEXT I
1320 YI = (YF - YI) / 150: REM FATTORE DI SCALA PER
1330 REM
1340 REM TRACCIA DEL QUADRATO GRADUATO
1350 REM
1360 HGR : HCOLOR= 3: REM INIZIALIZZAZIONE DELLA FUNZIONE GRAFICA
1370 HPL0T 0,9 TO 0,159 TO 270,159 TO 270,9 TO 0,9
1380 FOR I = 1 TO 9
1390 XP = 27 * I:YP = 159 - 15 * I
1400 HPL0T 0,YP TO 2,YP
1410 HPL0T 268,YP TO 270,YP
1420 HPL0T XP,9 TO XP,11
1430 HPL0T XP,157 TO XP,159
1440 NEXT I
1450 REM
1460 REM VISUALIZZAZIONE DEI LIMITI
1470 REM
1480 UTAB (21): REM PUNTATORE AL CAMPO TESTO
1490 PRINT "X:DA ";XI;" A ";XF;" CON PASSO";27 * XI
1500 PRINT "Y:DA ";YI;" A ";YF;" CON PASSO ";15* YI
1510 REM
1520 REM GRAFICO DELLA FUNZIONE
1530 REM
1540 HPL0T 0,159 - (Y(0) - YI) / YL
1550 FOR I = 1 TO 270
1560 YP = 159 - (Y(I) - YI) / YL
1570 HPL0T TO I,YP
1580 NEXT I
1590 REM
1600 REM VISUALIZZAZIONE DELL'ASSE
1610 REM
1620 IF XI * XF >= 0 THEN GOTO 1700
1630 XP = 270 - XF / XL
1640 HPL0T XP,9 TO XP,159
1650 REM SUDDIVISIONE DEL GRAFICO

```

```

1660 FOR I = 1 TO 9
1670 YP = 159 - 15 * I
1680 HPLOT XP,YP TO XP + 2,YP
1690 NEXT I
1700 REM
1710 REM      VISUALIZZAZIONE DELL'ASSE X
1720 REM
1730 IF YI * YF > = 0 THEN GOTO 1810
1740 YP = 159 + YI / YI
1750 HPLOT 0,YP TO 270,YP
1760 REM      SUDDIVISIONE DEL CAMPO
1770 FOR I = 1 TO 9
1780 XP = 27 * I
1790 HPLOT XP,YP TO XP,YP + 2
1800 NEXT I
1810 RETURN

```

## 4 — ESEMPIO PRATICO

Si è realizzato il grafico della funzione  $f(x) = (x - 1) (x + 1) (x - 2)$  nell'intervallo  $-3 \leq x \leq +3$

RUN

GRAFICO DI UNA FUNZIONE

AUTORE: H. HAUT

1. IL PROGRAMMA REALIZZA IL GRAFICO DI UNA FUNZIONE F(X) IN UN QUADRATO DI (270\*150) PUNTI.

LE SCALE, GLI ASSI E IL PASSO SONO CALCOLATI DAL PROGRAMMA

2. LA FUNZIONE F(X) DEVE ESSERE DEFINITA DALL'UTENTE ALLA LINEA 340 SECONDO IL MODELLO:

340 DEF F(X)=SIN(X)\*COS(X)

SE NON E' STATO ANCORA FATTO PREMERE IL TASTO RESET, DEFINIRE LA FUNZIONE E BATTERE RUN

3. INTRODURRE INFINE I VALORI INIZIALE E FINALE DELLA VARIABILE X CHE DELIMITANO L'INTERVALLO DI DEFINIZIONE

PREMERE UN TASTO PER CONTINUARE


X INIZIALE=-3.

X FINALE =+3

X:DA -3 A 3 CON PASSO.4

Y:DA -.939565076 A 14.1969636 CON PASSO 1.61365287

GRIGLIA (S O N)S



tempo d'esecuzione: 21''

memoria richiesta: 4830 bytes (senza REM : 3725).



## PROGRAMMA NUMERO 8


# RICERCA DEGLI ZERI DI UNA FUNZIONE

Il metodo utilizzato permette di trovare tutti gli "zeri" reali di una funzione  $f(x)$  definita in un intervallo  $a,b$ .

## 1 — METODO NUMERICO

Si utilizza il cosiddetto metodo della "bisezione" che permette di calcolare il punto di un intervallo  $[a,b]$  in cui la funzione ammette uno zero, purché in tale intervallo la funzione ammette un unico zero.

Sotto questa ipotesi valgono le seguenti considerazioni:



- 1) se in  $[x_a, x_b]$  vi è uno zero, allora risulta  $f(x_a) f(x_b) < 0$
- 2) sia allora  $x_c$  il punto medio di  $[x_a, x_b]$ :
  - se  $f(x_a) f(x_c) > 0$ , si considera il semintervallo  $[x_c, x_b]$
  - se  $f(x_a) f(x_c) < 0$ , si considera invece l'intervallo  $[x_a, x_c]$
- 3) si ripetono le operazioni di cui al punto 2 finché la lunghezza dell'intervallo ottenuto non risulta inferiore della precisione richiesta.

Riferimenti bibliografici: A2, C2, C3, C4, H1, L2.


## 2 — TECNICA DI PROGRAMMAZIONE


Per calcolare tutti gli zeri della funzione  $f(x)$  nell'intervallo  $[A, B]$ , bisogna innanzitutto suddividere quest'intervallo in sottointervalli di lunghezza  $DX$ , in modo tale che in ciascuno di questi cada al più uno zero.

Ognuno di questi intervallini (chiamati  $[XI, XF]$ ) viene a sua volta suddiviso in segmenti  $[XA, XB]$  utilizzando l'algoritmo esposto nel paragrafo precedente, finché  $XB - XA$  non risulta inferiore alla precisione  $ER$  voluta.


Le soluzioni vengono memorizzate in un vettore  $RA(I)$  che deve essere opportunamente dimensionato in base alla funzione prescelta. La variabile  $CI$  indica il numero di soluzioni trovate  $[RA(0), RA(1), \dots, RA(CI - 1)]$ .

3 — DIAGRAMMA  
A BLOCCHI









## 4 — PROGRAMMA

```
1  REM          CALCOLO DEGLI ZERI
2  REM          DI UNA FUNZIONE
3  REM
4  REM          AUTORE:H.H
5  REM
6  REM
7  REM          DESCRIZIONE:
8  REM          IL PROGRAMMA RICERCA CON IL METODO DELLA
9  REM          BISEZIONE, TUTTI GLI ZERI DI UNA FUNZIONE IN
10 REM          UN INTERVALLO A,B DATO.
11 REM          *****
12 REM
13 REM
100 REM
110 REM          REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 11);"CALCOLO DEGLI ZERI"
150 PRINT TAB( 11);"DI UNA FUNZIONE"
160 PRINT : PRINT TAB( 20);"AUTORE:H.HAUT"
170 PRINT
180 PRINT "1.LA FUNZIONE DI CUI SI CERCANO GLI ZE- RI DEVE ESSERE DEFIN
    ITA DALL'UTENTE ALLA LINEA 280 SECONDO IL MODELLO"
190 PRINT "      280 DEF FN F(X)= ..."
200 PRINT " SE NON E' GIA' STATO FATTO, PREMERE      RESET, DEFINIRE LA F
    UNZIONE E LANCIARE IL PROGRAMMA": PRINT
210 PRINT "2.INTRODURRE POI L'INTERVALLO A,B IN CUI SI CERCANO LE RADICI
    E IL PASSO SCELTO IN MODO CHE IN OGNI SUDDIVISIONE DI      A,B CADA
    AL PIU' UNA RADICE"
220 PRINT : PRINT "3.INTRODURRE INFINE LA PRECISIONE DESI- DERATA PER I
    L CALCOLO"
230 VTAB (23): PRINT "PREMERE UN TASTO PER CONTINUARE": GET A$
240 HOME
250 REM
260 REM          DEFINIZIONE DELLA FUNZIONE
270 REM
280 DEF FN F(X) = (X - 10) * (X - 5) * (X + 1) * (X + 3)
290 REM
300 REM          ISTRUZIONE DI DIMENSIONAMENTO
310 REM          DA MODIFICARE SE SI PREVEDONO PIU' DI 20 RADICI !!!
320 REM
330 REM
340 DIM RA(20)
350 REM
360 REM          GESTIONE DEI DATI IN INPUT
370 REM
380 PRINT "LIMITI DELL'INTERVALLO:"
390 INPUT "  A=";A
400 INPUT "  B=";B
410 PRINT : INPUT "SUDDIVISIONE DI A,B CON PASSO ";DX
420 PRINT : INPUT "PRECISIONE DESIDERATA: ";ER
430 PRINT
440 REM
450 REM          CHIAMATA DEL SOTTOPROGRAMMA
460 REM
470 GOSUB 1000
480 REM
490 REM          GESTIONE DEI RISULTATI
500 REM
510 IF CI = 0 THEN INVERSE : PRINT " NESSUNA RADICE !": NORMAL : END
520 FOR I = 0 TO CI - 1
530 PRINT " RADICE ";I + 1;" = ";RA(I)
540 NEXT I
550 END
```

```

1000 REM *****
1010 REM   SOTTOPROGRAMMA PER LA RICERCA DEGLI ZERI DI UNA FUNZIONE
1020 REM   IN UN INTERVALLO A,B (METODO DELLA BISEZIONE)
1040 REM
1050 REM   REGOLE PER L'USO DEL SOTTOPROGRAMMA
1060 REM
1070 REM       1.DATI NECESSARI:
1080 REM           * F(X)=FUNZIONE DI CUI SI RICERCANO
1090 REM             LE RADICI
1100 REM           * INTERVALLO IN CUI SI RICERCANO
1110 REM             LE RADICI, DEFINITO DA:
1120 REM             A=LIMITE INFERIORE
1130 REM             B=LIMITE SUPERIORE
1140 REM           * DX=VALORE DEL PASSO
1150 REM             (NON VI PUO' ESSERE PIU' DI UNA
1160 REM             RADICE IN OGNI SOTTOINTERVALLO
1170 REM             DI LUNGHEZZA DX)
1180 REM           * ER=PRECISIONE DESIDERATA
1190 REM       2.RISULTATI FORNITI:
1200 REM           * CI=NUMERO DI RADICI
1210 REM           * RA(0),RA(1),...,RA(CI-1) SONO LE
1220 REM             RADICI TROVATE
1230 REM       3.VARIABILI UTILIZZATE:
1240 REM           A,B,CI,DX,ER,TE,T2,XA,XB,XC,XI,XF,Y1,Y2
1250 REM       4.VETTORE UTILIZZATO:
1260 REM           RA(..) LA CUI DIMENSIONE DEVE ESSERE
1270 REM           SCELTA IN MODO CHE POSSA CONTENERE
1280 REM           TUTTE LE RADICI !!!
1290 REM *****
1300 CI = 0
1310 XF = A:XI = A
1320 IF FN F(A) = 0 THEN XC = A: GOTO 1490: REM   XC=A E' UNA RADICE
1330 IF XI > = B THEN RETURN : REM   LA RICERCA E' TERMINATA
1340 XI = XF: REM   DEFINIZIONE DELL'INTERVALLO
1350 XF = XF + DX: REM   DI LUNGHEZZA DX
1360 XA = XI:XB = XF: REM   PRIMO SEGMENTO NELL'INTERVALLO XI,XF
1370 Y1 = FN F(XI)
1380 Y2 = FN F(XF)
1390 IF Y2 = 0 THEN XC = XF: GOTO 1490: REM   XC=XF E' UNA RADICE
1400 IF Y1 * Y2 > = 0 THEN GOTO 1330: REM   CONTROLLO SULLA PRECISIONE
1410 XC = (XA + XB) / 2: REM   PUNTO MEDIO DI XA,XB
1420 TE = FN F(XC)
1430 T2 = XB - XC
1440 IF T2 < = ER THEN GOTO 1490: REM XC E' SOLUZIONE PER LA PRECISIONE
    VOLUTA
1450 REM   RIDEFINIZIONE DI UN NUOVO SEGMENTO XA,XB
1460 IF FN F(XA) * TE > 0 THEN XA = XC: GOTO 1480
1470 XB = XC
1480 GOTO 1410: REM   PROSECUZIONE DELLA RICERCA
1490 REM   MEMORIZZAZIONE DELLA SOLUZIONE
1500 RA(CI) = XC
1510 CI = CI + 1: REM   AGGIORNAMENTO DEL CONTATORE DI RADICI
1520 GOTO 1330: REM   PROSECUZIONE DELLA RICERCA

```

## 5 — ESEMPIO PRATICO

L'esempio illustra la ricerca degli zeri della funzione

$$f(x) = (x - 10) (x + 1) (x + 3)$$

nell'intervallo  $[-20, +20]$  con passo 1; la precisione richiesta è di  $1E-10$

RUN

CALCOLO DEGLI ZERI  
DI UNA FUNZIONE

AUTORE:H.HAUT

1.LA FUNZIONE DI CUI SI CERCANO GLI ZERI DEVE ESSERE DEFINITA DALL'UTENTE ALLA LINEA 280 SECONDO IL MODELLO ..  
280 DEF FN F(X)= ...  
SE NON E' GIA' STATO FATTO, PREMERE RESET, DEFINIRE LA FUNZIONE E LANCIARE IL PROGRAMMA

2.INTRODURRE POI L'INTERVALLO A,B IN CUI SI CERCANO LE RADICI E IL PASSO SCELTO IN MODO CHE IN OGNI SUDDIVISIONE DI A,B CADA AL PIU' UNA RADICE

3.INTRODURRE INFINE LA PRECISIONE DESIDERATA PER IL CALCOLO

PREMERE UN TASTO PER CONTINUARE  
LIMITI DELL'INTERVALLO:

A=-20

B=+20

SUDDIVISIONE DI A,B CON PASSO :1

PRECISIONE DESIDERATA: 1E-10

RADICE 1 = -3

RADICE 2 = -1

RADICE 3 = 5

RADICE 4 = 10

tempo d'esecuzione: 3"

memoria richiesta: 3795 (senza REM : 845).

## PROGRAMMA NUMERO 9

# RISOLUZIONE DELL'EQUAZIONE MATICIALE $AX = B$

Il programma permette di costruire la matrice inversa di una matrice  $A(n, n)$  e di calcolarne il determinante, nonché di risolvere uno o più sistemi lineari in più variabili. Inoltre il programma è concepito in modo da minimizzare l'area di memoria occupata.

## 1 — METODO NUMERICO

Siano  $A$  e  $B$  due matrici rispettivamente di dimensione  $(n, n)$  e  $(n, m)$ . Volendo risolvere l'equazione matriciale  $AX = B$ , per economizzare spazio in memoria, si dovrà fare in modo, nel corso del programma, di utilizzare lo spazio occupato inizialmente dalle matrici  $A$  e  $B$  per costruire le due matrici  $A^{-1}$  (inversa di  $A$ ) e  $X = A^{-1}B$  (soluzione). In questo modo, alla fine del programma, si troveranno memorizzate  $A^{-1}$  al posto di  $A$  e  $X$  al posto di  $B$ .

Per risolvere questo problema si è utilizzato il metodo di Gauss-Jordan o metodo del "cardine".

Questo metodo consiste nell'effettuare sulla matrice  $A$  una serie di trasformazioni  $\lambda_1, \lambda_2, \lambda_n$  in modo da ridurre la matrice  $A$  alla matrice unità  $I$ . Applicando queste trasformazioni la matrice  $A$ , dopo  $k-1$  passaggi avrà assunto la forma seguente:

$$\begin{pmatrix} 1 & 0 & \dots & 0 & | & a'_{1k} & \dots & a'_{1n} \\ 0 & 1 & \dots & 0 & | & & & \\ & & & & | & & & \\ & & & & | & & & \\ & & & & | & & & \\ 0 & 0 & & 1 & | & & & \\ \hline 0 & 0 & \dots & 0 & | & a'_{kk} & \dots & a'_{kn} \\ & & & & | & & & \\ & & & & | & & & \\ & & & & | & & & \\ & & & & | & & & \\ & & & & | & & & \\ 0 & 0 & \dots & 0 & | & a'_{nk} & & a'_{nn} \end{pmatrix} = \lambda_{k-1} \lambda_{k-2} \dots \lambda_1 A$$

La matrice inversa di A si può allora esprimere come:

$$A^{-1} = \lambda_n \lambda_{n-1} \dots \lambda_1 I \quad e$$

e la matrice X come:

$$X = \lambda_n \lambda_{n-1} \dots \lambda_1 B$$

Indicato con  $a_{ij}^{(k)}$  e  $b_{ij}^{(k)}$  gli elementi trasformati delle matrici A e B dopo k trasformazioni (k = 0 corrisponde alle matrici nella loro forma originaria) la trasformazione  $\lambda_k$  può essere definita tramite le seguenti operazioni:

$$1) \quad a_{ij}^{(k)} = a_{ij}^{(k-1)} - a_{ik}^{(k-1)} a_{kj}^{(k-1)} / a_{kk}^{(k-1)}$$

$$\text{per } \begin{cases} i = 1, \dots, k-1, k+1, \dots, n \\ j = k, \dots, n \end{cases}$$

$$2) \quad b_{ij}^{(k)} = b_{ij}^{(k-1)} - a_{ik}^{(k-1)} b_{kj}^{(k-1)} / a_{kk}^{(k-1)}$$

$$\text{per } \begin{cases} i = 1, \dots, k-1, k+1, \dots, n \\ j = 1, \dots, m \end{cases}$$

$$3) \quad a_{ij}^{(k)} = a_{ij}^{(k-1)}$$

$$\text{per } \begin{cases} i = 1, \dots, n \\ j = 1, \dots, k-1 \end{cases}$$

$$4) \quad a_{kj}^{(k)} = a_{kj}^{(k-1)} / a_{kk}^{(k-1)} \quad \text{per } j = k, \dots, n$$

$$5) \quad b_{kj}^{(k)} = b_{kj}^{(k-1)} / a_{kk}^{(k-1)} \quad \text{per } j = 1, \dots, m$$

Durante ogni passo occorre effettuare la divisione per l'elemento "pivot"  $a_{kk}$ . Per minimizzare l'errore di arrotondamento si cercherà il più grande (pivot) elemento della sottomatrice

$$\begin{pmatrix} a'_{kk} & \dots & a'_{kn} \\ & & \cdot \\ \cdot & & \cdot \\ a'_{nk} & & a'_{nn} \end{pmatrix}$$

e lo si trasporterà, effettuando le opportune permutazioni tra le righe e le colonne della matrice A nella k-esima posizione sulla diagonale principale.

Alla fine del calcolo occorrerà riordinare la matrice ottenuta tenendo conto di queste trasformazioni.

*Riferimenti bibliografici:* B1, C1, C2, C4, D1, J1, L1, L2, S1, S2, S3.

## 2 — TECNICA DI PROGRAMMAZIONE

Il calcolo si effettua utilizzando un metodo iterativo in n passi. ( $k = 1, \dots, n$ ).

Poiché si vogliono memorizzare ad ogni passo i risultati intermedi per la costruzione di  $A^{-1}$  e X nello spazio occupato dalle matrici A e B, l'algoritmo utilizzato risulta di una certa complessità. Si consiglia pertanto al lettore di effettuare una prova dettagliata del programma utilizzando un esempio molto semplice (per esempio un sistema di tre equazioni in tre incognite).

Definiamo ora le operazioni che costituiscono il k-esimo passo. Dette  $r_i$  e  $c_j$  le righe e le colonne di A e B si ha:

1) — Ricerca del "pivot": si ricerca l'elemento  $a_{ij}$  ( $i, j = k, \dots, n$ ) con valore assoluto maggiore. Tale elemento  $a_{ijk}$  verrà memorizzato in PV.

— Calcolo relativo al determinante:  $DET = DET * S * PV$ , ove S è l'indice relativo alle permutazioni e viene calcolato nel modo seguente:

$$S = S_r * S_c$$

$$\text{ove} \quad \begin{cases} S_r = \begin{cases} 1 & \text{se } i_k = k \\ -1 & \text{se } i_k \neq k \end{cases} \\ S_c = \begin{cases} 1 & \text{se } j_k = k \\ -1 & \text{se } j_k \neq k \end{cases} \end{cases}$$

– Memorizzazione di  $i_k$  e  $j_k$  in due vettori PL e PC

$$PL(K) = i_k \quad PC(K) = j_k$$

2) Posizionamento del pivot nella matrice (in posizione  $k, k$ ):

$$a) \text{ scambio di righe in A e B} \quad r_k \longleftrightarrow r_{ik}$$

$$b) \text{ scambio delle colonne in A} \quad c_k \longleftrightarrow c_{jk}$$

3) Memorizzazione della colonna  $k$ -esima di A in un vettore CS:

$$CS(I) = a_{ik} \quad (i = 1, \dots, k - 1, k + 1, \dots, n)$$

$$CS(K) = 0$$

4) Modifica della colonna  $k$ -esima di A tramite la trasformazione  $a_{ik} = \delta_{ik}$   
 ( $\delta_{ik} = 1$  se  $i = k$ ;  $\delta_{ik} = 0$  se  $i \neq k$ )

5) Trasformazione della riga  $k$ -esima di A e di B mediante la divisione per PV

6) Modifica delle restanti righe di A e B mediante la trasformazione:

$$r_j \longleftrightarrow r_j - CS(J) r_k \quad (j = 1, \dots, k - 1, k + 1, \dots, n).$$

Alla fine dell'ultimo passaggio ( $k = n$ ) bisogna riordinare le matrici A e B tenendo conto delle permutazioni effettuate tra le righe e le colonne (vedi punto 2). Per ottenere ciò si può operare nel modo seguente:

7) Scambio ordinato delle righe di A e B

$$r_n \longleftrightarrow r_{PC(N)}$$

$$r_{n-1} \longleftrightarrow r_{PC(N-1)}$$

.

.

$$r_1 \longleftrightarrow r_{PC(1)}$$

8) Scambio ordinato delle colonne di A:

$$c_n \longleftrightarrow c_{PL(N)}$$


$$c_{n-1} \longleftrightarrow c_{PL(N-1)}$$


.


$$c_1 \longleftrightarrow c_{PL(1)}$$





3 — DIAGRAMMA  
A BLOCCHI






















## 4 — PROGRAMMA

```

1  REM      RISOLUZIONE DELL'EQUAZIONE
2  REM      MATRICIALE AX=B.
3  REM
4  REM      AUTORE:H.HAUT
5  REM
6  REM
7  REM      DESCRIZIONE:
8  REM      IL PROGRAMMA PERMETTE DI RISOLVERE L'EQUAZIONE AX=B
9  REM      DOVE A E' UNA MATRICE (N,N)
10 REM      E B E' UNA MATRICE (N,M)
11 REM      IL PROGRAMMA CALCOLA
12 REM      - LA MATRICE SOLUZIONE X
13 REM      - L'INVERSA DI A
14 REM      - IL DETERMINANTE DI A
15 REM      SE M=0,SI CALCOLA SOLO LA MATRICE INVERSA E
16 REM      IL DETERMINANTE DI A
17 REM
18 REM      *****
19 REM
20 REM
100 REM
110 REM      REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 8);"RISOLUZIONE DELL'EQUAZIONE"
150 PRINT TAB( 11);"MATRICIALE AX=B"
160 PRINT : PRINT TAB( 20);"AUTORE H.HAUT"
170 PRINT
180 PRINT "1.IL PROGRAMMA RISOLVE L'EQUAZIONE AX=B      OVE"
190 PRINT "      A = MATRICE (N,N)"
200 PRINT "      B = MATRICE (N,M)      E CALCOLA:
      LA MATRICE SOLUZIONE X      L'INVE
      RSA E IL DETERMINANTE DI A"
210 PRINT " SE M=0,SI CALCOLA SOLO LA MATRICE      INVERSA ED IL DETERM
      INANTE DI A"
220 PRINT
230 PRINT "2.INSERIRE POI I VALORI DI N E M"
240 PRINT
250 PRINT "3.IN SEGUITO INTRODURRE UNA DOPO      L'ALTRA LE COLONNE D
      ELLE MATRICI A E B"
260 VTAB (23): PRINT "PREMERE UN TASTO PER CONTINUARE": GET A$
270 HOME
280 REM
290 REM      GESTIONE DEI DATI IN INPUT
300 REM
310 PRINT "DIMENSIONE DELLE MATRICI A(N,N) E B(N,M)"
320 INPUT "      N=";N
330 INPUT "      M=";M
340 N1 = N - 1;M1 = M - 1
350 REM
360 REM      ISTRUZIONI DI DIMENSIONAMENTO
370 REM
380 DIM AA(N1,N1),PC(N1),PL(N1),CS(N1)
390 IF M = 0 THEN GOTO 410
400 DIM BB(N1,M1)
410 REM
420 REM      ACQUISIZIONE NUOVI DATI
430 REM
440 REM
450 REM      INSERIMENTO DELLA MATRICE A
460 REM
470 PRINT
480 FOR I = 0 TO N1

```

```

490 PRINT "COLONNA ";I + 1;" DI A : "
500 FOR J = 0 TO N1
510 HTAB (5): INPUT AA(J,I)
520 NEXT J
530 PRINT
540 NEXT I
550 REM
560 REM   EVENTUALE INSERIMENTO DELLA MATRICE B
570 REM
580 IF M = 0 THEN GOTO 670
590 HOME
600 FOR I = 0 TO M1
610 PRINT "COLONNA ";I + 1;" DI B : "
620 FOR J = 0 TO N1
630 HTAB (5): INPUT BB(J,I)
640 NEXT J
650 PRINT
660 NEXT I
670 REM
680 REM   CHIAMATA DEL SOTTOPROGRAMMA
690 REM
700 GOSUB 2000
710 REM
720 REM   GESTIONE DEI RISULTATI
730 REM
740 CALL - 198: REM   SEGNALE SONORO
750 REM
760 REM   STAMPA DELLA MATRICE INVERSA DI A
770 REM
780 FOR I = 0 TO N1
790 HOME
800 PRINT "COLONNA ";I + 1;" DELL'INVERSA DI A : "
810 FOR J = 0 TO N1
820 HTAB (5): PRINT AA(J,I)
830 NEXT J
840 PRINT
850 PRINT "PREMERE RETURN PER CONTINUARE"
860 GET A$
870 IF ASC (A$) = 13 THEN GOTO 890
880 GOTO 860
890 NEXT I
900 REM
910 REM   STAMPA DELLA MATRICE X SE M>0
920 REM
930 FOR I = 0 TO M1
940 IF M = 0 THEN GOTO 1050
950 HOME
960 PRINT "COLONNA ";I + 1;" DELLA MATRICE SOLUZIONE X"
970 FOR J = 0 TO N1
980 HTAB (5): PRINT BB(J,I)
990 NEXT J
1000 PRINT : PRINT "PREMERE RETURN PER CONTINUARE"
1010 GET A$
1020 IF ASC (A$) = 13 THEN GOTO 1040
1030 GOTO 1010
1040 NEXT I
1050 PRINT : PRINT : PRINT
1060 PRINT "IL DETERMINANTE DI A= ";DET
1070 END
2000 REM *****
2010 REM   SOTTOPROGRAMMA D'INVERSIONE DI MATRICE E DI
2020 REM   RISOLUZIONE DELL'EQUAZIONE MATRICIALE AA X = BB
2030 REM
2040 REM
2050 REM   REGOLE PER L'USO DEL SOTTOPROGRAMMA
2060 REM
2070 REM   1.DEFINIZIONI:

```

```

2080 REM      NELL'EQUAZIONE AA X = BB,
2090 REM      AA=MATRICE N RIGHE,N COLONNE
2100 REM      BB=MATRICE N RIGHE,M COLONNE
2110 REM      2.DATI NECESSARI:
2120 REM      * N=DIMENSIONE DI AA
2130 REM      * M=NUMERO DI COLONNE DI BB
2140 REM      * AA MEMORIZZATA IN AA(I,J) (I,J=0,1,...,N-1)
2150 REM      * BB MEMORIZZATA IN BB(I,J) (I=0,1,...,N-1)
2160 REM      (J=0,1,...,M-1)
2170 REM      SE M=0 , BB NON DEVE ESSERE DEFINITA;
2180 REM      IN QUESTO CASO SI CALCOLA SOLO
2190 REM      L'INVERSA DI A E IL SUO DETERMINANTE
2200 REM      3.RISULTATI FORNITI:
2210 REM      * L'INVERSA DI AA SI TROVA IN AA(I,J)
2220 REM      (I,J=0,1,...,N-1)
2230 REM      * IL DETERMINANTE DI AA=DET
2240 REM      * SE M E' DIVERSO DA ZERO,LA MATRICE
2250 REM      SOLUZIONE X SI TROVA IN BB(I,J)
2260 REM      (I=0,1,...,N-1)
2270 REM      (J=0,1,...,M-1)
2280 REM      4.VARIABILI UTILIZZATE:
2290 REM      DET,I,IK,J,JK,K,N,N1,M,M1,PAV,PU,TT
2300 REM      5.VETTORI E MATRICI UTILIZZATI:
2310 REM      CS(N-1),PC(N-1),PL(N-1)
2320 REM      AA(N-1,N-1),BB(N-1,M-1)
2330 REM      SE M=0,BB NON VIENE UTILIZZATA
2340 REM      *****
2350 REM      INIZIALIZZAZIONE PER LA RICERCA
2360 N1 = N - 1:M1 = M - 1
2370 DET = 1: REM      INIZIALIZZAZIONE DEL DETERMINANTE
2380 REM      K=INDICE DEL PASSAGGIO EFFETTUATO
2390 FOR K = 0 TO N1
2400 REM
2410 REM      RICERCA DEL PIVOT
2420 REM
2430 PU = AA(K,K)
2440 IK = K:JK = K
2450 PAV = ABS (PU)
2470 FOR I = K TO N1
2480 FOR J = K TO M1
2490 IF ABS (AA(I,J)) > PAV THEN PU = AA(I,J):PAV = ABS (PU):IK = I:JK = J
2500 NEXT J
2510 NEXT I
2520 REM      RICERCA ULTIMATA, IL PIVOT E' PU NELLA POSIZIONE I=IK,J=JK
2530 PC(K) = JK:PL(K) = IK: REM      MEMORIZZAZIONE DELLA POSIZIONE DEL PIVOT

2540 REM
2550 REM      CALCOLO DEL DETERMINANTE
2560 REM
2570 IF IK > K THEN DET = - DET
2580 IF JK > K THEN DET = - DET
2590 DET = DET * PU
2600 REM
2610 REM      SE IL DETERMINANTE E' UGUALE A ZERO LA MATRICE NON E' INVERTIBILE
2620 REM      INVIO DI UN MESSAGGIO DI ERRORE
2630 REM
2640 IF DET = 0 THEN HOME : GOTO - 198: PRINT "IL DETERMINANTE E' NULLO
:NESSUNA SOLUZIONE": END
2650 REM
2660 REM      POSIZIONAMENTO DEL PIVOT IN K,K
2670 REM
2680 IF IK = K THEN GOTO 2870: REM      IL PIVOT E' GIA' SULLA GIUSTA RIGA

2690 REM
2700 REM      SCAMBIO DELLE RIGHE IK E K DI AA

```

```

2710 REM
2720 FOR I = 0 TO N1
2730 TT = AA(IK,I)
2740 AA(IK,I) = AA(K,I)
2750 AA(K,I) = TT
2760 NEXT I
2770 REM
2780 REM SCAMBIO DELLE RIGHE IK E K DI BB
2790 REM
2800 IF M1 = - 1 THEN GOTO 2870
2810 FOR I = 0 TO M1
2820 TT = BB(IK,I)
2830 BB(IK,I) = BB(K,I)
2840 BB(K,I) = TT
2850 NEXT I
2860 REM
2870 IF JK = K THEN GOTO 3030: REM IL PIVOT E' GIA' SULLA GIUSTA COLON-
NA
2880 REM
2890 REM SCAMBIO DELLE COLONNE JK E K DI AA
2900 REM
2910 FOR I = 0 TO N1
2920 TT = AA(I,JK)
2930 AA(I,JK) = AA(I,K)
2940 AA(I,K) = TT
2950 NEXT I
2960 REM
2970 REM IL PIVOT E' POSIZIONATO IN K,K
2980 REM
2990 REM
3000 REM MEMORIZZAZIONE DELLA COLONNA K DI AA NEL VETTORE CS
3010 REM E AZZERAMENTO DELLA STESSA
3020 REM
3030 FOR I = 0 TO N1
3040 CS(I) = AA(I,K)
3050 AA(I,K) = 0
3060 NEXT I
3070 REM
3080 CS(K) = 0:AA(K,K) = 1
3090 REM
3100 REM
3110 REM MODIFICA DELLA RIGA K DI AA
3120 REM
3130 FOR I = 0 TO N1
3140 AA(K,I) = AA(K,I) / PV
3150 NEXT I
3160 REM
3170 REM MODIFICA DELLA RIGA K DI BB TRANNE CHE SE M=0
3180 REM
3190 IF M1 = - 1 THEN GOTO 3260
3200 FOR I = 0 TO M1
3210 BB(K,I) = BB(K,I) / PV
3220 NEXT I
3230 REM
3240 REM MODIFICA DELLE ALTRE RIGHE DI AA
3250 REM
3260 FOR J = 0 TO N1
3270 IF J = K THEN GOTO 3440: REM NON SI MODIFICA LA RIGA K
3280 REM
3290 REM MODIFICA DELLA RIGA J DI AA
3300 REM
3310 FOR I = 0 TO N1
3320 AA(J,I) = AA(J,I) - CS(J) * AA(K,I)
3330 NEXT I
3340 REM
3350 REM MODIFICA DELLA RIGA J DI BB
3360 REM

```

```

3370 IF M1 = - 1 THEN GOTO 3440
3380 FOR I = 0 TO M1
3390 BB(J,I) = BB(J,I) - CS(J) * BB(K,I)
3400 NEXT I
3410 REM
3420 REM LA RIGA J E' MODIFICATA
3430 REM
3440 NEXT J
3450 REM
3460 REM FINE DEL PASSO K
3470 REM
3480 NEXT K
3490 REM
3500 REM LA MATRICE E' INVERTITA
3510 REM
3520 REM RIORDINAMENTO DELLA MATRICE
3530 REM
3540 REM
3550 REM SCAMBIO DELLE RIGHE
3560 REM
3570 FOR I = N1 TO 0 STEP - 1
3580 TK = PC(I)
3590 IF JK = I THEN GOTO 3800
3600 REM
3610 REM SCAMBIO DELLE RIGHE I E PC(I) DI AA
3620 REM
3630 FOR J = 0 TO N1
3640 TT = AA(I,J)
3650 AA(I,J) = AA(TK,J)
3660 AA(TK,J) = TT
3670 NEXT J
3680 REM
3690 REM SCAMBIO DELLE RIGHE I E PC(I) DI BB
3700 REM
3710 IF M1 = - 1 THEN GOTO 3800
3720 FOR J = 0 TO M1
3730 TT = BB(I,J)
3740 BB(I,J) = BB(TK,J)
3750 BB(TK,J) = TT
3760 NEXT J
3770 REM
3780 REM CAMBIO DELLA RIGA SUCCESSIVA
3790 REM
3800 NEXT I
3810 REM
3820 REM SCAMBIO TRA LE COLONNE DI AA
3830 REM
3840 FOR J = N1 TO 0 STEP - 1
3850 JK = PL(J)
3860 IF JK = J THEN GOTO 3960: REM SCAMBIO NON NECESSARIO
3870 REM
3880 REM SCAMBIO DELLE COLONNE J E PL(J) DI AA
3890 REM
3900 FOR I = 0 TO N1
3910 TT = AA(I,J)
3920 AA(I,J) = AA(I,JK)
3930 AA(I,JK) = TT
3940 NEXT I
3950 REM CAMBIO DELLA COLONNA SEGUENTE
3960 NEXT J
3970 REM
3980 REM RIORDINAMENTO CONCLUSO
3990 REM
4000 RETURN

```

## 5 — ESEMPIO PRATICO

Si voglia risolvere l'equazione matriciale  $AX = B$  ove le matrici A (di dimensioni 6,6; N = 6) e B (1 colonna; M = 1) sono date da:

$$A = \begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 1 \\ 1 & 1 & 0 & 0 & 0 & -1 \\ -1 & 1 & 1 & 0 & 0 & 1 \\ 1 & -1 & 1 & 1 & 0 & -1 \\ -1 & 1 & -1 & 1 & 1 & 1 \\ 1 & -1 & 1 & -1 & 1 & -1 \end{pmatrix} \text{ e } B = \begin{pmatrix} 1 \\ 0 \\ 1 \\ 0 \\ 1 \\ 0 \end{pmatrix}$$

Il programma fornisce la matrice inversa  $A^{-1}$ , la matrice soluzione X ed il determinante di A

```

RUN
      RISOLUZIONE DELL'EQUAZIONE
      MATRICIALE AX=B

      AUTORE H.HAUT

1.IL PROGRAMMA RISOLVE L'EQUAZIONE AX=B
   OVE
      A = MATRICE (N,N)
      B = MATRICE (N,M)
   E CALCOLA:
      LA MATRICE SOLUZIONE X
      L'INVERSA E IL DETERMINANTE DI A
   SE M=0,SI CALCOLA SOLO LA MATRICE
   INVERSA ED IL DETERMINANTE DI A

2.INSERIRE POI I VALORI DI N E M

3.IN SEGUITO INTRODURRE UNA DOPO
   L'ALTRA LE COLONNE DELLE MATRICI A E B

PREMERE UN TASTO PER CONTINUARE
DIMENSIONE DELLE MATRICI A(N,N) E B(N,M)

N=6
M=1

COLONNA 1 DI A :
?1
?1
?-1
?1
?-1
?1

COLONNA 2 DI A :
?0
?1
?1
?-1
?1
?-1

```

COLONNA 3 DI A :

?0  
?0  
?1  
?1  
?-1  
?1

COLONNA 4 DI A :

?0  
?0  
?0  
?1  
?1  
?-1

COLONNA 5 DI A :

?0  
?0  
?0  
?0  
?1  
?1

COLONNA 6 DI A :

?1  
?-1  
?1  
?-1  
?1  
?-1

COLONNA 1 DI B :

?1  
?0  
?1  
?0  
?1  
?0

COLONNA 1 DELL'INVERSA DI A:

.5  
0  
0  
0  
0  
.5

PREMERE RETURN PER CONTINUARE

COLONNA 2 DELL'INVERSA DI A:

.25  
.5  
0  
0  
0  
-.25

PREMERE RETURN PER CONTINUARE

COLONNA 3 DELL'INVERSA DI A:

-.125  
.25  
.5  
0  
0  
.125

PRESS RETURN PER CONTINUARE  
COLONNA 4 DELL'INVERSA DI A:

.0625  
-.125  
.25  
.5  
0  
-.0625

PRESS RETURN PER CONTINUARE  
COLONNA 5 DELL'INVERSA DI A:

-.03125  
.0625  
-.125  
.25  
.5  
.03125

PRESS RETURN PER CONTINUARE  
COLONNA 6 DELL'INVERSA DI A:

.03125  
-.0625  
.125  
-.25  
.5  
-.03125

PRESS RETURN PER CONTINUARE  
COLONNA 1 DELLA MATRICE SOLUZIONE X

.34375  
.3125  
.375  
.25  
.5  
.65625

PRESS RETURN PER CONTINUARE

IL DETERMINANTE DI A= -32

tempo d'esecuzione: 9"

memoria richiesta: 7249 bytes (senza REM : 2484).



## PROGRAMMA NUMERO 10

# AUTOVALORI ED AUTOVETTORI DI UNA MATRICE REALE SIMMETRICA

Questo programma realizza la “diagonalizzazione” di una matrice reale simmetrica con il metodo iterativo di Jacobi.

Il calcolo viene arrestato o quando si è raggiunta la precisione richiesta o quando si è raggiunto il numero massimo di iterazioni (tale numero viene definito dall'utente all'inizio del programma).

## 1 — METODO NUMERICO

Sia  $A = (a_{ij})$  la matrice reale simmetrica di ordine  $n$  da diagonalizzare. È noto che una tale matrice ammette degli autovalori e degli autovettori reali e che esiste una matrice ortogonale  $U$  che la diagonalizza, vale a dire che (posto  $U^T$  uguale alla matrice trasposta di  $U$ ) soddisfa alla seguente condizione:

$$U^T A U = \begin{pmatrix} \mu_1 & & & \\ & \mu_2 & 0 & \\ & & \ddots & \\ & 0 & & \mu_n \end{pmatrix}$$

I valori  $\mu_i$  rappresentano gli autovalori di  $A$  mentre gli autovettori corrispondenti sono dati dalle colonne della matrice  $U$ .

Dato che una trasformazione ortogonale conserva la norma della matrice (la norma è la somma dei quadrati di tutti gli elementi) si dovrà avere, dopo la diagonalizzazione:

$$\sum_i \mu_i^2 = \sum_{i,j} a_{ij}^2$$

Il principio su cui si basa il metodo Jacobi consiste nel considerare  $U$  come prodotto di rotazioni elementari.

Esprimendo la trasformazione  $U$  nella forma:

$$U = U_1 U_2 U_3 \dots$$

si effettuano le trasformazioni  $U_i$

$$A' = \dots U_3^T U_2^T U_1^T A U_1 U_2 U_3 \dots$$

finché la somma dei quadrati degli elementi diagonali di  $A'$  non approssima, con la precisione voluta, la norma di  $A$ .

Ricordiamo che una rotazione elementare  $U_r$  si rappresenta nel modo seguente:

$$U^{(ij)}_r = \begin{pmatrix} 1 & & & & & \\ & \ddots & & & & \\ & & 1 & & & \\ & & & c & \dots & s \\ 0 & & & s & \dots & c \\ & & -s & \dots & c & \\ 0 & & & & 0 & \ddots & 1 \end{pmatrix} \begin{matrix} \rightarrow i \\ \rightarrow j \end{matrix}$$

con  $c = \cos(\theta)$  e  $s = \sin(\theta)$

La trasformazione

$$A' = U^{(ij)T} A U^{(ij)}$$

opera sugli elementi di  $A$  le seguenti variazioni:

- (1)  $a'_{ik} = c a_{ik} + s a_{jk} \quad (k \neq i, j)$
- (2)  $a'_{jk} = c a_{jk} - s a_{ik} \quad (k \neq i, j)$
- $a'_{ki} = a'_{ik} \quad (k \neq i, j)$
- $a'_{kj} = a'_{jk} \quad (k \neq i, j)$

- $$\begin{aligned}
 (3) \quad a'_{ii} &= c^2 a_{ii} + 2sc a_{ij} + s^2 a_{jj} \\
 (4) \quad a'_{jj} &= s^2 a_{ii} - 2sc a_{ij} + c^2 a_{jj} \\
 (5) \quad a'_{ij} &= a'_{ji} = (c^2 - s^2) a_{ij} + cs (a_{jj} - a_{ii})
 \end{aligned}$$

Gli altri elementi rimangono inalterati.

La somma dei quadrati degli elementi posti sulla diagonale principale viene modificata eseguendo la sostituzione:

$$a_{ii}'^2 + a_{jj}'^2 = a_{ii}^2 + a_{jj}^2 + 2a_{ij}^2$$

Si sceglie allora l'angolo  $\theta$  in modo da rendere nullo  $a'_{ij}$  e si procede così nell'iterazione fino a giungere alla diagonalizzazione di A.

I valori di  $\cos(\theta)$  e  $\sin(\theta)$  possono essere calcolati secondo la seguente legge:

$$(6) \quad c = \cos(\theta) = \sqrt{\frac{1}{2} \left( 1 + \frac{q}{p^2 + q^2} \right)}$$

$$(7) \quad s = \sin(\theta) = \frac{p}{2\sqrt{p^2 + q^2} \cos(\theta)}$$

$$\begin{aligned}
 \text{ove: } q &= |b| \\
 p &= 2ba_{ij}/|b| \\
 b &= a_{ii} - a_{jj}
 \end{aligned}$$

Se  $b = 0$  si ha:

$$c = 1/\sqrt{2} \text{ ed } s = \text{sgn}(a_{ij})/\sqrt{2}$$

*Riferimenti bibliografici:* A2, B1, C4, D1, J1, L1, L2, S3.

## 2 — TECNICA DI PROGRAMMAZIONE

L'algoritmo può essere scomposto in quattro passi successivi:

- 1) Ad ogni iterazione si ricerca, tra gli elementi non disposti sulla diagonale principale, quello avente valore assoluto maggiore.
- 2) Si calcolano i parametri della rotazione  $U^{(ij)}$  che permette di azzerare tale elemento [formule (6) e (7)].

- 3) Si effettua su A la trasformazione  $U^{(ij)} T_{AU}^{(ij)}$  (cfr. formule da (1) a (5) ) e si realizza la U con  $UU^{(ij)}$ .
- 4) Si termina l'iterazione quando si è ottenuta la precisione voluta.

Il programma è concepito in modo da permettere all'utente di definire un estremo superiore per il numero di iterazioni. Se dopo un tale numero di iterazioni non si è ancora ottenuto il risultato con la precisione voluta, si può richiedere di effettuare un numero di iterazioni supplementari.

Le principali variabili utilizzate nel programma sono:

- $A_{(i,j)}$  = matrice reale simmetrica di ordine N
- ER = precisione desiderata
- NA = norma di A
- ND = somma dei quadrati degli elementi della diagonale principale; l'iterazione ha termine quando  $|1 - NA/ND| < ER$
- IM = numero massimo di iterazioni volute
- $U_{(i,j)}$  = matrice di diagonalizzazione, inizializzata con la matrice unità.


Alla fine del programma gli autovalori sono dati da:


$$\mu_i = A(i,i) \quad (i = 0, 1, \dots, N - 1)$$


e gli autovettori corrispondenti da:


$$\vec{V}_i = [U(0,i), U(1,i), \dots, U(N - 1,i)] \quad (i = 0, 1, \dots, N - 1)$$

3 — DIAGRAMMA A BLOCCHI














## 4 — PROGRAMMA

```

1  REM      CALCOLO DEGLI AUTOVALORI E AUTOVETTORI
2  REM      DI UNA MATRICE REALE SIMMETRICA
4  REM
5  REM      AUTORE:H.HAUT
6  REM
7  REM
8  REM      DESCRIZIONE:
9  REM      QUESTO PROGRAMMA UTILIZZA IL METODO ITERATIVO
10 REM      DI JACOBI PER CALCOLARE GLI AUTOVALORI E GLI
11 REM      AUTOVETTORI DI UNA MATRICE REALE, SIMMETRICA
12 REM      CON LA PRECISIONE VOLUTA E ASSEGNANDO UN
13 REM      NUMERO MASSIMO DI ITERAZIONI
14 REM
15 REM      *****
16 REM
17 REM
100 REM
110 REM      REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 5);"CALCOLO DEGLI AUTOVALORI E DEGLI"
150 PRINT TAB( 5);"AUTOVETTORI DI UNA MATRICE "
160 PRINT TAB( 11);"REALE SIMMETRICA. "
170 PRINT TAB( 6);"(METODO ITERATIVO DI JACOBI)": PRINT
180 PRINT : PRINT TAB( 24);"AUTORE:H.HAUT"
190 PRINT
200 PRINT "INSERIRE I SEGUENTI DATI:"
210 PRINT " 1.L'ORDINE DELLA MATRICE": PRINT
220 PRINT " 2.LE COLONNE UNA DOPO L'ALTRA          (LA MATRICE E' SIMM
    ETRICA E QUINDI SONO NECESSARI SOLO I PRIMI TERMINI DI OGNI
    COLONNA)"
230 PRINT
240 PRINT " 3.LA PRECISIONE VOLUTA"
250 PRINT
260 PRINT " 4.IL NUMERO MASSIMO DI ITERAZIONI"
270 VTAB (22): PRINT "PREMERE UN TASTO PER PROSEGUIRE ": GET A$
280 HOME
290 REM
300 REM      GESTIONE DEI DATI IN INPUT
310 REM
320 PRINT : INPUT "ORDINE DELLA MATRICE=";N
330 N = N - 1
340 REM
350 REM      ISTRUZIONI DI DIMENSIONAMENTO
360 REM
370 DIM A(N,N),U(N,N)
380 REM
390 REM      PROSECUZIONE INSERIMENTO DATI
400 REM
410 FOR I = 0 TO N
420 PRINT : PRINT
430 PRINT "I ";I + 1;" PRIMI TERMINI DELLA COLONNA ";I + 1: PRINT "SONO:"

440 FOR J = 0 TO I
450 HTAB (6): INPUT A(J,I)
460 NEXT J
470 NEXT I
480 PRINT : PRINT : PRINT
490 INPUT "LA PRECISIONE VOLUTA=";FR
500 PRINT : PRINT
510 INPUT "IL NUMERO MASSIMO D'ITERAZIONI=";IM
520 REM
530 REM      CHIAMATA DEL SOTTOPROGRAMMA
540 REM

```

```

550 GOSUB 1000
560 CALL - 198
570 REM
580 REM GESTIONE DEI RISULTATI
590 REM
600 HOME
610 FOR I = 0 TO N
620 PRINT : PRINT
630 PRINT "AUTOVALORE ";I + 1;" = ";A(I,I): PRINT
640 PRINT "AUTOVETTORE CORRISPONDENTE:"
650 FOR J = 0 TO N
660 HTAB (20): PRINT U(J,I)
670 NEXT J
680 PRINT : PRINT "PREMERE RETURN PER CONTINUARE"
690 GET A$
700 IF ASC (A$) 13 THEN GOTO 690
710 NEXT I
720 END

1000 REM *****
1010 REM SOTTOPROGRAMMA PER IL CALCOLO DEGLI AUTOVALORI
1020 REM ED AUTOVETTORI DI UNA MATRICE SIMMETRICA REALE
1030 REM CON IL METODO ITERATIVO DI JACOBI
1040 REM
1050 REM
1060 REM REGOLE PER L'USO DEL SOTTOPROGRAMMA
1070 REM
1080 REM 1.DATI NECESSARI:
1090 REM * LA MATRICE DA DIAGONALIZZARE MEMORIZZATA IN
1100 REM A(I,J) (I,J=0,1,...,N)
1110 REM * N CHE DEFINISCE L'ORDINE DELLA MATRICE
1120 REM ( N= ORDINE - 1 !!)
1130 REM * ER=PRECISIONE VOLUTA
1140 REM * IM=NUMERO MASSIMO DI ITERAZIONI
1150 REM 2.RISULTATI FORNITI:
1160 REM * A(I,I) DA' L'I-ESIMO AUTOVALORE
1170 REM (I=0,1,...,N)
1180 REM * L'I-ESIMO AUTOVETTORE E' LA I-ESIMA
1190 REM COLONNA DELLA MATRICE U:
1200 REM (U(0,I),U(1,I),...,U(N,I))
1210 REM I=0,1,...,N
1220 REM 3.VARIABILI UTILIZZATE:
1230 REM A$,B,C,ER,I,IM,J,K,N,NA,ND,NT,NJ
1240 REM P,PR,Q,S,ST,TE
1250 REM 4.MATRICI UTILIZZATE:
1260 REM A(N,N),U(N,N)
1270 REM (NOTA: N=ORDINE-1)
1280 REM *****
1290 REM INIZIALIZZAZIONE
1300 IT = 0
1310 NA = 0
1320 ND = 0
1330 ST = 0
1340 REM
1350 REM CALCOLO DI ND,NA E INIZIALIZZAZIONE DELLA MATRICE U
1360 REM
1370 FOR I = 0 TO N
1380 U(I,I) = 1
1390 ND = ND + A(I,I) ^ 2
1400 IF I = N THEN GOTO 1470
1410 FOR J = I + 1 TO N
1420 ST = ST + A(I,J) ^ 2: REM SOMMA TRANSITORIA DEI QUADRATI DEGLI ELEM-
1430 U(I,J) = 0:U(J,I) = 0
1440 NEXT J
1450 NEXT I
1460 REM
1470 NA = ND + 2 * ST

```

```

1480 REM
1490 REM RICERCA DELL'ELEMENTO (NON SULLA DIAGONALE) MAGGIORE IN VALORE
    ASSOLUTO
1500 REM
1510 ST = 0
1520 FOR I = 0 TO N - 1
1530 FOR J = I + 1 TO N
1540 TE = ABS (A(I,J))
1550 IF TE > ST THEN ST = TE:NI = I:NJ = J
1560 NEXT J
1570 NEXT I
1580 REM
1590 REM L'ELEMENTO CERCATO E' IN NI,NJ
1600 REM
1610 REM CALCOLO DEI PARAMETRI DELLA ROTAZIONE
1620 REM
1630 R = A(NI,NI) - A(NJ,NJ)
1640 IF R = 0 THEN C = 1 / SQR (2):S = C * SGN (A(NI,NJ)):GOTO 1710: REM
    CASO PARTICOLARE CON R=0
1650 Q = ABS (R):P = 2 * A(NI,NJ) * SGN (R)
1660 ST = SQR (P * P + Q * Q)
1670 C = SQR ((1 + Q / ST) / 2):S = P / (2 * ST * C)
1680 REM
1690 REM C=cos(ANGOLO DI ROTAZIONE);S=sin(ANGOLO)
1700 REM
1710 REM CALCOLO DELLA MATRICE U
1720 REM
1730 FOR K = 0 TO N
1740 ST = U(K,NI)
1750 U(K,NI) = C * ST + S * U(K,NJ)
1760 U(K,NJ) = C * U(K,NJ) - S * ST
1770 NEXT K
1780 REM
1790 REM TRASFORMAZIONE DELLA MATRICE A
1800 REM
1810 REM
1820 REM TRASFORMAZIONI (1) E (2) DEL TESTO
1830 REM IN TRE TAPPE:
1840 REM 1.K=0,...,NI-1 SALVO SE NI=0
1850 REM 2.K=NI+1,...,NJ-1 SALVO SE NJ=NI+1
1860 REM 3.K=NJ+1,...,N SALVO SE NJ=N
1870 REM
1880 REM PRIMA TAPPA
1890 REM
1900 IF NI = 0 THEN GOTO 1960
1910 FOR K = 0 TO NI - 1
1920 ST = A(K,NI)
1930 A(K,NI) = C * ST + S * A(K,NJ)
1940 A(K,NJ) = C * A(K,NJ) - S * ST
1950 NEXT K
1960 REM
1970 REM SECONDA TAPPA
1980 REM
1990 IF NJ = (NI + 1) THEN GOTO 2050
2000 FOR K = NI + 1 TO NJ - 1
2010 ST = A(NI,K)
2020 A(NI,K) = C * ST + S * A(K,NJ)
2030 A(K,NJ) = C * A(K,NJ) - S * ST
2040 NEXT K
2050 REM
2060 REM TERZA TAPPA
2070 REM
2080 IF NJ = N THEN GOTO 2170
2090 FOR K = NJ + 1 TO N
2100 ST = A(NI,K)
2110 A(NI,K) = C * ST + S * A(NJ,K)
2120 A(NJ,K) = C * A(NJ,K) - S * ST

```

```

2130 NEXT K
2140 REM
2150 REM   CALCOLO DI ND
2160 REM
2170 ND = ND + 2 * A(NI,NJ) ^ 2
2180 REM
2190 REM   TRASFORMAZIONE (3) DEL TESTO
2200 REM
2210 ST = A(NI,NI)
2220 A(NI,NI) = C * C * ST + 2 * S * C * A(NI,NJ) + S * S * A(NJ,NJ)
2230 REM
2240 REM   TRASFORMAZIONE (4) DEL TESTO
2250 REM
2260 A(NJ,NJ) = C * C * A(NJ,NJ) + S * S * ST - 2 * S * C * A(NI,NJ)
2270 REM
2280 REM   TRASFORMAZIONE (5) DEL TESTO
2290 REM
2300 A(NI,NJ) = 0
2310 REM
2320 IT = IT + 1: REM   AGGIORNAMENTO DEL NUMERO DI ITERAZIONI
2330 PR = ABS (1 - NA / ND): REM   CALCOLO DELLA PRECISIONE
2340 IF PR < ER THEN RETURN : REM   IL CALCOLO E' TERMINATO CON LA PRECI-
    SIONE VOLUTA
2350 IF IT < IM THEN GOTO 1510: REM PROSECUZIONE DELLA DIAGONALIZZAZIONE

2360 REM
2370 REM   IL NUMERO DI ITERAZIONI E' ULTIMATO: OPZIONE PER EFFETTUARNE
    ALTRE SUPPLEMENTARI
2380 REM
2390 CALL - 198
2400 HOME : PRINT "IL NUMERO DI ITERAZIONI PERMESSE E'      RAGGIUNTO"
2410 PRINT "LA PRECISIONE OTTENUTA E' ";PR
2420 INPUT "VOLETE PROSEGUIRE?";A$
2430 IF A$ = "N" THEN RETURN
2440 INPUT "NUMERO DI ITERAZIONI SUPPLEMENTARI=";IM
2450 IT = 0: REM   RIINIZIALIZZAZIONE DEL CONTATORE DI ITERAZIONI
2460 GOTO 1510: REM   PROSECUZIONE DELLA DIAGONALIZZAZIONE

```

## 5 — ESEMPIO PRATICO

L'esempio illustra la ricerca degli autovalori ed autovettori della matrice:

$$A = \begin{pmatrix} 4 & 2 & 2 \\ 2 & 5 & 1 \\ 2 & 1 & 6 \end{pmatrix}$$

con una precisione di  $1E - 10$  ed un massimo di 10 iterazioni.

```

RUN
CALCOLO DEGLI AUTOVALORI E DEGLI
AUTOVETTORI DI UNA MATRICE
    REALE SIMMETRICA.
(METODO ITERATIVO DI JACOBI)

```

AUTORE:H.HAUT

```

INSERIRE I SEGUENTI DATI:
1.L'ORDINE DELLA MATRICE

2.LE COLONNE UNA DOPO L'ALTRA
  (LA MATRICE E' SIMMETRICA E QUINDI

```

SONO NECESSARI SOLO I PRIMI TERMINI  
DI OGNI COLONNA)

3.LA PRECISIONE VOLUTA

4.IL NUMERO MASSIMO DI ITERAZIONI  
PREMERE UN TASTO PER PROSEGUIRE

ORDINE DELLA MATRICE=3

I 1 PRIMI TERMINI DELLA COLONNA 1  
SONO:

74

I 2 PRIMI TERMINI DELLA COLONNA 2  
SONO:

72

75

I 3 PRIMI TERMINI DELLA COLONNA 3  
SONO:

72

71

76

LA PRECISIONE VOLUTA=1E-10

IL NUMERO MASSIMO D'ITERAZIONI=10

AUTOVALORE 1 = 2.12592447

AUTOVETTORE CORRISPONDENTE:

.828033347

-.46965459

-.306243926

PREMERE RETURN PER CONTINUARE

AUTOVALORE 2 = 8.38761906

AUTOVETTORE CORRISPONDENTE:

.538678589

.514885452

.666872064

PREMERE RETURN PER CONTINUARE

AUTOVALORE 3 = 4.48645648

AUTOVETTORE CORRISPONDENTE:

-.155518984

-.717159353

.679335196

PREMERE RETURN PER CONTINUARE

tempo d'esecuzione: 6"8

memoria richiesta: 5636 bytes (senza REM : 2049)

## PROGRAMMA NUMERO 11

# CALCOLO DELLE FUNZIONI STATISTICHE PER UNA O DUE VARIABILI

Il programma permette di calcolare il valore delle varie funzioni statistiche sia per una variabile  $x$ , che assume i valori  $[x_1, x_2, \dots, x_n]$ , sia per una coppia di variabili  $x, y$  che assumono i valori  $[(x_1, y_1), \dots, (x_n, y_n)]$ .

## 1 — DESCRIZIONE DEL PROGRAMMA E DEL METODO

Le funzioni calcolate dal programma sono definite dalle formule seguenti:

1. media di  $x$  :  $\bar{x} = (\sum x_i)/n$
2. media di  $y$  :  $\bar{y} = (\sum y_i)/n$
3. scarto quadratico medio di  $x$  (divisione per  $n - 1$ ):

$$S_x = \sqrt{\frac{\sum x_i^2 - n\bar{x}^2}{n - 1}}$$

4. scarto quadratico medio di  $y$  (divisione per  $n - 1$ ):

$$S_y = \sqrt{\frac{\sum y_i^2 - n\bar{y}^2}{n - 1}}$$

5. scarto quadratico medio di  $x$  (divisione per  $n$ ):

$$S'_x = \sqrt{\frac{n-1}{n}} S_x$$

6. scarto quadratico medio di  $y$  (divisione per  $n$ ):

$$S'_y = \sqrt{\frac{n-1}{n}} S_y$$

7. scarto quadratico medio di  $\bar{x}$  (divisione per  $n - 1$ ):

$$S_{\bar{x}} = S_x / \sqrt{n}$$

8. scarto quadratico medio di  $\bar{y}$  (divisione per  $n - 1$ ):

$$S_{\bar{y}} = S_y / \sqrt{n}$$

9. scarto quadratico medio di  $\bar{x}$  (divisione per  $n$ ):

$$S'_{\bar{x}} = S'_x / \sqrt{n}$$

10. scarto quadratico medio di  $\bar{y}$  (divisione per  $n$ ):

$$S'_{\bar{y}} = s'_y / \sqrt{n}$$

11. covarianza di  $x, y$  (divisione per  $n - 1$ ):

$$S_{xy} = \frac{1}{n-1} (\sum x_i y_i - 1/n \sum x_i \sum y_i)$$

12. covarianza di  $x, y$  (divisione per  $n$ ):

$$S_{xy} = 1/n (\sum x_i y_i - 1/n \sum x_i \sum y_i)$$

13. coefficiente di correlazione:


$$P = S_{xy} / S_x S_y$$


Man mano che i valori  $x_i$  ed  $y_i$  vengono trasmessi al calcolatore, vengono memorizzati nei due vettori  $X(I)$  e  $Y(I)$  di dimensione  $n$ . L'algoritmo segue rigorosamente le definizioni date e non presenta alcuna difficoltà.


*Riferimenti bibliografici:* J2, M1.



## 2 — DIAGRAMMA A BLOCCHI







### 3 — PROGRAMMA

```

1 REM      CALCOLO DELLE FUNZIONI STATISTICHE
2 REM      PER UNA O DUE VARIABILI.
3 REM
4 REM      AUTORE:H.HAUT
5 REM
6 REM
7 REM      DESCRIZIONE:
8 REM      IL PROGRAMMA PERMETTE DI CALCOLARE LE FUNZIONI
9 REM      STATISTICHE PARTENDO DA N VALORI DI UNA
10 REM     VARIABILE X O DI UNA COPPIA DI VARIABILI X,Y
11 REM
12 REM
13 REM     *****
14 REM
15 REM
100 REM
110 REM     REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 3);"CALCOLO DELLE FUNZIONI STATISTICHE"
150 PRINT TAB( 8);"PER UNA O DUE VARIABILI"
160 PRINT : PRINT TAB( 20);"AUTORE:H.HAUT": PRINT : PRINT
170 PRINT "1.DEFINIRE IL TIPO DI CALCOLO:                1=FUNZIONI STATISTI
    CHE PER UN INSIE-      MEDI N VALORI X(I)                2=FUNZIO
    NI STATISTICHE PER UN INSIE-      ME DI COPPIE (X(I),Y(I))"
180 PRINT
190 PRINT "2.INTRODURRE POI IL NUMERO DI VALORI      O DI COPPIE"
200 PRINT
210 PRINT "3.INTRODURRE SUCCESSIVAMENTE I DATI      SPERIMENTALI"
220 VTAB (21): PRINT "PREMERE UN TASTO PER PROSEGUIRE  ": GET A$
230 HOME
240 REM
250 REM     GESTIONE DEI DATI IN INPUT
260 REM
270 INPUT "SCELTA DEL TIPO DI CALCOLO (1 O 2) ":NV
280 PRINT : INPUT "NUMERO DI VALORI O DI COPPIE: ":N
290 REM
300 REM     ISTRUZIONI DI DIMENSIONAMENTO
310 REM
320 IF NV = 2 THEN DIM Y(N)
330 DIM X(N)
340 REM
350 REM     INSERIMENTO DATI SPERIMENTALI
360 REM
370 HOME
380 FOR I = 1 TO N
390 PRINT I;" : "
400 CV = PEEK (37): VTAB (CV): HTAB (6): INPUT "X=":X(I)
410 IF NV = 2 THEN HTAB (6): INPUT "Y=":Y(I)
420 PRINT
430 NEXT I
440 REM
450 REM     CHIAMATA DELLA SUBROUTINE
460 REM
470 GOSUB 1000
480 REM
490 REM     GESTIONE DEI RISULTATI
500 REM
510 HOME
520 PRINT "MEDIA DI X = . ":U1
530 PRINT
540 PRINT "(N-1)SCARTO QUADRATICO DI X=":A3
550 PRINT "( N )SCARTO QUADRATICO DI X=":A5
560 PRINT

```

```

570 PRINT "(N-1)SCARTO QUADR. DI X MEDIO=";U3
580 PRINT "( N )SCARTO QUADR. DI X MEDIO=";U5
600 IF NV = 1 THEN END
610 PRINT "MEDIA DI Y   =";U2
620 PRINT
630 PRINT "(N-1)SCARTO QUADRATICO DI Y=";A4
640 PRINT "( N )SCARTO QUADRATICO DI Y=";A6
660 PRINT "(N-1)SCARTO QUADR. DI Y MEDIO=";U4
670 PRINT "( N )SCARTO QUADR. DI Y MEDIO=";U6
680 PRINT
690 PRINT "(N-1) COVARIANZA DI X,Y = ";U7
700 PRINT "( N ) COVARIANZA DI X,Y = ";A7
710 PRINT : PRINT
720 PRINT "COEF. DI CORRELAZIONE = ";U8
730 END

1000 REM *****
1010 REM SUBROUTINE PER IL CALCOLO DELLE FUNZIONI
1020 REM STATISTICHE DI UNA O DUE VARIABILI
1030 REM
1040 REM REGOLE PER L'USO DEL SOTTOPROGRAMMA
1050 REM
1060 REM 1.DATI NECESSARI:
1070 REM * NV=TIPO DI CALCOLO,
1080 REM NV=1:FUNZIONI STATISTICHE PER UNA VARIABILE
1090 REM NV=2:FUNZIONI STATISTICHE PER DUE VARIABILI
1100 REM * N=NUMERO DI DATI SPERIMENTALI
1110 REM * SE NV=1,
1120 REM X(I) (I=1,...,N) SONO GLI N VALORI DELLA
1130 REM VARIABILE
1140 REM * SE NV=2,
1150 REM (X(I),Y(I)) (I=1,...,N) SONO LE N COPPIE
1160 REM 2.RISULTATI FORNITI:
1170 REM * V1=MEDIA DI X
1180 REM * V2=MEDIA DI Y
1190 REM * V3=(N-1) SCARTO QUADRATICO DI X MEDIO
1200 REM * V4=(N-1) SCARTO QUADRATICO DI Y MEDIO
1210 REM * V5=(N-1)SCARTO QUADRATICO DI Y MEDIO
1220 REM * V6= (N) SCARTO QUADRATICO DI Y MEDIO
1230 REM * V7=(N-1) COVARIANZA DI X,Y
1240 REM * V8= COEFFICIENTE DI CORRELAZIONE
1250 REM * A3=(N-1) SCARTO QUADRATICO DI X
1260 REM * A4=(N-1) SCARTO QUADRATICO DI Y
1270 REM * A5= (N) SCARTO QUADRATICO DI X
1280 REM * A6= (N) SCARTO QUADRATICO DI Y
1290 REM * A7= (N) COVARIANZA DI X,Y
1300 REM 3.VARIABILI UTILIZZATE:
1310 REM A3,A4,A5,A6,A7,I,N,N1,NV,NR,V1,V2,V3,V4,
1320 REM V5,V6,V7,V8
1330 REM 4.VEVITORI UTILIZZATI:
1340 REM X(N),Y(N)
1350 REM *****
1360 REM
1370 REM INIZIALIZZAZIONE
1380 REM
1390 V1 = 0:V2 = 0:V3 = 0
1400 V4 = 0:V5 = 0
1410 N1 = N - 1:NR = SQR (N)
1420 REM
1430 REM SCELTA RELATIVA AL TIPO DI CALCOLO
1440 REM
1450 ON NV GOTO 1460,1600
1460 REM
1470 REM CALCOLO PER UNA VARIABILE
1480 REM
1490 FOR I = 1 TO N
1500 V1 = V1 + X(I)
1510 V3 = V3 + X(I) ^ 2

```

```

1520 NEXT I
1530 U6 = U3 - U1 * U1 / N
1540 U1 = U1 / N
1550 A3 = SQR (U6 / N1)
1560 A5 = SQR (U6 / N)
1570 U5 = A5 / NR
1580 U3 = A3 / NR
1590 RETURN
1600 REM
1610 REM  CALCOLO PER DUE VARIABILI
1620 REM
1630 FOR I = 1 TO N
1640 U1 = U1 + X(I)
1650 U2 = U2 + Y(I)
1660 U3 = U3 + X(I) ^ 2
1670 U4 = U4 + Y(I) ^ 2
1680 U5 = U5 + X(I) * Y(I)
1690 NEXT I
1700 U6 = U3 - U1 * U1 / N
1710 U7 = U4 - U2 * U2 / N
1720 U8 = U5 - (U1 * U2) / N
1730 U1 = U1 / N
1740 U2 = U2 / N
1750 A3 = SQR (U6 / N1)
1760 A4 = SQR (U7 / N1)
1770 A5 = SQR (U6 / N)
1780 A6 = SQR (U7 / N)
1790 U7 = U8 / N1
1800 A7 = U8 / N
1810 U3 = A3 / NR
1820 U4 = A4 / NR
1830 U5 = A5 / NR
1840 U6 = A6 / NR
1850 IF (A3 = 0) OR (A4 = 0) THEN RETURN : REM  TEST SULL'ESISTENZA DEL
      COEFFICIENTE DI CORRELAZIONE
1860 U8 = U8 / (N * A5 * A6)
1870 RETURN

```

## 4 — ESEMPIO PRATICO

Il programma viene utilizzato per il calcolo delle funzioni statistiche di due variabili  $x$  ed  $y$  che assumono i seguenti valori:

$x = 1, 2, 3, 4, 5$   
 $y = 12, 9, 7, 15, 6$

```

RUN
CALCOLO DELLE FUNZIONI STATISTICHE
PER UNA O DUE VARIABILI

```

AUTORE:H.HAUT

```

1. DEFINIRE IL TIPO DI CALCOLO:
1=FUNZIONI STATISTICHE PER UN INSIE-
ME DI N VALORI X(I)
2=FUNZIONI STATISTICHE PER UN INSIE-
ME DI COPPIE (X(I),Y(I))

```

2. INTRODURRE POT TI NUMERO DI VALORI  
O DI COPPIE

3. INTRODURRE SUCCESSIVAMENTE I DATI  
SPERIMENTALI  
PREMERE UN TASTO PER PROSEGUIRE  
SCELTA DEL TIPO DI CALCOLO (1 O 2) 2

NUMERO DI VALORI O DI COPPIE: 5

1 :  
X=1  
Y=12

2 :  
X=2  
Y=9

3 :  
X=3  
Y=7

4 :  
X=4  
Y=15

5 :  
X=5  
Y=6

MEDIA DI X = 3

(N-1)SCARTO QUADRATICO DI X=1.58113883

( N )SCARTO QUADRATICO DI X=1.41421356

(N-1)SCARTO QUADR. DI X MEDIO=.707106782

( N )SCARTO QUADR. DI X MEDIO=.632455532

MEDIA DI Y =9.8

(N-1)SCARTO QUADRATICO DI Y=3.70135111

( N )SCARTO QUADRATICO DI Y=3.31058908

(N-1)SCARTO QUADR. DI Y MEDIO=1.65529454

( N )SCARTO QUADR. DI Y MEDIO=1.48054045

(N-1) COVARIANZA DI X,Y = -1.5

( N ) COVARIANZA DI X,Y = -1.2

COEF. DI CORRELAZIONE = -.256307296

tempo d'esecuzione: 1''4

memoria richiesta: 3744 bytes (senza REM : 1584)





## PROGRAMMA NUMERO 12

# CALCOLO DELLE MEDIE E DEI MOMENTI DI UNA VARIABILE STOCASTICA

### 1 — DESCRIZIONE DEL METODO

Il programma realizzato permette di calcolare il valore delle medie e dei momenti relativi ad una variabile stocastica  $x$  che assume i valori  $x_1, x_2, \dots, x_n$  con frequenza  $f_1, f_2, \dots, f_n$ .

Le medie ed i momenti sono definiti dalle formule:

- media aritmetica

$$A = 1/M \sum f_i x_i \quad (i = 1, 2, \dots, n) \quad M = \sum f_i$$

- media geometrica

$$G = (\prod x_i^{f_i})^{1/M}$$

- media armonica

$$H = M / (\sum f_i / x_i)$$

- media generalizzata

$$M(t) = (1/M \sum f_i (x_i)^t)^{1/t}$$

— momenti

$$m1 = A$$

$$m2 = 1/M \sum f_i (x_i - A)^2$$

$$m3 = 1/M \sum f_i (x_i - A)^3$$

$$m4 = 1/M \sum f_i (x_i - A)^4$$

— coefficiente di appiattimento

$$AP = m4/(m2)^2$$

— coefficiente di asimmetria

$$AS = m3/(m2)^{3/2}$$

Si utilizzano le seguenti variabili:

$$V1 = \sum f_i x_i$$

$$V4 = \sum f_i x_i^4$$

$$V7 = \sum f_i (x_i)^t$$

$$V2 = \sum f_i x_i^2$$

$$V5 = \prod x_i^{f_i}$$

$$M = \sum f_i$$

$$V3 = \sum f_i x_i^3$$

$$V6 = \sum f_i / x_i$$

e si ottiene:

$$A = V1/M$$

$$G = (V5)^{1/M}$$

$$H = M/V6$$

$$M(t) = MT = (V7/M)^{1/t}$$


$$m2 = M2 = (V2 - 2 * A * V1 + A^2 M)/M$$


$$m3 = M3 = (V3 - 3 * A * V2 + 3 * A^2 V1 - A^3 M)/M$$


$$m4 = M4 = (V4 + 4 * A * V3 + 6 * A^2 V2 - 4 * A^3 V1 + A^4 M)/M$$

**Nota:** Se uno degli  $x_i$  è uguale a zero, la media geometrica è nulla e quella armonica non è definita.

2 — DIAGRAMMA A BLOCCHI







### 3 — PROGRAMMA

```

1  REM          CALCOLO DELLE MEDIE
2  REM          E DEI MOMENTI,
3  REM
4  REM          AUTORE:H.HAUT
5  REM
6  REM
7  REM  DESCRIZIONE:
8  REM      IL PROGRAMMA PERMETTE DI CALCOLARE LE VARIE
9  REM      MEDIE ED I MOMENTI DI UNA VARIABILE STATISTICA
10 REM      X PRENDENDO I VALORI X(1),...,X(N) CON
11 REM      FREQUENZE F(1),...,F(N)
12 REM
13 REM *****
14 REM
15 REM
100 REM
110 REM      REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 3);"CALCOLO DELLE MEDIE E DEI MOMENTI"
150 PRINT TAB( 6);"DI UNA VARIABILE STATISTICA,"
160 PRINT : PRINT TAB( 20);"AUTORE:H.HAUT"
170 UTAB (9)
180 PRINT "1.DEFINIRE IL NUMERO DI VALORI          PER LA VARIABILE STA
TISTICA"
190 PRINT
200 PRINT "2.INTRODURRE SUCCESSIVAMENTE I          VALORI E LE RELATIVE
FREQUENZE"
210 UTAB (19): PRINT "PREMERE UN TASTO PER PROSEGUIRE  ": GET A$
220 HOME
230 REM
240 REM      GESTIONE DEI DATI IN INPUT
250 REM
260 INPUT "NUMERO DI VALORI= ";N
270 REM
280 REM      ISTRUZIONI DI DIMENSIONAMENTO
290 REM
300 DIM X(N),F(N)
310 REM
320 REM      INSERIMENTO DATI
330 REM
340 FOR I = 1 TO N
350 PRINT : PRINT I
360 CV = PEEK (37): UTAB (CV)
370 HTAB (5): INPUT "X=";X(I)
380 HTAB (5): INPUT "F=";F(I)
390 NEXT I
400 PRINT : INPUT "CALCOLO DELLA MEDIA GENERALIZZATA M(T)  PER T= ";T
410 REM
420 REM      CHIAMATA DEL SOTTOPROGRAMMA
430 REM
440 GOSUB 1000
450 REM
460 REM      GESTIONE DEI RISULTATI
470 REM
480 HOME
490 PRINT "MEDIA ARITMETICA = ";A
500 PRINT
510 PRINT "MEDIA GEOMETRICA = ";G: PRINT
520 IF TE = 1 THEN PRINT "MEDIA ARMONICA NON DEFINITA": PRINT : GOTO 540
530 PRINT "MEDIA ARMONICA  = ";H: PRINT
540 PRINT "MOMENTI:"

```

```

550 PRINT "      M1=";M1
560 PRINT "      M2=";M2
570 PRINT "      M3=";M3
580 PRINT "      M4=";M4: PRINT
590 PRINT "COEF. D'APPIATTIMENTO=";AP: PRINT
600 PRINT "COEF. D'ASIMMETRIA   " ";AS: PRINT
610 PRINT "MEDIA GENER. M(';T;')"; TAB( 22); "=";MT
620 END
1000 REM *****
1010 REM SOTTOPROGRAMMA PER IL CALCOLO DELLE MEDIE E DEI MOMENTI
1020 REM DI UNA VARIABILE STATISTICA X
1030 REM
1040 REM REGOLE PER L'USO DEL SOTTOPROGRAMMA
1050 REM
1060 REM      1.DATI NECESSARI:
1070 REM      * N=NUMERO DI VALORI RELATIVI ALLA
1080 REM        VARIABILE STATISTICA X
1090 REM      * X(1),...,X(N) =VALORI SPERIMENTALI
1100 REM        PER LA VARIABILE X
1110 REM      * F(1),...,F(N) =FREQUENZE DI TALI VALORI
1120 REM      * T=COEFFICIENTE PER LA MEDIA GENERALIZZATA
1130 REM        DA CALCOLARE
1140 REM      2.RISULTATI FORNITI:
1150 REM      * A=MEDIA ARITMETICA
1160 REM      * G=MEDIA GEOMETRICA
1170 REM      * H=MEDIA ARMONICA
1180 REM      * M1,M2,M3,M4 = MOMENTI DI ORDINE 1,2,3,4
1190 REM      * AP=COEFFICIENTE DI APPIATTIMENTO
1200 REM      * AS=COEFFICIENTE DI ASIMMETRIA
1210 REM      * MT=MEDIA GENERALIZZATA M(T)
1220 REM      * TE= VARIABILE DI CONTROLLO: SE TE=1
1230 REM        LA MEDIA ARMONICA NON E' DEFINITA
1240 REM      3.VARIABILI UTILIZZATE:
1250 REM      A,AP,AS,G,H,I,M,M1,M2,M3,M4,MT
1260 REM      N,T,TF,U1,U2,U3,U4,U5,U6,U7,UT
1270 REM      4.VEVETTORI UTILIZZATI:
1280 REM      F(N),X(N)
1290 REM *****
1300 REM *****
1310 REM *****
1320 REM INIZIALIZZAZIONI
1330 REM
1340 TE = 0
1350 U1 = 0:U2 = 0:U3 = 0:U4 = 0
1360 U5 = 1:U6 = 0:U7 = 0:M = 0
1370 REM
1380 REM CALCOLO DELLE SOMMATORIE
1390 REM
1400 FOR I = 1 TO N
1410 UT = F(I) * X(I)
1420 U1 = U1 + UT
1430 U2 = U2 + UT * X(I)
1440 U3 = U3 + UT * X(I) ^ 2
1450 U4 = U4 + UT * X(I) ^ 3
1460 M = M + F(I)
1470 U7 = U7 + F(I) * X(I) ^ T
1480 IF X(I) = 0 THEN TE = 1: GOTO 1520: REM TEST SUL VALORE DEGLI X(I)
1490 IF TF = 1 THEN GOTO 1520: REM SE UN X=0 LA MEDIA ARMONICA NON ESISTE E QUELLA GEOMETRICA E' ZERO
1500 U5 = U5 * X(I) ^ F(I)
1510 U6 = U6 + F(I) / X(I)
1520 NEXT I
1530 REM
1540 REM CALCOLO DELLE FUNZIONI
1550 REM
1560 A = U1 / M
1570 MT = (U7 / M) ^ (1 / T)

```

```

1580 M1 = A
1590 M2 = (U2 - 2 * A * U1 + A * A * M) / M
1600 M3 = (U3 - 3 * A * U2 + 3 * A * A * U1 - M * A ^ 3) / M
1610 M4 = (U4 - 4 * A * U3 + 6 * A * A * U2 - 4 * U1 * A ^ 3 + M * A ^ 4) /
      M
1620 AP = M4 / M2 / M2
1630 AS = M3 / (M2 ^ 1.5)
1640 IF TF = 1 THEN G = 0: RETURN : REM UN X(T)=0
1650 G = (U5) ^ (1 / M)
1660 H = M / U6
1670 RETURN

```

## 4 — ESEMPIO PRATICO

Il calcolo è effettuato per le tre coppie di valori dato/frequenza: (1,6), (3,4), (5,8) utilizzando il coefficiente 2 per la media generalizzata.

```

RUN
CALCOLO DELLE MEDIE E DEI MOMENTI
DI UNA VARIABILE STATISTICA.

      AUTORE:H.HAUT
1.DEFINIRE IL NUMERO DI VALORI
  PER LA VARIABILE STATISTICA

2.INTRODURRE SUCCESSIVAMENTE I
  VALORI E LE RELATIVE FREQUENZE
PREMERE UN TASTO PER PROSEGUIRE
NUMERO DI VALORI= 3

1
  X=1
  F=6

2
  X=3
  F=4

3
  X=5
  F=8

CALCOLO DELLA MEDIA GENERALIZZATA M(T)
PER T= 2
MEDIA ARITMETICA = 3.22222222
MEDIA GEOMETRICA = 2.61023904
MEDIA ARMONICA   = 2.01492537

MOMENTI:
  M1=3.22222222
  M2=3.0617284
  M3=-1.16323733
  M4=12.5688157

COEF. D'APPIATTIMENTO=1.34079084
COEF. D'ASIMMETRIA   =-.217129252
MEDIA GENER. M(2)    =3.66666667

```

tempo d'esecuzione: 1''2

memoria richiesta: 3745 bytes (senza REM : 1584).



## PROGRAMMA NUMERO 13

# DISTRIBUZIONI STATISTICHE

Il programma è applicabile ad una qualsiasi delle sei distribuzioni statistiche seguenti: binomiale, di Poisson, normale, normale a due variabili, "chi quadro", t di Student.

Per la maggior parte di esse viene inoltre calcolata sia la funzione di densità di probabilità che quella di probabilità cumulata.

*Riferimenti bibliografici:* A1, G1, H2, J2, M1.

## 1 — DEFINIZIONI E METODI

### 1.1 — Distribuzione binomiale

#### a) Definizione

Sia  $p$  la probabilità favorevole per il verificarsi di un certo evento. La probabilità che tale evento si presenti  $x$  volte nel corso di  $N$  prove è data dalla legge binomiale:

$$P(x) = \binom{N}{x} p^x (1-p)^{N-x}$$

ove

$$\binom{N}{x} = \frac{N!}{x! (N-x)!}$$

La funzione cumulativa di probabilità in  $x$  (vale a dire la probabilità di ottenere al più  $x$  successi) è data da:

$$\sum_{k=0}^x p(k)$$

*b) Metodo*

Si utilizza la formula di ricorrenza:

$$p(l) = \frac{N + 1 - l}{l} \cdot \frac{p}{q} p(l - 1) \quad l = 1, 2, \dots, x$$

con  $q = (1 - p)$  e  $p(0) = q^N$

## 1.2 — Distribuzione di Poisson

*a) Definizione*

La funzione di densità di probabilità di una distribuzione di Poisson avente media  $\mu$  è definita come:

$$p(x) = \frac{\mu^x}{x!} e^{-\mu}$$

e la funzione cumulativa è data:

$$\sum_{k=0}^x p(k)$$

*b) Metodo*

Si utilizza la formula di ricorrenza:

$$p(x) = \frac{\mu}{x} p(x - 1)$$

con

$$p(0) = e^{-\mu}$$

## 1.3 — Distribuzione normale

*a) Definizione*

La densità di probabilità relativa ad una distribuzione normale (di media  $\mu$  e varianza  $\sigma^2$ ) viene definita nel modo seguente:

$$f(x) = \frac{1}{\sqrt{2\pi}} \cdot \frac{1}{\sigma} \exp\left(-\frac{(x-\mu)^2}{2\sigma^2}\right) = Z\left(\frac{x-\mu}{\sigma}\right)$$

*b) Metodo*

La funzione cumulativa PX si ottiene nel modo seguente:

$$PX = \int_{-\infty}^{\frac{x-\mu}{\sigma}} Z(t) dt = P\left(\frac{x-\mu}{\sigma}\right) = 1 - Q\left(\frac{x-\mu}{\sigma}\right)$$

utilizzando per il calcolo la funzione  $Q(y)$  definita con il seguente sviluppo:

$$Q(y) = Z(y) = [b_4 t + b_3 t^2 + b_2 t^3 + b_1 t^4 + b_0 t^5] + \varepsilon(x)$$

In questo sviluppo l'errore è limitato dalla condizione  $[\varepsilon(x)] < 7.5 \cdot 10^{-8}$  e i coefficienti  $t$  e  $b_i$  sono definiti tramite le formule seguenti:

$$\begin{aligned} t &= 1/(1 + py) & p &= .2316419 \\ b_0 &= 1.330274429 & b_1 &= -1.821255978 \\ b_2 &= 1.781477937 & b_3 &= -.356563782 & b_4 &= .319381530 \end{aligned}$$

## 1.4 — Distribuzione normale a due variabili

Per questa distribuzione è stato previsto solo il calcolo della densità di probabilità per una coppia di valori delle variabili aleatorie  $x$  e  $y$ .

La funzione di densità viene definita nel modo seguente:

$$f(x, y) = \frac{1}{2\pi\sigma_x \sigma_y} \frac{1}{\sqrt{1 - \rho^2}} \exp - \left[ \frac{\left(\frac{x - \mu_x}{\sigma_x}\right)^2 - 2\rho\left(\frac{x - \mu_x}{\sigma_x}\right)\left(\frac{y - \mu_y}{\sigma_y}\right) + \left(\frac{y - \mu_y}{\sigma_y}\right)^2}{2(1 - \rho^2)} \right]$$

ove  $\mu_x$  = media di  $x$   
 $\sigma_x^2$  = varianza di  $x$   
 $\mu_y$  = media di  $y$   
 $\sigma_y^2$  = varianza di  $y$   
 $\rho$  = coefficiente di correlazione tra  $x$  e  $y$

## 1.5 — Distribuzione “chi quadro”

### a) Definizione

La funzione di densità relativa alla distribuzione “chi quadro” a  $V$  gradi di libertà, è definita come:

$$f_X = f(x|v) = \frac{1}{2^{v/2} \Gamma(v/2)} e^{-x/2} x^{(v/2 - 1)} \quad (x > 0)$$

La funzione cumulativa di probabilità si può ottenere tramite il seguente sviluppo in serie:

$$PX = \int_0^x f(x | v) dx = P(x | v) = (x/2)^{v/2} \frac{e^{-x/2}}{\Gamma(v/2 + 1)} \left( 1 + \sum_{r=1}^{\infty} \frac{x^r}{(v+2)(v+4) \dots (v+2r)} \right)$$

#### b) Metodo

Si calcola dapprima  $F(v/2)$  che, a seconda che  $v$  sia pari o dispari, è dato da:

$$\text{se } v = 2n \quad \Gamma(v/2) = 1.2.3 \dots (n-1)$$

$$\text{se } v = 2n + 1, \quad (v/2) = 1/2.3/2.5/2 \dots \frac{2n-1}{2} \sqrt{\pi}$$

Si utilizza poi questo sviluppo per calcolare  $PX$ , troncando la serie quando il termine  $n$ -esimo risulta inferiore a  $10^{-8}$ .

## 1.6 — Distribuzione $t$ di Student

#### a) Definizione

La densità di probabilità relativa alla distribuzione di Student a  $v$  gradi di libertà è definita come:

$$f(x | v) = \frac{[(v-1)/2]!}{\sqrt{\pi v} [(v-2)/2]!} (1 + x^2/v)^{-(v+1)/2}$$

Si definisce poi come probabilità che la variabile aleatoria  $x$  assuma un valore più piccolo di  $t$  (in valore assoluto):

$$FX = \text{Prob}(|x| \leq t) = A(t | v) = \int_{-t}^t f(x | v) dx$$

La funzione cumulativa di probabilità risulta pertanto data da:

$$PX = \text{Prob}(x \leq t) = \frac{1 + A(t | v)}{2}$$


b) Metodo


La funzione  $A(t | \nu)$  può essere calcolata tramite il seguente sviluppo in serie:


$$A(t | \nu) = \begin{cases} (\nu > 1 \text{ e } \nu \text{ dispari}) : 2/\pi \left[ \Theta + \sin \Theta (\cos \Theta + 2/3 \cos^3 \Theta + \dots + \frac{2 \cdot 4 \cdot \dots \cdot (\nu - 3)}{3 \cdot 5 \cdot \dots \cdot (\nu - 2)} \cos^{\nu-2} \Theta) \right] \\ (\nu = 1) & 2/\pi \Theta \\ (\nu \text{ pari}) & \sin \Theta \left[ 1 + 1/2 \cos^2 \Theta + \dots + \frac{1 \cdot 3 \cdot 5 \cdot \dots \cdot (\nu - 3)}{2 \cdot 4 \cdot 6 \cdot \dots \cdot (\nu - 2)} \cos^{\nu-2} \Theta \right] \end{cases}$$

$$\text{con } \text{tg } \Theta = t/\sqrt{\nu}$$


## 2 — DIAGRAMMA A BLOCCHI





























### 3 — PROGRAMMA

```

1  REM      CALCOLO DELLE DISTRIBUZIONI
2  REM      STATISTICHE.
3  REM
4  REM      AUTORE:H.HAUT
5  REM
6  REM
7  REM      DESCRIZIONE:
8  REM      IL PROGRAMMA PERMETTE DI CALCOLARE
9  REM      LE DISTRIBUZIONI SEGUENTI
10 REM      1.DISTRIBUZIONE BINOMIALE
11 REM      2.DISTRIBUZIONE DI POISSON
12 REM      3.DISTRIBUZIONE NORMALE
13 REM      4.DISTRIBUZIONE NORMALE A 2 VARIABILI
14 REM      5.DISTRIBUZIONE CHI-QUADRO
15 REM      6.DISTRIBUZIONE T DI STUDENT
16 REM
17 REM      *****
18 REM
19 REM
100 REM
110 REM      REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 5);"CALCOLO DELLE DISTRIBUZIONI"
150 PRINT TAB( 14);"STATISTICHE "
160 PRINT
170 PRINT TAB( 20);"AUTORE:H.HAUT"
180 PRINT : PRINT
190 PRINT "1.SCEGLIERE LA DISTRIBUZIONE:          1=DISTR. BINOMIALE"
200 PRINT " 2=DISTR. DI POISSON                    3=DISTR. NORMALE
        4=DISTR. NORMALE A 2 VARIABILI          5=DISTR.
        CHI-QUADRO                            6=DISTR. T DI STUDENT"
210 PRINT
220 PRINT "2.DEFINIRE I PARAMETRI DELLA DISTRIBU-   ZIONE SCELTA"
230 PRINT
240 PRINT "3.INSERIRE IL VALORE DELLA VARIABILE     ALEATORIA PER CUI DE
VE ESSERE          EFFETTUATO IL CALCOLO"
250 VTAB (22): PRINT "PREMERE UN TASTO PER PROSEGUIRE  ": GET A$
260 REM
270 REM      ISTRUZIONI DI DIMENSIONAMENTO
280 REM
290 DIM R(4)
300 REM
310 REM      GESTIONE DEI DATI IN INPUT
320 REM
330 HOME
340 INPUT "SCELTA DELLA DISTRIBUZIONE  ":CH
350 HOME
360 ON CH GOTO 370,410,450,490,590,630
370 PRINT "DISTRIBUZIONE BINOMIALE:          P=PROBABILITA' DI S
SUCCESSO          N=NUMERO DI PROVE
      DI EVENTI FAVOREVOLI"
380 PRINT : INPUT "P=":P
390 INPUT "N=":N: PRINT
400 INPUT "X=":X: GOTO 490
410 PRINT "DISTRIBUZIONE DI POISSON:          MU=MEDIA"
420 PRINT : INPUT "MU=":M
430 PRINT : INPUT " X=":X
440 GOTO 490
450 PRINT "DISTRIBUZIONE NORMALE:          MU=MEDIA
      S=SCARTO QUADRATICO"
460 PRINT : INPUT "MU=":M
470 INPUT " S=":S: PRINT
480 INPUT " X=":X: GOTO 490

```



```

490 PRINT "DISTRIBUZIONE NORMALE A 2 VARIABILI X,Y:      MX=MEDIA PER X
      MY=MEDIA PER Y      SX=SCAR
      TO QUADRATICO DI X
500 PRINT "      SY=SCARTO QUADRATICO DI Y      RO=COEFFICIENTE DI
      CORRELAZIONE"
510 PRINT : INPUT "MX=";MX
520 INPUT "SX=";SX
530 INPUT "MY=";MY
540 INPUT "SY=";SY
550 INPUT "RO=";RO
560 PRINT : INPUT "X=";X
570 INPUT "Y=";Y
580 GOTO 690
590 PRINT "DISTRIBUZIONE CHI-QUADRO :      NU=NUMERO DI GRADI D
      I LIBERTA'"
600 PRINT : INPUT "NU=";NU
610 PRINT : INPUT "X=";X
620 GOTO 690
630 PRINT "DISTRIBUZIONE T DI STUDENT:      NU=NUMERO DI GRADI
      DI LIBERTA'"
640 PRINT : INPUT "NU=";NU
650 PRINT : INPUT "X=";X
660 REM
670 REM   CHIAMATA DEL SOTTOPROGRAMMA PRESCELTO
680 REM
690 ON CH GOSUB 1000,2000,3000,4000,5000,6000
700 REM
710 REM   GESTIONE DEI RISULTATI:
720 REM
730 PRINT
740 A$ = "PROB(VAR.ALEATORIA =X)="
750 B$ = "PROB(VAR.ALEATORIA<=X)="
760 C$ = "PROB(VAR.ALEATORIA>=X)="
770 IF CH = 4 THEN PRINT "PROB(VAR.ALEATORIE =X,Y)=";FXY: GOTO 860
780 IF CH < > 6 THEN GOTO 830
790 PRINT "PROB(-X<=VAR.ALEAT.<=X)=";BT
800 PRINT B$;PX
810 PRINT C$;QX
820 GOTO 860
830 PRINT A$;FX
840 PRINT B$;PX
850 PRINT C$;QX
860 PRINT : INPUT "ALTRO CALCOLO? (S O N): ";D$
870 IF D$ = "N" THEN END
880 PRINT : INPUT "X=";X
890 IF CH = 4 THEN INPUT "Y=";Y
900 GOTO 690
1000 REM *****
1010 REM   SUBROUTINE   DISTRIBUZIONE BINOMIALE
1020 REM
1030 REM   REGOLE PER L'USO DEL SOTTOPROGRAMMA
1040 REM       1.DATI NECESSARI:
1050 REM           * P=PROBABILITA' DI SUCCESSO
1060 REM           * N=NUMERO DI PROVE
1070 REM           * X=VALORE DELLA VARIABILE ALEATORIA
1080 REM       2.RISULTATI FORNITI:
1090 REM           * FX=PROB(VAR.ALEATORIA =X)
1100 REM           * PX=PROB(VAR.ALEATORIA<=X)
1110 REM           * QX=PROB(VAR.ALEATORIA>=X)
1120 REM       3.VARIABILI UTILIZZATE:
1130 REM           FX,I,N,N1,P,PX,Q,QX,T,X
1140 REM *****
1150 Q = 1 - P
1160 T = P / Q
1170 N1 = N + 1
1180 FX = Q ^ N: REM   FX=PROB(0)
1190 PX = FX

```

```

1200 IF X = 0 THEN GOTO 1260: REM  TERMINE
1210 REM  CALCOLO PER RICORRENZA
1220 FOR I = 1 TO X
1230 FX = (N1 - I) * T * FX / I
1240 PX = PX + FX
1250 NEXT I
1260 QX = 1 + FX - PX
1270 RETURN
2000 REM  *****
2010 REM  SUBROUTINE  DISTRIBUZIONE DI POISSON
2020 REM
2030 REM  REGOLE PER L'USO DEL SOTTOPROGRAMMA
2040 REM  1.DATI NECESSARI:
2050 REM      * M=MEDIA
2060 REM      * X=VALORE DELLA VARIABILE ALEATORIA
2070 REM  2.RISULTATI FORNITI:
2080 REM      * FX=PROB(VAR.ALEATORIA =X)
2090 REM      * PX=PROB(VAR.ALEATORIA<=X)
2100 REM      * QX=PROB(VAR.ALEATORIA>=X)
2110 REM  3.VARIABILI UTILIZZATE:
2120 REM      FX,I,M,PX,QX,X
2130 REM  *****
2140 FX = EXP ( - M)
2150 PX = FX
2160 IF X = 0 THEN GOTO 2220: REM  TERMINE
2170 REM  FORMULA DI RICORRENZA
2180 FOR I = 1 TO X
2190 FX = FX * M / I
2200 PX = PX + FX
2210 NEXT I
2220 QX = 1 + FX - PX
2230 RETURN
3000 REM  *****
3010 REM  SUBROUTINE  DISTRIBUZIONE NORMALE
3020 REM
3030 REM  REGOLE PER L'USO DEL SOTTOPROGRAMMA
3040 REM  1.DATI NECESSARI:
3050 REM      * M=MEDIA
3060 REM      * S=SCARTO QUADRATICO
3070 REM      * X=VALORE DELLA VARIABILE ALEATORIA
3080 REM  2.RISULTATI FORNITI:
3090 REM      * FX=PROB(VAR.ALEATORIA =X)
3100 REM      * PX=PROB(VAR.ALEATORIA<=X)
3110 REM      * QX=PROB(VAR.ALEATORIA>=X)
3120 REM  3.VARIABILI UTILIZZATE:
3130 REM      FX,I,M,PO,PP,PX,QX,S,T,X,Y,ZY
3140 REM  4.ETTORE UTILIZZATO:
3150 REM      B(4)
3160 REM  *****
3170 REM
3180 REM  DEFINIZIONE DEI COEFFICIENTI B(I)
3190 REM
3200 B(0) = 1,330274429:B(1) = - 1,821255978:B(2) = 1,781477937:B(3) = -
    .356563782:B(4) = .319381530
3210 PP = .2316419
3220 Y = (X - M) / S
3230 ZY = EXP ( - Y * Y / 2) / 2,506628275
3240 FX = ZY / S
3250 REM
3260 REM  CALCOLO DI QX CON IL METODO DI HORNER
3270 REM
3280 T = 1 / (1 + PP * Y)
3290 PO = 0
3300 FOR I = 0 TO 4
3310 PO = PO * T + B(I)
3320 NEXT I
3330 PO = PO * T

```

```

3340 QX = ZY * P0
3350 PX = 1 - QX
3360 RETURN
4000 REM *****
4010 REM SUBROUTINE DISTRIBUZIONE NORMALE A 2 VARIABILI
4020 REM
4030 REM REGOLE PER L'USO DEL SOTTOPROGRAMMA
4040 REM 1.DATI NECESSARI:
4050 REM * MX=MEDIA DI X
4060 REM * MY=MEDIA DI Y
4070 REM * SX=SCARTO QUADRATICO DI X
4080 REM * SY=SCARTO QUADRATICO DI Y
4090 REM * RO=COEFFICIENTE DI CORRELAZIONE
4100 REM * X=VALORE DELLA PRIMA VARIABILE ALEATORIA
4110 REM * Y=VALORE DELLA SECONDA VARIABILE ALEATORIA
4120 REM 2.RISULTATO FORNITO:
4130 REM * FXY=PROB(VAR.ALEATORIE =X,Y)
4140 REM 3.VARIABILI UTILIZZATE:
4150 REM FXY,MX,MY,R,RO,S,SX,SY,T,UT,X,Y
4160 REM *****
4170 S = (X - MX) / SX
4180 T = (Y - MY) / SY
4190 R = 2 * (1 - RO * RO)
4200 UT = S * S + T * T - 2 * RO * S * T
4210 UT = EXP (- UT / R)
4220 S = SX * SY * SQR (2 * R) * 3.141592654
4230 FXY = UT / S
4240 RETURN
5000 REM *****
5010 REM SUBROUTINE DISTRIBUZIONE CHI-QUADRO
5020 REM
5030 REM REGOLE PER L'USO DEL SOTTOPROGRAMMA:
5040 REM 1.DATI NECESSARI:
5050 REM * NU=NUMERO DI GRADI DI LIBERTA'
5060 REM * X=VALORE DELLA VARIABILE ALEATORIA
5070 REM 2.RISULTATI FORNITI:
5080 REM * FX=PROB(VAR.ALEATORIA =X)
5090 REM * FX=PROB(VAR.ALEATORIA<=X)
5100 REM * QX=PROB(VAR.ALEATORIA>=X)
5110 REM * GN=GAMMA(NU/2)
5120 REM 3.VARIABILI UTILIZZATE:
5130 REM AA,BB,FA,FM,FX,GN,I,N,N2,NU,PX,QX,X
5140 REM *****
5150 REM
5160 REM CALCOLO DI GN=GAMMA(NU/2)
5170 REM
5180 N2 = NU / 2
5190 GN = 1
5200 N = INT (N2)
5210 IF NU = 2 * N THEN GOTO 5310
5220 REM
5230 REM CALCOLO DI GN PER NU DISPARI
5240 REM
5250 IF NU = 1 THEN GOTO 5290
5260 FOR I = 1 TO 2 * N - 1 STEP 2
5270 GN = GN * I / 2
5280 NEXT I
5290 GN = GN * 1.7724528509
5300 GOTO 5380
5310 REM
5320 REM CALCOLO DI GN PER NU PARI
5330 REM
5340 IF NU = 2 THEN GOTO 5380
5350 FOR I = 1 TO N - 1 STEP 2
5360 GN = GN * I
5370 NEXT I
5380 REM

```

```

5390 REM      CALCOLO DI FX
5400 REM
5410 FX = X ^ (N2 - 1)
5420 AA = EXP ( - X / 2)
5430 BR = GN * (2 ^ N2)
5440 FX = FX * AA / BR
5450 REM
5460 REM      CALCOLO DELLO SVILUPPO DI FX
5470 REM
5480 FA = (X / 2) ^ N2
5490 FA = FA * AA / (N2 * GN)
5500 FM = 1
5510 PX = 1
5520 I = 2
5530 FM = FM * X / (NU + I)
5540 PX = PX + FM
5550 IF (FM * FA) > 1E - 8 THEN I = I + 2: GOTO 5530
5560 REM      LA PRECISIONE E' OTTENUTA
5570 PX = PX * FA
5580 QX = 1 - PX
5590 RETURN
6000 REM *****
6010 REM      SUBROUTINE      DISTRIBUZIONE T DI STUDENT
6020 REM
6030 REM      REGOLE PER L'USO DEL SOTTOPROGRAMMA
6040 REM      1.DATI NECESSARI:
6050 REM          * NU=NUMERO DI GRADI DI LIBERTA'
6060 REM          * X=VALORE DELLA VARIABILE ALEATORIA
6070 REM      2.RISULTATI FORNITI:
6080 REM          * RT=PROB(-X<=VAR.ALEATORIA<=X)
6090 REM          * PX=PROB(VAR.ALEATORIA<=X)
6100 REM          * QX=PROB(VAR.ALEATORIA>=X)
6110 REM      3.VARIABILI UTILIZZATE:
6120 REM          RT,I,NT,NU,PX,QX,T,TT,UT,X
6130 REM *****
6140 X = X / SQR (NU)
6150 T = ATN (X): REM      T=THETA
6160 IF NU = 1 THEN RT = T * .6366197724: GOTO 6450: REM      RT=A(X/1)
6170 NT = 2 * INT (NU / 2)
6180 IF NT = NU THEN GOTO 6330
6190 REM
6200 REM      CALCOLO DELLO SVILUPPO DI A(X/NU) PER NU DISPARI
6210 REM
6220 RT = COS (T)
6230 TT = RT * RT
6240 UT = RT
6250 IF NU = 3 THEN GOTO 6300
6260 FOR I = 2 TO NU - 3 STEP 2
6270 UT = UT * I * TT / (I + 1)
6280 RT = RT + UT
6290 NEXT I
6300 RT = RT * SIN (T)
6310 RT = (RT + T) * .6366197724
6320 GOTO 6450
6330 REM
6340 REM      CALCOLO DELLO SVILUPPO DI A(X/NU) PER NU PARI
6350 REM
6360 TT = COS (T) ^ 2
6370 RT = 1
6380 UT = 1
6390 IF NU = 2 THEN GOTO 6440
6400 FOR I = 1 TO NU - 3 STEP 2
6410 UT = UT * I * TT / (I + 1)
6420 RT = RT + UT
6430 NEXT I
6440 RT = RT * SIN (T)
6450 PX = (1 + RT) / 2

```

```

6460 QX = 1 - PX
6470 RETURN

```

## 4 — ESEMPIO PRATICO

L'esempio permette di calcolare le funzioni relative a ciascuna delle sei distribuzioni partendo dai seguenti parametri iniziali:

1 — Binomiale:       $p = .5$   
                           $n = 3$   
                           $x = 1$

2 — Poisson:         $\mu = 1.8$   
                           $x = 4$

3 — Normale:         $\mu = 2$   
                           $\sigma = 3$   
                           $x = 5$

4 — Doppia normale:  
                           $\mu_x = .1$   
                           $\sigma_x = 1.4$   
                           $\mu_y = 2.4$   
                           $\sigma_y = .9$   
                           $\rho = .75$   
                           $x = 1$   
                           $y = 2$

5 — Chi-quadro:     $\nu = 3$   
                           $x = 1$

6 — t di Student:     $\nu = 19$   
                           $x = .257$

```

RUN
  CALCOLO DELLE DISTRIBUZIONI
    STATISTICHE

    AUTORE:H.HAUT

```

```

1.SCEGLIERE LA DISTRIBUZIONE:
1=DISTR. BINOMIALE
2=DISTR. DI POISSON
3=DISTR. NORMALE
4=DISTR. NORMALE A 2 VARIABILI
5=DISTR. CHI-QUADRO
6=DISTR. T DI STUDENT

```

2.DEFINIRE I PARAMETRI DELLA DISTRIBUZIONE SCELTA

3.INSERIRE IL VALORE DELLA VARIABILE ALEATORIA PER CUI DEVE ESSERE EFFETTUATO IL CALCOLO

PREMERE UN TASTO PER PROSEGUIRE SCELTA DELLA DISTRIBUZIONE 1  
DISTRIBUZIONE BINOMIALE:

P=PROBABILITA' DI SUCCESSO

N=NUMERO DI PROVE

X=NUMERO DI EVENTI FAVOREVOLI

P=.5

N=3

X=1

PROB(VAR.ALEATORIA =X)=.375

PROB(VAR.ALEATORIA<=X)=.5

PROB(VAR.ALEATORIA>=X)=.875

ALTRO CALCOLO? (S O N): N

]

RUN

CALCOLO DELLE DISTRIBUZIONI  
STATISTICHE

AUTORE:H.HAUT

1.SCEGLIERE LA DISTRIBUZIONE:

1=DISTR. BINOMIALE

2=DISTR. DI POISSON

3=DISTR. NORMALE

4=DISTR. NORMALE A 2 VARIABILI

5=DISTR. CHI-QUADRO

6=DISTR. T DI STUDENT

2.DEFINIRE I PARAMETRI DELLA DISTRIBUZIONE SCELTA

3.INSERIRE IL VALORE DELLA VARIABILE ALEATORIA PER CUI DEVE ESSERE EFFETTUATO IL CALCOLO

PREMERE UN TASTO PER PROSEGUIRE SCELTA DELLA DISTRIBUZIONE 2  
DISTRIBUZIONE DI POISSON:

MU=MEDIA

MU=1.8

X=4

PROB(VAR.ALEATORIA =X)=.0723017337

PROB(VAR.ALEATORIA<=X)=.963593339

PROB(VAR.ALEATORIA>=X)=.108708394

ALTRO CALCOLO? (S O N): N

]

```

RUN
  CALCOLO DELLE DISTRIBUZIONI
    STATISTICHE

    AUTORE:H.HAUT

```

```

1.SCEGLIERE LA DISTRIBUZIONE:
  1=DISTR. BINOMIALE
  2=DISTR. DI POISSON
  3=DISTR. NORMALE
  4=DISTR. NORMALE A 2 VARIABILI
  5=DISTR. CHI-QUADRO
  6=DISTR. T DI STUDENT

2.DEFINIRE I PARAMETRI DELLA DISTRIBU-
  ZIONE SCELTA

3.INSERIRE IL VALORE DELLA VARIABILE
  ALEATORIA PER CUI DEVE ESSERE
  EFFETTUATO IL CALCOLO
PREMERE UN TASTO PER PROSEGUIRE
SCELTA DELLA DISTRIBUZIONE      3
DISTRIBUZIONE NORMALE:
  MU=MEDIA
  S=SCARTO QUADRATICO

MU=2
S=3

X=5

PROB(VAR.ALEATORIA =X)=.0806569082
PROB(VAR.ALEATORIA<=X)=.84134474
PROB(VAR.ALEATORIA>=X)=.15865526

ALTRO CALCOLO? (S O N): N

]

```

```

RUN
  CALCOLO DELLE DISTRIBUZIONI
    STATISTICHE

    AUTORE:H.HAUT

```

```

1.SCEGLIERE LA DISTRIBUZIONE:
  1=DISTR. BINOMIALE
  2=DISTR. DI POISSON
  3=DISTR. NORMALE
  4=DISTR. NORMALE A 2 VARIABILI
  5=DISTR. CHI-QUADRO
  6=DISTR. T DI STUDENT

2.DEFINIRE I PARAMETRI DELLA DISTRIBU-
  ZIONE SCELTA

3.INSERIRE IL VALORE DELLA VARIABILE
  ALEATORIA PER CUI DEVE ESSERE
  EFFETTUATO IL CALCOLO
PREMERE UN TASTO PER PROSEGUIRE

```

```

SCELTA DELLA DISTRIBUZIONE      4
DISTRIBUZIONE NORMALE A 2 VARIABILI X,Y:
  MX=MEDIA PER X
  MY=MEDIA PER Y
  SX=SCARTO QUADRATICO DI X
  SY=SCARTO QUADRATICO DI Y
  RO=COEFFICIENTE DI CORRELAZIONE

```

```

MX=1.1
SX=1.4
MY=2.4
SY=.9
RO=.75

```

```

X=1
Y=2

```

```

PROB(VAR.ALEATORIE =X,Y)=.159963727

```

```

ALTRO CALCOLO? (S O N): N

```

```

]

```

```

RUN

```

```

  CALCOLO DELLE DISTRIBUZIONI
  STATISTICHE

```

```

  AUTORE:H.HAUT

```

```

1.SCEGLIERE LA DISTRIBUZIONE:

```

```

  1=DISTR. BINOMIALE
  2=DISTR. DI POISSON
  3=DISTR. NORMALE
  4=DISTR. NORMALE A 2 VARIABILI
  5=DISTR. CHI-QUADRO
  6=DISTR. T DI STUDENT

```

```

2.DEFINIRE I PARAMETRI DELLA DISTRIBU-
  ZIONE SCELTA

```

```

3.INSERIRE IL VALORE DELLA VARIABILE
  ALEATORIA PER CUI DEVE ESSERE
  EFFETTUATO IL CALCOLO
PREMERE UN TASTO PER PROSEGUIRE
SCELTA DELLA DISTRIBUZIONE      5
DISTRIBUZIONE CHI-QUADRO :
  NU=NUMERO DI GRADI DI LIBERTA'

```

```

NU=3

```

```

X=1

```

```

PROB(VAR.ALEATORIA =X)=.241970861
PROB(VAR.ALEATORIA<=X)=.198748155
PROB(VAR.ALEATORIA>=X)=.801251845

```

```

ALTRO CALCOLO? (S O N): N

```

```

]

```



```

RUN
  CALCOLO DELLE DISTRIBUZIONI
  STATISTICHE

  AUTORE:H.HAUT

```

```

1.SCEGLIERE LA DISTRIBUZIONE:
  1=DISTR. BINOMIALE
  2=DISTR. DI POISSON
  3=DISTR. NORMALE
  4=DISTR. NORMALE A 2 VARIABILI
  5=DISTR. CHI-QUADRO
  6=DISTR. T DI STUDENT

2.DEFINIRE I PARAMETRI DELLA DISTRIBU-
  ZIONE SCELTA

3.INSERIRE IL VALORE DELLA VARIABILE
  ALFATORIA PER CUI DEVE ESSERE
  EFFETTUATO IL CALCOLO
  PREMERE UN TASTO PER PROSEGUIRE
  SCELTA DELLA DISTRIBUZIONE      6
  DISTRIBUZIONE T DI STUDENT:
  NU=NUMERO DI GRADI DI LIBERTA'

  NU=19

  X=.257

  PROR(-X<=VAR.AL FATORIA<=X)=.200058704
  PROR(VAR.AL FATORIA<=X)=.400029352
  PROR(VAR.AL FATORIA>=X)=.399970648

  ALTRO CALCOLO? (S O N): N

```

tempo d'esecuzione: 1"

memoria richiesta: 8446 bytes (senza REM : 3108).



## PROGRAMMA NUMERO 14

# REGRESSIONE AD UN PARAMETRO

Le regressioni ad un parametro, presentate in questo capitolo, permettono di adattare dei dati sperimentali ad una retta o una curva polinomiale di ordine superiore o a una curva di tipo esponenziale o ad una curva di tipo logaritmico.

## 1 \_ METODI NUMERICI

### 1.1 — Regressione lineare semplice

Siano date due variabili  $x$  e  $y$  legate tra loro da una legge lineare del tipo:  
 $y = \alpha + \beta x$

Il problema della regressione consiste nello stimare i valori da attribuire ai parametri  $\alpha$  e  $\beta$  a partire da un insieme di  $n$  dati sperimentali  $x_i, y_i$  ( $i = 1, 2, \dots, n$ ).

Supponiamo ora che siano verificate le seguenti ipotesi supplementari:

- a) i valori  $x_i$  sono esenti da errori sperimentali di misurazione
- b) i valori  $y_i$  sono soggetti a degli errori di misura  $\varepsilon_i$  statisticamente indipendenti e aventi distribuzione normale con media 0 e varianza  $\sigma^2$ .

Abbiamo risolto il problema della regressione applicando il metodo dei minimi quadrati che consiste nel calcolare la somma degli scarti quadratici

$$Q = \sum_i (y_i - \alpha - \beta x_i)^2$$

e nello stimare i parametri  $\alpha$  e  $\beta$  con i valori  $a$  e  $b$  che rendono minimo il valore di  $Q$ .

Per ottenere ciò basta risolvere il sistema:

$$\begin{aligned} \frac{\delta Q}{\delta \alpha} &= \sum_i (y_i - a - bx_i) = 0 \\ \frac{\delta Q}{\delta \beta} &= \sum_i (y_i - a - bx_i) x_i = 0 \end{aligned}$$

che ha per soluzione:

$$\begin{aligned} b &= \frac{\sum (x_i - \bar{x}) y_i}{\sum (x_i - \bar{x})^2} \\ (1) \quad a &= \bar{y} - b\bar{x} \\ \text{ove } \bar{x} &= 1/n \sum x_i \quad \text{e} \quad \bar{y} = 1/n \sum y_i \end{aligned}$$

Sostituendo nella (1) l'espressione  $y_i = \alpha + \beta x_i + \varepsilon_i$  si giunge ad esprimere a e b in funzione di  $\alpha$  e  $\beta$ .

$$\begin{aligned} b &= \beta + \frac{\sum \varepsilon_i (x_i - \bar{x})}{\sum (x_i - \bar{x})^2} \\ a &= \alpha + \frac{\sum \varepsilon_i}{n} - \frac{\bar{x} \sum \varepsilon_i (x_i - \bar{x})}{\sum (x_i - \bar{x})^2} \end{aligned}$$

Queste espressioni permettono di calcolare i valori "aspettati" per a e b;  $E[a]$  ed  $E[b]$  (valori medi di a e b).

Poiché  $E[a] = \alpha$  e  $E[b] = \beta$  si può dire che a e b sono dei buoni estimatori per i parametri  $\alpha$  e  $\beta$ .

Il calcolo della varianza e della covarianza per a e b dà poi i seguenti risultati:

$$\begin{aligned} \sigma_a^2 &= E[(a - \alpha)^2] = \sigma^2 \left( 1/n + \frac{\bar{x}^2}{\sum (x_i - \bar{x})^2} \right) \\ (2) \quad \sigma_b^2 &= E[(b - \beta)^2] = \frac{\sigma^2}{\sum (x_i - \bar{x})^2} \\ \sigma_{a,b} &= E[(a - \alpha)(b - \beta)] = \frac{-\bar{x} \sigma^2}{\sum (x_i - \bar{x})^2} \end{aligned}$$

Si può dimostrare infine che  $s^2 = \Sigma(y_i - \hat{y}_i)^2 / (n - 2)$  è un buon estimatore di  $\sigma^2$  ( $E[s^2] = \sigma^2$ ) e tramite esso calcolare i valori delle (2).

Il programma realizzato calcola, a partire dai dati sperimentali i valori di  $a, b, \sigma^2, \sigma_a^2, \sigma_b^2, \sigma_{ab}$  e permette di effettuare un'interpolazione per  $x$  o  $y$  dati.

## 1.2 — Adattamento ad altre curve ad un parametro

L'adattamento a una curva del tipo  $y = ax^b$  si riduce ad una regressione lineare passando ai logaritmi:

$$\ln y = \ln a + b \cdot \ln x$$

Applicando allora il metodo illustrato in precedenza all'insieme di dati  $(\ln x_i, \ln y_i)$  si possono stimare i valori di  $\ln a$  e di  $b$ .


Nello stesso modo si può operare nel caso di una funzione esponenziale del tipo  $y = a e^{bx}$ , applicando la trasformazione:


$$\ln y = \ln a + bx$$


Nel caso di una funzione di tipo logaritmico del tipo  $y = a + b \ln x$  non si rendono necessarie ulteriori trasformazioni essendo la stessa già lineare nei parametri  $a$  e  $b$ .

*Riferimenti bibliografici:* B2, C5, G1, H2, M1.


## 2 — DIAGRAMMA A BLOCCHI

















### 3 — PROGRAMMA

```

1  REM          REGRESSIONI AD UN PARAMETRO
3  REM
4  REM          AUTORE:H.HAUT
5  REM
6  REM
7  REM          DESCRIZIONE:
8  REM          IL PROGRAMMA PERMETTE DI ADATTARE UN INSIEME
9  REM          DI DATI(X(I),Y(I)) A UNA RETTA, A UNA CURVA
10 REM          DEL TIPO A*X^B, A UNA CURVA ESPONENZIALE
11 REM          O A UNA CURVA LOGARITMICA
12 REM
13 REM          *****
14 REM
15 REM
100 REM
110 REM          REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 5);"REGRESSIONI AD UN PARAMETRO"
160 PRINT : PRINT TAB( 20);"AUTORE:H.HAUT"
170 PRINT : PRINT
180 PRINT "1.DEFINIRE IL TIPO DI CURVA A CUI SI          VOGLIONO ADATTARE I
    DATI:                      1=RETTA                      2=FUNZI
    ONE DEL TIPO A*X^B                      3=CURVA ESPONENZIALE
    4=CURVA LOGARITMICA"
190 PRINT
200 PRINT "2.INTRODURRE IL NUMERO DI DATI SPERIMEN- TALI"
210 PRINT
220 PRINT "3.INTRODURRE LE COPPIE (X(I),Y(I)) L'UNA DOPO L'ALTRA"
230 VTAB (22): PRINT "PREMERE UN TASTO PER PROSEGUIRE ": GET A$
240 HOME
250 REM
260 REM          GESTIONE DEI DATI IN INPUT
270 REM
280 DIM M$(4),VE$(4),VA$(4)
290 FOR I = 1 TO 4: READ M$(I),VE$(I),VA$(I): NEXT I: REM          PREPARAZIONE
    PER L'IMPAGINAZIONE
300 DATA "Y=A + B*X", " Y ", " A ", "Y=A * (X^B)", "LN(Y)", "LN(A)", "Y=
    A* EXP(B*X)", "LN(Y)", "LN(A)", "Y=A + B*LN(X)", " Y ", " A "
310 PRINT "TIPO DI REGRESSIONE:"
320 FOR I = 1 TO 4: PRINT TAB( 20);I;"=";M$(I): NEXT I
330 INPUT "VOSTRA SCELTA (1,2,3,4 ):";CH
340 PRINT : PRINT
350 INPUT "NUMERO DI PUNTI=";N
360 REM
370 REM          ISTRUZIONI DI DIMENSIONAMENTO
380 REM
400 REM
410 DIM X(N),Y(N)
415 REM          INSERIMENTO DATI
420 FOR I = 1 TO N
430 PRINT
440 PRINT I
450 CV = PEEK (37): VTAB (CV): HTAB (10): INPUT "X=";X(I)
460 HTAB (10): INPUT "Y=";Y(I)
470 NEXT I
480 REM
490 REM          CHIAMATA DEL SOTTOPROGRAMMA RICHIESTO
500 REM
510 ON CH GOSUB 1000,2000,3000,4000
520 REM
530 REM          GESTIONE DEI RISULTATI
540 REM
550 HOME

```

```

560 PRINT "MODELLO: ";M0$(CH)
570 PRINT
580 PRINT "STIMA DI A      =" ;A
590 PRINT "STIMA DI B      =" ;B
600 PRINT
610 PRINT "VARIANZA DI ";VF$(CH);"=" ;SS
620 PRINT
630 PRINT "VARIANZA DI ";VA$(CH);"=" ;SA
640 PRINT "VARIANZA DI   B =" ;SB
650 PRINT : PRINT "COVARIANZA DI ";VA$(CH);",B =" ;AB
660 PRINT : PRINT "=====
670 REM
680 REM  POSSIBILITA' DI INTERPOLAZIONE
690 REM
700 PRINT : INPUT "CALCOLO DI INTERPOLAZIONE (S O N) :";A$
710 IF A$ = "N" THEN END
720 INPUT "INTERPOLAZIONE PER X O Y?:"A$
730 IF A$ = "Y" THEN GOTO 820
740 PRINT : HTAB (5): INPUT "X=";X
750 ON CH GOTO 760,770,780,790: REM          INTERPOLAZIONE IN Y
760 Y = A + B * X: GOTO 800
770 Y = A * (X ^ B): GOTO 800
780 Y = A * EXP (B * X): GOTO 800
790 Y = A + B * LOG (X)
800 PRINT TAB( 5);"Y=";Y
810 GOTO 700
820 PRINT : HTAB (5): INPUT "Y=";Y
830 ON CH GOTO 840,850,860,870: REM  INTERPOLAZIONE IN X
840 X = (Y - A) / B: GOTO 880
850 X = (Y / A) ^ (1 / B): GOTO 880
860 X = LOG (Y / A) / B: GOTO 880
870 X = EXP ((Y - A) / B)
880 PRINT TAB( 5);"X=";X
890 GOTO 700
1000 REM *****
1010 REM  SOTTOPROGRAMMA PER LA REGRESSIONE LINEARE
1020 REM
1030 REM  REGOLE PER L'USO DEL SOTTOPROGRAMMA
1040 REM
1050 REM      1.DATI NECESSARI:
1060 REM      * N=NUMERO DI PUNTI
1070 REM      * X(I),Y(I) (I=1,...,N) =COORDINATE DI TALI PUNTI
1090 REM      2.RISULTATI FORNITI:
1100 REM      * A,B = PARAMETRI DELLA RETTA DI
1110 REM      REGRESSIONE  $Y = A + B \cdot X$ 
1120 REM      * SS=VARIANZA DEGLI Y(I)
1130 REM      * SA=VARIANZA DI A
1140 REM      * SB=VARIANZA DI B
1150 REM      * AB=COVARIANZA DI A,B
1160 REM      3.VARIABILI UTILIZZATE:
1170 REM      A,AB,B,I,N,SA,SB,SS,X2,XM,XY,YM
1180 REM      4.VETTORI UTILIZZATI:
1190 REM      X(N),Y(N)
1200 REM *****
1210 REM
1220 REM  INIZIALIZZAZIONE
1230 REM
1240 XM = 0:YM = 0
1250 XY = 0:X2 = 0
1260 SS = 0
1270 REM
1280 REM  CALCOLO DI XM E YM (VALORI MEDI DI X E Y)
1290 REM
1300 FOR I = 1 TO N
1310 XM = XM + X(I)
1320 YM = YM + Y(I)
1330 NEXT I

```

```

1340 XM = XM / N
1350 YM = YM / N
1360 REM
1370 REM   CALCOLO DI X2=SOMMA DEGLI SCARTI QUADRATICI E XY=SOMMA DEGLI
      (X(I)-XM)*Y(I)
1380 REM
1390 FOR I = 1 TO N
1400 X2 = X2 + (X(I) - XM) ^ 2
1410 XY = XY + (X(I) - XM) * Y(I)
1420 NEXT I
1430 REM
1440 REM   CALCOLO DEI PARAMETRI A E B
1450 REM
1460 B = XY / X2
1470 A = YM - B * XM
1480 REM
1490 REM   CALCOLO DELLA VARIANZA DI Y(I)
1500 REM
1510 FOR I = 1 TO N
1520 SS = SS + (Y(I) - A - B * X(I)) ^ 2
1530 NEXT I
1540 SS = SS / (N - 2)
1550 REM
1560 REM   CALCOLO DELLE ALTRE VARIANZE
1570 REM
1580 SA = SS * ((1 / N) + (XM * XM / X2))
1590 SB = SS / X2
1600 AB = - SS * XM / X2
1610 RETURN
2000 REM *****
2010 REM   SOTTOPROGRAMMA PER L'ADATTAMENTO AD UNA CURVA DI POTENZA
2020 REM
2030 REM   REGOLE PER L'USO DEL SOTTOPROGRAMMA
2040 REM
2050 REM   1.DATI NECESSARI:
2060 REM       * N=NUMERO DI PUNTI
2070 REM       * X(I),Y(I) (I=1,...,N) =COORDINATE DI TALI PUNTI
2080 REM   2.RISULTATI FORNITI:
2090 REM       * A,B = PARAMETRI DELLA CURVA DI REGRESSIONE :
2100 REM           Y = A*(X^B)
2110 REM       * SS=VARIANZA DI LN(Y(I))
2120 REM       * SA=VARIANZA DI LN(A)
2130 REM       * SB=VARIANZA DI B
2140 REM       * AB=COVARIANZA DI LN(A),B
2150 REM   3.VARIABILI UTILIZZATE:
2160 REM       A,I,N
2170 REM   4.VETTORI UTILIZZATI:
2180 REM       X(N),Y(N)
2190 REM   5.SOTTOPROGRAMMA CHIAMATO:
2200 REM       1000 : REGRESSIONE LINEARE
2210 REM *****
2220 REM
2230 REM   CAMBIAMENTO DI VARIABILI
2240 REM
2250 FOR I = 1 TO N
2260 X(I) = LOG (X(I))
2270 Y(I) = LOG (Y(I))
2280 NEXT I
2290 REM
2300 REM   CHIAMATA DEL SOTTOPROGRAMMA PER LA REGRESSIONE LINEARE
2310 REM
2320 GOSUB 1000
2330 A = EXP (A): REM   TRASFORMAZIONE INVERSA
2340 RETURN
3000 REM *****
3010 REM   SOTTOPROGRAMMA D'ADATTAMENTO AD UNA CURVA ESPONENZIALE
3020 REM

```

```

3030 REM      REGOLE PER L'USO DEL SOTTOPROGRAMMA
3040 REM
3050 REM      1.DATI NECESSARI:
3060 REM          * N=NUMERO DI PUNTI
3070 REM          * X(I),Y(I) (I=1,J..,N) =COORDINATE DI TALI PUNTI
3080 REM      2.RISULTATI FORNITI:
3090 REM          * A,B = PARAMETRI DELLA CURVA DI REGRESSIONE:
3100 REM              Y = A*EXP(B*X)
3110 REM          * SS=VARIANZA DEI LN(Y(I))
3120 REM          * SA=VARIANZA DI LN(A)
3130 REM          * SB=VARIANZA DI B
3140 REM          * AB=COVARIANZA DI LN(A),B
3150 REM      3.VARIABILI UTILIZZATE:
3160 REM          A,I,N
3170 REM      4.VETTORI UTILIZZATI:
3180 REM          X(N),Y(N)
3190 REM      5.SOTTOPROGRAMMA UTILIZZATO:
3200 REM          1000 : REGRESSIONE LINEARE
3210 REM *****
3220 REM
3230 REM      CAMBIAMENTO DI VARIABILE
3240 REM
3250 FOR I = 1 TO N
3260 Y(I) = LOG (Y(I))
3270 NEXT I
3280 REM
3290 REM      CHIAMATA DEL SOTTOPROGRAMMA PER LA REGRESSIONE LINEARE
3300 REM
3310 GOSUB 1000
3320 A = EXP (A): REM      TRASFORMAZIONE INVERSA
3330 RETURN
4000 REM *****
4010 REM      SOTTOPROGRAMMA PER L'ADATTAMENTO AD UNA CURVA LOGARITMICA
4020 REM
4030 REM      REGOLE PER L'USO DEL SOTTOPROGRAMMA
4040 REM
4050 REM      1.DATI NECESSARI:
4060 REM          * N=NUMERO DI PUNTI
4070 REM          * X(I),Y(I) (I=1,...,N) =COORDINATE DI TALI PUNTI
4080 REM      2.RISULTATI FORNITI:
4090 REM          * A,B = PARAMETRI DELLA CURVA DI REGRESSIONE:
4100 REM              Y = A+B*LN(X)
4110 REM          * SS=VARIANZA DI Y(I)
4120 REM          * SA=VARIANZA DI A
4130 REM          * SB=VARIANZA DI B
4140 REM          * AB=COVARIANZA DI A,B
4150 REM      3.VARIABILI UTILIZZATE:
4160 REM          I,N
4170 REM      4.VETTORI UTILIZZATI:
4180 REM          X(N),Y(N)
4190 REM      5.SUBROUTINE CHIAMATA:
4200 REM          1000 : REGRESSIONE LINEARE
4210 REM *****
4220 REM
4230 REM      CAMBIAMENTO DI VARIABILI
4240 REM
4250 FOR I = 1 TO N
4260 X(I) = LOG (X(I)).
4270 NEXT I
4280 REM
4290 REM      CHIAMATA DEL SOTTOPROGRAMMA PER LA REGRESSIONE LINEARE
4300 REM
4310 GOSUB 1000
4320 RETURN

```

## 4 — ESEMPIO PRATICO

Vengono presi in considerazione l'adattamento ad una retta e ad una funzione del tipo  $y = ax^b$ . Per ciascuno dei due casi sono state scelte 5 coppie di valori sperimentali e precisamente:

|                     |       |      |      |       |       |       |
|---------------------|-------|------|------|-------|-------|-------|
| a) per $y = a + bx$ | $x =$ | 1    | 2    | 3     | 4     | 5     |
|                     | $y =$ | 4.95 | 8.03 | 10.78 | 14.06 | 16.39 |

|                   |       |      |       |        |        |        |
|-------------------|-------|------|-------|--------|--------|--------|
| b) per $y = ax^b$ | $x =$ | 1    | 2     | 3      | 4      | 5      |
|                   | $y =$ | 4.97 | 39.81 | 136.01 | 318.76 | 627.20 |

```
RUN
REGRESSIONI AD UN PARAMETRO

AUTORE:H.HAUT
```

```
1.DEFINIRE IL TIPO DI CURVA A CUI SI
VOGLIONO ADATTARE I DATI:
1=RETTA
2=FUNZIONE DEL TIPO A**X^B
3=CURVA ESPONENZIALE
4=CURVA LOGARITMICA

2.INTRODURRE IL NUMERO DI DATI SPERIMEN-
TALI

3.INTRODURRE LE COPPIE (X(I),Y(I)) L'UNA
DOPO L'ALTRA
PREMERE UN TASTO PER PROSEGUIRE
TIPO DI REGRESSIONE:
1=Y=A + B*X
2=Y=A * (X^B)
3=Y=A* EXP(B*X)
4=Y=A + B*LN(X)
VOSTRA SCELTA (1,2,3,4 ):1
```

NUMERO DI PUNTI= 5

```
1
X=1
Y=4.95

2
X=2
Y=8.03

3
X=3
Y=10.78

4
X=4
Y=14.06

5
X=5
Y=16.39
MODELLO: Y=A + B*X
```



```

STIMA DI A      =2.169
STIMA DI B      =2.891

VARIANZA DI Y   =.0612900006

VARIANZA DI A   =.0674190006
VARIANZA DI B   =6.12900006E-03

COVARIANZA DI A ,B =-.0183870002

```

```
=====
```

```

CALCOLO D'INTERPOLAZIONE (S O N) :S
INTERPOLAZIONE PER X O Y?:Y

```

```

Y=10
X=2.7087513

```

```

CALCOLO D'INTERPOLAZIONE (S O N) :N

```

```

RUN
  REGRESSIONI AD UN PARAMETRO

      AUTORE:H.HAUT

```

```

1.DEFINIRE IL TIPO DI CURVA A CUI SI
VOGLIONO ADATTARE I DATI:

```

```

1=RETTA
2=FUNZIONE DEL TIPO A**X^B
3=CURVA ESPONENZIALE
4=CURVA LOGARITMICA

```

```

2.INTRODURRE IL NUMERO DI DATI SPERIMENTALI

```

```

3.INTRODURRE LE COPPIE (X(I),Y(I)) L'UNA
DOPO L'ALTRA
PREMERE UN TASTO PER PROSEGUIRE
TIPO DI REGRESSIONE:

```

```

1=Y=A + B**X
2=Y=A * (X^B)
3=Y=A* EXP(B**X)
4=Y=A + B*LN(X)

```

```

VOSTRA SCELTA (1,2,3,4 ):2

```

```

NUMERO DI PUNTI= 5

```

```

1
      X=1
      Y=4.97

2
      X=2
      Y=39.81

3
      X=3
      Y=136.01

4
      X=4
      Y=318.76

```

```

5
      X=5
      Y=427.20
MODELLO: Y=A * (X^B)

STIMA DI A      =4.97061355
STIMA DI B      =3.00538459

VARIANZA DI LN(Y)=3.07652912E-05
VARIANZA DI LN(A)=2.34126102E-05
VARIANZA DI B    =1.90439498E-05
COVARIANZA DI LN(A),B =-1.82345505E-05
=====
CALCOLO DI INTERPOLAZIONE (S O N) :S
INTERPOLAZIONE PER X O Y?:X
      X=6
      Y=1084.06114
CALCOLO DI INTERPOLAZIONE (S O N) :N

```

tempo d'esecuzione: 2"

memoria richiesta: 7063 bytes (senza REM : 2111)

## PROGRAMMA NUMERO 15

# REGRESSIONE LINEARE MULTIPLA

Questo programma sulla regressione lineare multipla permette di adattare dei dati sperimentali ad un modello teorico funzione lineare in diversi parametri, che dovranno essere opportunamente stimati. Il programma fornisce i valori relativi agli estimatori di questi parametri, i valori delle loro varianze e delle loro covarianze ed il valore della varianza della variabile dipendente.

### 1 — METODO NUMERICO

Il problema della regressione lineare multipla consiste nello stimare gli  $r$  parametri  $\beta_1, \beta_2, \dots, \beta_r$  di un modello lineare del tipo:

$$y = \beta_1 x_{i1} + \beta_2 x_{i2} + \dots + \beta_r x_{ir}$$

sulla base di  $n$  insiemi di dati  $(x_{i1}, x_{i2}, \dots, x_{ir}, y_i)$  per  $i = 1, 2, \dots, n$ .

Come nel caso della regressione lineare semplice si supporrà che i valori  $x_{ik}$  rappresentino delle misure esenti da errori sperimentali e che le  $y_i$  siano invece soggette ad un errore aleatorio  $\varepsilon_i$ , di media 0 e varianza  $\sigma^2$ .

La soluzione si basa sul metodo dei minimi quadrati ed in particolare gli estimatori fedeli dei parametri  $\beta_1, \dots, \beta_r$  sono i valori che rendono minimo lo scarto quadratico:

$$Q = \sum_i (y_i - \beta_1 x_{i1} - \dots - \beta_r x_{ir})^2$$

Il valore degli estimatori si ottiene quindi risolvendo il sistema di  $r$  equazioni in  $r$  incognite:

$$(1) \quad \frac{\delta Q}{\delta \beta_k} = 0 \quad (k = 1, 2, \dots, r)$$

Introduciamo ora le matrici:

$$X_k = [x_{1k}, x_{2k}, \dots, x_{nk}]^T \quad (k = 1, \dots, r)$$

$$Y = [y_1, y_2, \dots, y_n]^T$$

$$B = [b_1, b_2, \dots, b_r]^T$$

$$\beta = [\beta_1, \beta_2, \dots, \beta_r]^T$$

$$X = \text{matrice } (n, r) \text{ le cui colonne sono } X_1, X_2, \dots, X_r$$

ove T denota la trasposta della matrice.

Con queste notazioni, il sistema (1) può essere scritto nella forma matriciale  $(X^T X) B = X^T Y$  e la soluzione (vale a dire la matrice colonna B degli estimatori dei  $\beta_i$ ), è data da:

$$(2) \quad B = (X^T X)^{-1} X^T Y$$

Si può allora calcolare la matrice delle covarianze  $\Sigma_b$ :

$$\Sigma_b = E [(B - \beta) (B - \beta)^T] = \sigma^2 (X^T X)^{-1}$$

in cui un estimatore fedele di  $\sigma^2$  è dato da:

$$s^2 = \frac{(\text{somma residua dei quadrati})}{n - r}$$

ove la somma residua dei quadrati è data da

$$\sum_i (y_i - \hat{y}_i)^2 \text{ con } \hat{y}_i = b_1 x_{i1} + b_2 x_{i2} + \dots + b_r x_{ir}$$

*Riferimenti bibliografici:* C5, G1, H2.

## 2 — TECNICA DI PROGRAMMAZIONE

Dopo aver definito le matrici X e Y nel modo seguente:

$$X = \begin{pmatrix} x_{11} & x_{12} & \dots & x_{1r} \\ x_{21} & x_{22} & \dots & x_{2r} \\ . & . & . & . \\ . & . & . & . \\ x_{n1} & x_{n2} & \dots & x_{nr} \end{pmatrix} \quad Y = \begin{pmatrix} y_1 \\ y_2 \\ . \\ . \\ y_n \end{pmatrix}$$

si calcola il valore di  $X^T X$  e lo si memorizza in una matrice AA(r,r) mentre in una matrice BB si memorizza  $X^T Y$ .

L'uso del sottoprogramma per l'inversione di una matrice permette di risolvere l'equazione (2) fornendo la soluzione nella matrice BB e  $(X^TX)^{-1}$  in AA. Quest'ultima, moltiplicata per  $\sigma^2$  fornirà la matrice delle covarianze  $\Sigma_b$ .

Per calcolare la somma residua dei quadrati, si calcola dapprima  $\hat{y}_i = \sum_j X_{ij} B_j$ , poi  $\sum_i (y_i - \hat{y}_i)^2$  ed infine l'estimatore fedele di  $\sigma^2$ .

*Nota:*


Il programma può essere utilizzato per un qualsiasi modello lineare nei parametri da stimare. Il modello generale di tale funzione è dato da:


$$y_i = \beta_1 f_1(x_{i1}, \dots) + \beta_2 f_2(x_{i1}, \dots) + \dots + \beta_r f_r(x_{i1}, \dots)$$


in cui le  $f_i$  sono funzioni arbitrarie delle variabili  $x_{i1}, x_{i2}, \dots, x_{is}$

$$X = \begin{pmatrix} f_1(x_{11}, \dots) & f_2(x_{11}, \dots) & \dots & f_r(x_{11}, \dots) \\ f_1(x_{21}, \dots) & f_2(x_{21}, \dots) & \dots & f_r(x_{21}, \dots) \\ \cdot & \cdot & & \cdot \\ \cdot & \cdot & & \cdot \\ f_1(x_{n1}, \dots) & f_2(x_{n1}, \dots) & \dots & f_r(x_{n1}, \dots) \end{pmatrix}$$


### 3 — DIAGRAMMA A BLOCCHI














## 4 — PROGRAMMA

```

1  REM      PROGRAMMA DI REGRESSIONE
2  REM      LINEARE MULTIPLA
3  REM
4  REM      AUTORE:H.HAUT
5  REM
6  REM      DESCRIZIONE:
7  REM      IL PROGRAMMA PERMETTE DI ADATTARE UN
8  REM      INSIEME DI DATI(X(1),...,X(NP),Y) AL MODELLO
9  REM       $Y=B(1)*X(1)+...+B(NR)*X(NP)$ 
10 REM
11 REM
100 REM
110 REM      REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 8);"PROGRAMMA DI REGRESSIONE"
150 PRINT TAB( 11);"LINEARE MULTIPLA"
160 PRINT
170 PRINT TAB( 20);"AUTORE:H.HAUT"
180 PRINT : PRINT
190 PRINT "1.DEFINIRE IL NUMERO DI GRUPPI DI DATI      N E DI PARAMETRI NP
    ": PRINT
200 PRINT "2.INSERIRE POI GLI ND GRUPPI DI DATI      SECONDO IL SEGUENTE
    MODELLO:"
210 PRINT
220 PRINT "      Y(I)=B(1)*X(I,1) + B(2)*X(I,2) + ..      ..+B(NP)*X(I,
    NP)  I=
    1,2,...,ND"
230 PRINT
240 PRINT "3.SE B(1) RAPPRESENTA UN TERMINE COSTAN- TE, TUTTI GLI X(I,1)
    DEVONO ESSERE      UGUALI A 1)
250 VTAB (22): PRINT "PREMERE UN TASTO PER PROSEGUIRE": GET A$
260 REM
270 REM      GESTIONE DEI DATI IN INPUT
280 REM
290 HOME
300 PRINT "MODELLO: "
310 PRINT "      Y(I)=B(1)*X(I,1) + B(2)*X(I,2) + ..      ..+B(NP)*X(I,
    NP)  I=
    1,2,...,ND"
320 PRINT
330 INPUT "NUMERO DI PARAMETRI: NP=":NP
340 INPUT "NUMERO DI DATI:      ND=":ND
350 REM
360 REM      ISTRUZIONI DI DIMENSIONAMENTO
370 REM
380 DIM X(ND,NP),Y(ND),AA(NP - 1,NP - 1),BB(NP - 1,0)
390 REM
400 REM      INSERIMENTO DATI
410 REM
420 FOR I = 1 TO ND
430 PRINT : PRINT "GRUPPO ";I;" DI DATI"
440 FOR J = 1 TO NP: HTAB (10)
450 PRINT "X(";I;";";J;")":CV = PEEK (37): VTAB (CV): HTAB (20): INPUT "
    =" :X(I,J)
460 NEXT J
470 HTAB (10): PRINT "Y(";I;")":CV = PEEK (37): VTAB (CV): HTAB (20): INPUT
    " =" :Y(I)
480 NEXT I
490 REM
500 REM      CHIAMATA DEL SOTTOPROGRAMMA
510 REM
520 GOSUB 1000
530 REM

```

```

540 REM    GESTIONE DEI RISULTATI
550 REM
560 HOME : CALL - 198
570 PRINT "SOMMA RESIDUA DEI QUADRATI= ";SR
580 PRINT "VARIANZA DEGLI X(I)          = ";SS
590 PRINT
600 PRINT "STIMA DEI "; TAB( 25);"DEVIAZIONI"
610 PRINT "PARAMETRI"; TAB( 25);"STANDARD"
620 PRINT "*****"; TAB( 25);"*****"
630 FOR I = 1 TO NP
640 I1 = I - 1
650 SD = SQR (AA(I1,I1))
660 PRINT "R(";I;")=";BB(I1,0); TAB( 25);SD
670 NEXT I
680 REM
690 REM    STAMPA EVENTUALE DELLA MATRICE DELLE COVARIANZE
700 REM
710 PRINT : PRINT "SE NON SI VUOLE LA MATRICE DELLE          COVARIANZE, PR
EMERE ESC ALTRIMENTI UN ALTRO TASTO": GET A$
720 IF ASC (A$) = 27 THEN GOTO 870
730 FOR I = 1 TO NP
740 HOME
750 PRINT "COLONNA ";I;" DELLA MATRICE DI COVARIANZA;"
760 I1 = I - 1
770 FOR J = 1 TO NP
780 J1 = J - 1
790 PRINT TAB( 5);AA(I1,J1)
800 NEXT J
810 PRINT "PREMERE UN TASTO PER CONTINUARE"
820 GET A$
830 NEXT I
840 REM
850 REM    SCELTA PER L'INTERPOLAZIONE
860 REM
870 PRINT : INPUT "INTERPOLAZIONE?(S O N): ";A$
880 IF A$ = "N" THEN END
890 YT = 0
900 FOR I = 1 TO NP
910 PRINT "X";I;"=";CV = PEEK (37); UTAB (CV); HTAB (6); INPUT " ";X
920 YT = YT + X * BB(I - 1,0); REM    CALCOLO DELL'INTERPOLAZIONE SECONDO IL
MODELLO DATO
930 NEXT I
940 PRINT : PRINT "Y= "; TAB( 7);YT; PRINT
950 GOTO 870
1000 REM *****
1010 REM    SOTTOPROGRAMMA DI REGRESSIONE LINEARE MULTIPLA
1020 REM    ESSO PERMETTE DI ADATTARE UN INSIEME DI DATI
1030 REM    SPERIMENTALI(I,1),..., (I,NP),Y(I) AL SEGUENTE MODELLO
1040 REM    AD NP PARAMETRI:
1050 REM    
$$Y(I)=B(1)*X(I,1) + B(2)*X(I,2) + \dots + B(NP)*X(I,NP)$$

1060 REM    I=1,2,...,ND
1070 REM
1080 REM    REGOLE PER L'USO DEL SOTTOPROGRAMMA
1090 REM    1.DATI NECESSARI:
1100 REM    * ND=NUMERO DI DATI
1110 REM    * NP=NUMERO DI PARAMETRI
1120 REM    * X(I,J)=VALORI DELLE VARIABILI INDIPENDENTI
1130 REM    DEFINITE NEL MODELLO
1140 REM    I=1,2,...,ND
1150 REM    J=1,2,...,NP
1160 REM    * Y(I)=VALORI DELLA VARIABILE DIPENDENTE
1170 REM    I=1,2,...,ND
1180 REM    2.RISULTATI FORNITI:
1190 REM    * BB(I-1,0)=STIMATORE DEL PARAMETRO B(I)
1200 REM    I=0,1,...,NP-1
1210 REM    * AA(I-1,J-1)=COVARIANZA DI B(I),B(J)
1220 REM    I=0,1,...,NP-1

```

```

1230 REM          J=0,1,...,NP-1
1240 REM          * SR=SOMMA RESIDUA DEI QUADRATI
1250 REM          * SS=VARIANZA DEGLI Y(I)
1260 REM          3.VARIABILI UTILIZZATE:
1270 REM          I,I1,J,J1,K,M,N,ND,NP,SR,SS,ST
1280 REM          4.VETTORI E MATRICI UTILIZZATI:
1290 REM          AA(NP-1,NP-1),BB(NP-1,0),X(ND,NP),Y(ND)
1300 REM          5.SOTTOPROGRAMMA CHIAMATO:
1310 REM          SUBROUTINE 2000 PER L'INVERSIONE DI UNA MATRICE
1320 REM          *****
1330 REM
1340 REM          INIZIALIZZAZIONI
1350 REM
1360 SR = 0:N = NP:M = 1
1370 FOR I = 0 TO NP - 1
1380 FOR J = 0 TO NP - 1
1390 AA(I,J) = 0
1400 NEXT J
1410 BB(I,0) = 0
1420 NEXT I
1430 REM
1440 REM          CALCOLO DELLA MATRICE X(T)X DA MEMORIZZARE IN AA
1450 REM          (SI USA X(T) PER INDICARE LA TRASPOSTA DI X)
1460 REM
1470 FOR I = 1 TO NP
1480 FOR J = I TO NP
1490 FOR K = 1 TO ND
1500 I1 = I - 1:J1 = J - 1
1510 AA(I1,J1) = AA(I1,J1) + X(K,I) * X(K,J)
1520 NEXT K
1530 AA(J1,I1) = AA(I1,J1)
1540 NEXT J
1550 NEXT I
1560 REM
1570 REM          CALCOLO DI X(T)Y DA MEMORIZZARE IN BB
1580 REM
1590 FOR I = 1 TO NP
1600 I1 = I - 1
1610 FOR K = 1 TO ND
1620 BB(I1,0) = BB(I1,0) + X(K,I) * Y(K)
1630 NEXT K
1640 NEXT I
1650 REM
1660 REM          CHIAMATA DELLA SUBROUTINE PER LA RISOLUZIONE DI UN SISTEMA
1670 REM
1680 GOSUB 2000
1690 REM
1700 REM          CALCOLO DEGLI Y(I) TEORICI DEL MODELLO
1720 REM
1730 FOR I = 1 TO ND
1740 ST = 0
1750 FOR J = 1 TO NP
1760 J1 = J - 1
1770 ST = ST + X(I,J) * BB(J1,0)
1780 NEXT J
1790 X(I,0) = ST
1800 NEXT I
1810 REM
1820 REM          CALCOLO DELL'ESTIMATORE DELLA VARIANZA DEGLI Y(I)
1830 REM
1840 FOR I = 1 TO ND
1850 SR = SR + ((Y(I) - X(I,0)) ^ 2)
1860 NEXT I
1870 REM          SR=SOMMA RESIDUA DEI QUADRATI
1880 SS = SR / (ND - NP)
1890 REM
1900 REM          CALCOLO DELLA MATRICE DELLE COVARIANZE

```

```

1910 REM    MEMORIZZATA IN AA(I,J)
1920 REM
1930 FOR I = 0 TO NP - 1
1940 FOR J = 0 TO NP - 1
1950 AA(I,J) = AA(I,J) * SS
1960 NEXT J
1970 NEXT I
1980 RETURN
2000 REM *****
2010 REM    SOTTOPROGRAMMA PER LA RISOLUZIONE
2020 REM    DEL SISTEMA MATRICIALE AA*X=BB
2030 REM    PER I DETTAGLI CFR PROGRAMMA SPECIFICO
2040 REM *****
2050 N1 = N - 1:M1 = M - 1:DET = 1
2060 FOR K = 0 TO N1
2070 PV = AA(K,K):IK = K:JK = K:PAV = ABS (PV)
2080 FOR I = K TO N1: FOR J = K TO N1
2090 IF ABS (AA(I,J)) > PAV THEN PV = AA(I,J):PAV = ABS (PV):IK = I:JK =
J
2100 NEXT J,I
2110 PC(K) = JK:PL(K) = IK
2120 IF IK < > K THEN DET = - DET
2130 IF JK < > K THEN DET = - DET
2140 DET = DET * PV
2150 IF DET = 0 THEN HOME : PRINT "IL DETERMINANTE E'NULLO,NESSUNA SOLUZ
IONE": END
2160 IF IK = K THEN GOTO 2200
2170 FOR I = 0 TO N1:TT = AA(IK,I):AA(IK,I) = AA(K,I):AA(K,I) = TT: NEXT
I
2180 IF M1 = - 1 THEN GOTO 2200
2190 FOR I = 0 TO M1:TT = BB(IK,I):BB(IK,I) = BB(K,I):BB(K,I) = TT: NEXT
I
2200 IF JK = K THEN GOTO 2220
2210 FOR I = 0 TO N1:TT = AA(I,JK):AA(I,JK) = AA(I,K):AA(I,K) = TT: NEXT
I
2220 FOR I = 0 TO N1:CS(I) = AA(I,K):AA(I,K) = 0: NEXT I
2230 CS(K) = 0:AA(K,K) = 1
2240 FOR I = 0 TO N1:AA(K,I) = AA(K,I) / PV: NEXT I
2250 IF M1 = - 1 THEN GOTO 2270
2260 FOR I = 0 TO M1:BB(K,I) = BB(K,I) / PV: NEXT I
2270 FOR J = 0 TO N1: IF J = K THEN GOTO 2310
2280 FOR I = 0 TO N1:AA(J,I) = AA(J,I) - CS(J) * AA(K,I): NEXT I
2290 IF M1 = - 1 THEN GOTO 2310
2300 FOR I = 0 TO M1:BB(J,I) = BB(J,I) - CS(J) * BB(K,I): NEXT I
2310 NEXT J
2320 NEXT K
2330 FOR I = N1 TO 0 STEP - 1:IK = PC(I)
2340 IF IK = I THEN GOTO 2380
2350 FOR J = 0 TO N1:TT = AA(I,J):AA(I,J) = AA(IK,J):AA(IK,J) = TT: NEXT
J
2360 IF M1 = - 1 THEN GOTO 2380
2370 FOR J = 0 TO M1:TT = BB(I,J):BB(I,J) = BB(IK,J):BB(IK,J) = TT: NEXT
J
2380 NEXT I
2390 FOR J = N1 TO 0 STEP - 1:JK = PL(J)
2400 IF JK = J THEN GOTO 2420
2410 FOR I = 0 TO N1:TT = AA(I,J):AA(I,J) = AA(I,JK):AA(I,JK) = TT: NEXT
I
2420 NEXT J
2430 RETURN

```

## 5 — ESEMPIO PRATICO

L'esempio presentato è relativo ad una regressione lineare a 4 parametri ( $NP = 4$ ) relativa a 6 gruppi di dati ( $ND = 6$ ):

|         |      |      |      |       |        |    |
|---------|------|------|------|-------|--------|----|
| $X_1 =$ | 1    | 1    | 1    | 1     | 1      | 1  |
| $X_2 =$ | 0    | 1    | 1    | 4     | 1      | 0  |
| $X_3 =$ | 0    | 0    | 1    | 3     | 3      | 1  |
| $X_4 =$ | 0    | 0    | 0    | 2     | 2      | 0  |
| $y =$   | 0.98 | 3.96 | 1.99 | -2.98 | -12.01 | -1 |

viene inoltre calcolata l'interpolazione per  $x_1 = 1$ ,  $x_2 = 1$ ,  $x_3 = 1$ ,  $x_4 = 1$ .

```
RUN
      PROGRAMMA DI REGRESSIONE
      LINEARE MULTIPLA
      AUTORE:H.HAUT

1.DEFINIRE IL NUMERO DI GRUPPI DI DATI
  N E DI PARAMETRI NP

2.INSERIRE POI GLI ND GRUPPI DI DATI
  SECONDO IL SEGUENTE MODELLO:

      Y(I)=B(1)*X(I,1) + B(2)*X(I,2) + ..
          ..+B(NP)*X(I,NP)

      I=1,2,...,ND

3.SE B(1) RAPPRESENTA UN TERMINE COSTAN-
  TE, TUTTI GLI X(I,1) DEVONO ESSERE
  UGUALI A 1)
  PREMERE UN TASTO PER PROSEGUIRE
  MODELLO:
      Y(I)=B(1)*X(I,1) + B(2)*X(I,2) + ..
          ..+B(NP)*X(I,NP)

      I=1,2,...,ND

NUMERO DI PARAMETRI: NP=4
NUMERO DI DATI:      ND=6

GRUPPO 1 DI DATI
      X(1,1)
          =1
      X(1,2)
          =0
      X(1,3)
          =0
      X(1,4)
          =0
      Y(1)
          =0.98
```

## GRUPPO 2 DI DATI

$X(2,1)$   
 $X(2,2)$   
 $X(2,3)$   
 $X(2,4)$   
 $Y(2)$

=1

=1

=0

=0

=3,96

## GRUPPO 3 DI DATI

$X(3,1)$   
 $X(3,2)$   
 $X(3,3)$   
 $X(3,4)$   
 $Y(3)$

=1

=1

=1

=0

=1,99

## GRUPPO 4 DI DATI

$X(4,1)$   
 $X(4,2)$   
 $X(4,3)$   
 $X(4,4)$   
 $Y(4)$

=1

=4

=3

=2

=-2,98

## GRUPPO 5 DI DATI

$X(5,1)$   
 $X(5,2)$   
 $X(5,3)$   
 $X(5,4)$   
 $Y(5)$

=1

=1

=3

=2

=-12,01

## GRUPPO 6 DI DATI

$X(6,1)$   
 $X(6,2)$   
 $X(6,3)$   
 $X(6,4)$   
 $Y(6)$

=1

=0

=1

=0

=-1

SOMMA RESIDUA DEI QUADRATI= 5,36363637E-04

VARIANZA DEGLI X(I) = 2,68181818E-04



| STRIMA DEI<br>PARAMETRI | DEVIAZIONI<br>STANDARD |
|-------------------------|------------------------|
| *****                   | *****                  |
| B(1)=.967272719         | .0120946679            |
| B(2)=3.00545455         | 6.98285977E-03         |
| B(3)=-1.97499998        | .0163762578            |
| B(4)=-5.02545457        | .0227613381            |

SE NON SI VUOLE LA MATRICE DELLE  
COVARIANZE, PREMERE ESC ALTRIMENTI UN AL  
TRO TASTO

COLONNA 1 DELLA MATRICE DI COVARIANZA:

1.46280992E-04  
-2.43801653E-05  
-1.34090909E-04  
1.58471075E-04

PREMERE UN TASTO PER CONTINUARE

COLONNA 2 DELLA MATRICE DI COVARIANZA:

-2.43801653E-05  
4.87603306E-05  
-3.50010768E-14  
-4.87603306E-05

PREMERE UN TASTO PER CONTINUARE

COLONNA 3 DELLA MATRICE DI COVARIANZA:

-1.34090909E-04  
4.06110055E-14  
2.68181819E-04  
-3.35227274E-04

PREMERE UN TASTO PER CONTINUARE

COLONNA 4 DELLA MATRICE DI COVARIANZA:

1.58471075E-04  
-4.87603307E-05  
-3.35227274E-04  
5.18078513E-04

PREMERE UN TASTO PER CONTINUARE

INTERPOLAZIONE?(S O N): S

X1=

1

X2=

1

X3=

1

X4=

1

Y= -3.02772728

INTERPOLAZIONE?(S O N): N

tempo d'esecuzione: 7''3

memoria richiesta: 6881 bytes (senza REM : 3385).



## PROGRAMMA NUMERO 16

# ADATTAMENTO AD UN POLINOMIO

### 1 — METODO NUMERICO

L'adattamento di dati sperimentali ad una funzione polinomiale può essere visto come un caso particolare di regressione lineare multipla applicata al modello teorico

$$y_i = \beta_0 + \beta_1 x_i + \beta_2 x_i^2 + \dots + \beta_d x_i^d$$

Il problema consiste pertanto nella ricerca degli estimatori  $b_0, b_1, \dots, b_d$  dei coefficienti  $\beta_0, \beta_1, \dots, \beta_d$  utilizzando le  $n$  coppie di dati sperimentali  $(x_i, y_i)$ .

Il metodo ricalca essenzialmente quello svolto nel capitolo precedente. Le matrici  $X$  e  $Y$  vengono definite nel modo seguente:

$$X = \begin{pmatrix} 1 & x_1 & x_1^2 & \dots & x_1^d \\ 1 & x_2 & x_2^2 & \dots & x_2^d \\ & & \cdot & & \cdot \\ & & & & \cdot \\ \cdot & \cdot & \cdot & & \cdot \\ 1 & x_n & x_n^2 & \dots & x_n^d \end{pmatrix} \quad Y = \begin{pmatrix} y_1 \\ y_2 \\ \cdot \\ \cdot \\ y_n \end{pmatrix}$$

Il numero dei parametri  $\beta_i$  risulta essere  $r = d + 1$  (ove  $d$  è il grado del polinomio).

Il programma fornisce gli estimatori dei parametri  $\beta_i$ , la loro deviazione standard, lo scarto quadratico, la varianza di  $y_i$ , la matrice  $\Sigma_b$  delle covarianze e permette di stimare il valore  $y$  relativo ad un qualsiasi valore  $x$ . Il programma infine fornisce, a richiesta, un grafico del polinomio interpolatore con una rappresentazione dei dati sperimentali.

*Riferimenti bibliografici:* C5, G1, H2, W1.


*Nota tecnica:*


Il programma corrispondente al listing presentato qui di seguito richiede la disponibilità di una memoria superiore a 16K per fornire una rappresentazione grafica del polinomio. Per far "girare" ugualmente il programma su un elaboratore con una memoria di 16K occorre eliminare buona parte delle linee di commento. In particolare si consiglia di sopprimere le seguenti linee presenti nel listing:


1—120, 190—280, 360—380, 450—470, 490—510, 610—630, 760—780, 820,  
900—920, 950—960, 1040—1070, 1120, 1220—1230, 1290—1310, 1330, 1410,  
1450, 1520, 2010—2270, 2290—2310, 2380—2400, 3010—3080, 4010—4050.

Si può inoltre notare che le due sub-routines relative alla regressione lineare multipla ed alla risoluzione di un sistema sono presentate in forma compatta; ciò permetterà al lettore, confrontandole con quelle presentate nei capitoli precedenti, di rendersi conto dei vantaggi ottenuti sia in termini di memoria utilizzata che di velocità di esecuzione.

2 — DIAGRAMMA  
A BLOCCHI







### 3 — PROGRAMMA

```

1  REM          ADATTAMENTO AD UN POLINOMIO
2  REM          DI GRADO DE.
3  REM
4  REM          AUTORE:H.HAUT
5  REM
6  REM
7  REM  DESCRIZIONE:
8  REM  IL PROGRAMMA PERMETTE DI ADATTARE UN INSIEME
9  REM  DI DATI SPERIMENTALI AD UN POLINOMIO DI GRADO
10 REM  DE:  $Y=B(0) + B(1)*X + B(2)*X^2 + \dots + B(DE)*X^{DE}$ 
11 REM
12 REM
100 REM
110 REM  REGOLE D'USO
120 REM
130 HOME
140 PRINT TAB( 8);"ADATTAMENTO A UN POLINOMIO"
150 PRINT TAB( 14);"DI GRADO DE "
160 PRINT
170 PRINT TAB( 20);"AUTORE:H.HAUT"
180 PRINT : PRINT
190 PRINT "1.DEFINIRE IL GRADO DEL POLINOMIO          E IL NUMERO DEI DATI
    SPERIMENTALI"
200 PRINT
210 PRINT "2.INTRODURRE POI LE ND COPPIE DI DATI      X(I),Y(I) DA ADATTAR
    E AL POLINOMIO:       $Y=B(0) + B(1)*X + \dots + B(DE)*X^{DE}$ "
220 PRINT
230 PRINT "3.IL PROGRAMMA FORNISCE GLI ESTIMATORI      DEI B(I),LA MATRICE
    DELLE COVARIANZE, DA' LA POSSIBILITA' DI EFFETTUARE UNA  INTERPOLA
    ZIONE E IL RELATIVO GRAFICO"
240 VTAB (22): PRINT "PREMERE UN TASTO PER PROSEGUIRE  ": GET A$
250 REM
260 REM  GESTIONE DEI DATI IN INPUT
270 REM
280 HOME
290 INPUT "GRADO DEL POLINOMIO= ";DE
300 PRINT
310 INPUT "NUMERO DI DATI= ";ND
320 REM
330 REM  ISTRUZIONI DI DIMENSIONAMENTO
340 REM
350 DIM X(ND,DE + 1),YD(ND),AA(DE,DE),BB(DE,0),Y(268),XD(ND)
360 REM
370 REM  INSERIMENTO DATI
380 REM
390 PRINT
400 FOR I = 1 TO ND
410 PRINT
420 PRINT I:CV = PEEK (37): VTAB (CV): HTAB (10): INPUT "X=";XD(I)
430 HTAB (10): INPUT "Y=";YD(I)
440 NEXT I
450 REM
460 REM  CHIAMATA DEL SOTTOPROGRAMMA
470 REM
480 GOSUB 2000
490 REM
500 REM  GESTIONE DEI RISULTATI
510 REM
520 HOME
530 PRINT "SOMMA RESIDUA DEI QUADRATI= ";SR
540 PRINT "VARIANZA D XI= ";SS
550 PRINT "ESTIMATORI DEF": TAB( 25);"DEVIAZIONI"
560 PRINT "  PARAMETRI "; TAB( 25);"STANDARD "
570 PRINT "*****"; TAB( 25);"*****"

```



```

580 FOR I = 0 TO DE
590 SD = SQR (AA(I,I))
600 PRINT "B(;"I;")=";BB(I,0); TAB( 25);SD
610 REM
620 REM      STAMPA A RICHIESTA DELLA MATRICE DI COVARIANZA
630 REM
640 NEXT I
650 PRINT : PRINT "SE SI VUOLE LA MATRICE DELLE COVARIANZE BATTERE UN TAS
    TO ALTRIMENTI ESC": GET A$
660 IF ASC (A$) = 27 THEN GOTO 790
670 FOR I = 0 TO DE
680 HOME
690 PRINT "MATRICE DI COVARIANZA"
700 PRINT "COLONNA "I;"":
710 FOR J = 0 TO DE
720 PRINT TAB( 5);AA(J,I)
730 NEXT J
740 PRINT : PRINT "PREMERE UN TASTO PER CONTINUARE": GET A$
750 NEXT I
760 REM
770 REM      FACOLTA' DI INTERPOLAZIONE
780 REM
790 PRINT : INPUT "INTERPOLAZIONE? (S O N)";A$
800 IF A$ = "N" THEN GOTO 930
810 PRINT : INPUT "X = ";X
820 REM      CALCOLO DELL'INTERPOLAZIONE CON IL METODO DI HORNER
830 P = BB(DE,0)
840 FOR I = 1 TO DE
850 J1 = DE - I
860 P = P * X + BB(J1,0)
870 NEXT I
880 PRINT "Y=";P
890 GOTO 790
900 REM
910 REM      STAMPA DEL GRAFICO (SE RICHIESTO)
920 REM
930 PRINT : INPUT "RAPPRESENTAZIONE GRAFICA? (S O N)";A$
940 IF A$ = "N" THEN END
950 REM
960 REM      CALCOLO DI X MINIMO E X MASSIMO
970 REM
980 XI = 1E38;XF = - 1E38
990 FOR I = 1 TO ND
1000 IF XD(I) > XF THEN XF = XD(I)
1010 IF XD(I) < XI THEN XI = XD(I)
1020 NEXT I
1030 XL = (XF - XI) / 266
1040 REM      CALCOLO DI Y MINIMO E Y MASSIMO
1050 REM      E CALCOLO SIMULTANEO DEL VETTORE Y(268) CHE
1060 REM      DEFINISCE IL GRAFICO DEL POLINOMIO
1070 REM
1080 P1 = BB(DE,0)
1090 YI = 1E38;YF = - 1E38
1100 FOR I = 0 TO 268
1110 X = XI + I * XL
1120 REM      CALCOLO DEL VALORE DEL POLINOMIO IN X
1130 P = P1
1140 FOR J = 1 TO DE
1150 J1 = DE - J
1160 P = P * X + BB(J1,0)
1170 NEXT J
1180 Y(I) = P
1190 IF P > YF THEN YF = P
1200 IF P < YI THEN YI = P
1210 NEXT I
1220 REM      PROSECUZIONE DEL CALCOLO PER Y MIN. E Y MAX.
1230 REM      TRA I VALORI DATI

```

```

1240 FOR I = 1 TO ND
1250 IF YD(I) > YF THEN YF = YD(I)
1260 IF YD(I) < YI THEN YI = YD(I)
1270 NEXT I
1280 YL = (YF - YI) / 148
1290 REM
1300 REM INIZIO DEL GRAFICO
1310 REM
1320 HOME : HGR : HCOLOR= 3
1330 REM REALIZZAZIONE DEL INQUADRATURA GRADUATA
1340 HPLLOT 0,9 TO 0,159 TO 270,159 TO 270,9 TO 0,9
1350 FOR I = 1 TO 9
1360 XP = 27 * I:YP = 159 - 15 * I
1370 HPLLOT 268,YP TO 270,YP
1380 HPLLOT XP,9 TO XP,11
1390 HPLLOT XP,157 TO XP,159
1400 NEXT I
1410 REM STAMPA DEI LIMITI
1420 VTAB (21)
1430 PRINT "X:":XI:" A ":XF:" PAR ":27 * XI
1440 PRINT "Y:":YI:" A ":YF:" PAR ":15 * YI
1445 VTAB (1)
1450 REM VISUALIZZAZIONE DEI PUNTI SPERIMENTALI
1460 FOR I = 1 TO ND
1470 XP = 268 + (XD(I) - XF) / XL
1480 YP = 157 - (YD(I) - YI) / YL
1490 HPLLOT XP - 2,YP TO XP + 2,YP
1500 HPLLOT XP,YP - 2 TO XP,YP + 2
1510 NEXT I
1520 REM GRAFICO DEL POLINOMIO INTERPOLATORE
1530 HPLLOT 0,157 - (Y(0) - YI) / YL
1540 FOR I = 1 TO 268
1550 YP = 157 - (Y(I) - YI) / YL
1560 HPLLOT TO I,YP
1570 NEXT I
1580 END
2000 REM *****
2010 REM SOTTOPROGRAMMA DI ADATTAMENTO AL POLINOMIO
2020 REM  $Y=B(0) + B(1)*X + B(2)*X^2 + \dots + B(DE)*X^DE$ 
2030 REM DEI DATI SPERIMENTALI X(I),Y(I)
2040 REM
2050 REM REGOLE PER L'USO DEL SOTTOPROGRAMMA
2060 REM 1.DATI NECESSARI:
2070 REM * DE=GRADO DEL POLINOMIO
2080 REM * ND=NUMERO DI DATI
2090 REM * XD(I)=ASCISSE DEI PUNTI
2100 REM I=1,2,...,ND
2110 REM * YD(I)=ORDINATE DEI PUNTI
2120 REM I=1,2,...,ND
2130 REM 2.RISULTATI FORNITI:
2140 REM * BB(I,0)=STIMA DI B(I)
2150 REM I=0,1,...,DE
2160 REM * AA(I,J)=COVARIANZA DI B(I),B(J)
2170 REM I,J=0,1,...,DE
2180 REM * SR=SOMMA RESIDUA DEI QUADRATI
2190 REM * SS=VARIANZA DEGLI Y(I)
2200 REM 3.VARIABILI UTILIZZATE:
2210 REM DE,I,J,ND,NP
2220 REM 4.VETTORI E MATRICI UTILIZZATI:
2230 REM X(ND,DE+1),XD(ND)
2240 REM 5.SOTTOPROGRAMMI RICHIAMATI:
2250 REM 3000:REGRESSIONE LINEARE MULTIPLA
2260 REM 4000:RISOLUZIONE DI UN SISTEMA LINEARE
2270 REM *****
2280 NP = DE + 1
2290 REM
2300 REM DEFINIZIONE DELLA MATRICE X DEL TESTO

```

```

2310 REM
2320 FOR I = 1 TO ND
2330 X(I,1) = 1
2340 FOR J = 2 TO NP
2350 X(I,J) = XD(I) ^ (J - 1)
2360 NEXT J
2370 NEXT I
2380 REM
2390 REM CHIAMATA DEL SOTTOPROGRAMMA DI REGRESSIONE LINEARE MULTIPLA
2400 REM
2410 GOSUB 3000
2420 RETURN
3000 REM *****
3010 REM SOTTOPROGRAMMA DI REGRESSIONE LINEARE MULTIPLA
3020 REM VERSIONE COMPATTATA
3030 REM
3040 REM CER IL PROGRAMMA DI REGRESSIONE LINEARE
3050 REM MULTIPLA PER LE REGOLE D'USO
3060 REM IL VETTORE Y(ND) E' SOSTITUITO DAL
3070 REM VETTORE YD(ND)
3080 REM *****
3090 SR = 0: N = NP: M = 1
3100 FOR I = 0 TO NP - 1: FOR J = 0 TO NP - 1: AA(I,J) = 0: NEXT J: BB(I,0)
= 0: NEXT I
3110 FOR I = 1 TO NP: FOR J = I TO NP: FOR K = 1 TO ND
3120 T1 = T - 1: J1 = J - 1: AA(I1,J1) = AA(I1,J1) + X(K,I) * X(K,J): NEXT K
3130 AA(J1,I1) = AA(I1,J1): NEXT J,I
3140 FOR I = 1 TO NP: I1 = I - 1: FOR K = 1 TO ND: BB(I1,0) = BB(I1,0) + X(
K,I) * YD(K): NEXT K,I
3150 GOSUB 4000
3160 FOR I = 1 TO ND: ST = 0
3170 FOR J = 1 TO NP: J1 = J - 1: ST = ST + X(I,J) * BB(J1,0): NEXT J
3180 X(I,0) = ST: NEXT I
3190 FOR I = 1 TO ND: SR = SR + ((YD(I) - X(I,0)) ^ 2): NEXT I
3200 SS = SR / (ND - NP)
3210 FOR I = 0 TO NP - 1: FOR J = 0 TO NP - 1: AA(I,J) = AA(I,J) * SS: NEXT
J,I
3220 RETURN
4000 REM *****
4010 REM SOTTOPROGRAMMA PER L'INVERSIONE DI UNA MATRICE
4020 REM IN VERSIONE COMPATTATA
4030 REM
4040 REM LA VARIABILE DET E' SOSTITUITA CON DT
4050 REM *****
4060 N1 = N - 1: M1 = M - 1: DT = 1
4070 FOR K = 0 TO N1
4080 PV = AA(K,K): IK = K: JK = K: PAV = ABS (PV)
4090 FOR I = K TO N1: FOR J = K TO N1
4100 IF ABS (AA(I,J)) > PAV THEN PV = AA(I,J): PAV = ABS (PV): IK = I: JK =
J
4110 NEXT J,I
4120 PC(K) = JK: PL(K) = IK
4130 IF IK < > K THEN DT = - DT
4140 IF JK < > K THEN DT = - DT
4150 DT = DT * PV
4160 IF DT = 0 THEN HOME: PRINT "IL DETERMINANTE E' NULLO, NESSUNA SOLU
ZIONE": END
4170 IF IK = K THEN GOTO 4210
4180 FOR I = 0 TO N1: TT = AA(IK,I): AA(IK,I) = AA(K,I): AA(K,I) = TT: NEXT
I
4190 IF M1 = - 1 THEN GOTO 4210
4200 FOR I = 0 TO M1: TT = BB(IK,I): BB(IK,I) = BB(K,I): BB(K,I) = TT: NEXT
I
4210 IF JK = K THEN GOTO 4230
4220 FOR I = 0 TO N1: TT = AA(I,JK): AA(I,JK) = AA(I,K): AA(I,K) = TT: NEXT
I
4230 FOR I = 0 TO N1: CS(I) = AA(I,K): AA(I,K) = 0: NEXT I

```

```

4240 CS(K) = 0:AA(K,K) = 1
4250 FOR I = 0 TO N1:AA(K,I) = AA(K,I) / PV: NEXT I
4260 IF M1 = - 1 THEN GOTO 4280
4270 FOR I = 0 TO M1:BB(K,I) = BB(K,I) / PV: NEXT I
4280 FOR J = 0 TO N1: IF J = K THEN GOTO 4320
4290 FOR I = 0 TO N1:AA(J,I) = AA(J,I) - CS(J) * AA(K,I): NEXT I
4300 IF M1 = - 1 THEN GOTO 4320
4310 FOR I = 0 TO M1:BB(J,I) = BB(J,I) - CS(J) * BB(K,I): NEXT I
4320 NEXT J
4330 NEXT K
4340 FOR I = N1 TO 0 STEP - 1:IK = PC(I)
4350 IF IK = I THEN GOTO 4390
4360 FOR J = 0 TO N1:TT = AA(I,J):AA(I,J) = AA(IK,J):AA(IK,J) = TT: NEXT
J
4370 IF M1 = - 1 THEN GOTO 4390
4380 FOR J = 0 TO M1:TT = BB(I,J):BB(I,J) = BB(IK,J):BB(IK,J) = TT: NEXT
J
4390 NEXT I
4400 FOR J = N1 TO 0 STEP - 1:JK = PL(J)
4410 IF JK = J THEN GOTO 4430
4420 FOR I = 0 TO N1:TT = AA(I,J):AA(I,J) = AA(I,JK):AA(I,JK) = TT: NEXT
I
4430 NEXT J
4440 RETURN

```

## 4 — ESEMPIO PRATICO

L'esempio illustra l'adattamento ad un polinomio di terzo grado delle 7 coppie di valori sperimentali:

|     |        |      |      |      |      |       |      |
|-----|--------|------|------|------|------|-------|------|
| x = | - 3,   | - 2, | - 1, | 0,   | 1,   | :     | 3    |
| y = | - 4,5, | 2,7, | 3,1, | 0,9, | 2,6, | 15,1, | 42,6 |

e fornisce la stima dei parametri, la matrice delle covarianze, una interpolazione per  $x = 0.5$  ed una rappresentazione grafica dei dati sperimentali e della funzione polinomiale calcolata.

IRUN

ADATTAMENTO A UN POLINOMIO  
DI GRADO DE

AUTORE:H.HAUT

1.DEFINIRE IL GRADO DEL POLINOMIO  
E IL NUMERO DEI DATI SPERIMENTALI

2.INTRODURRE POI LE ND COPPIE DI DATI  
X(I),Y(I) DA ADATTARE AL POLINOMIO:  
Y=B(0) + B(1)\*X + ... + B(DE)\*X^DE

3.IL PROGRAMMA FORNISCE GLI ESTIMATORI  
DEI B(I),LA MATRICE DELLE COVARIANZE,  
DA' LA POSSIBILITA' DI EFFETTUARE UNA  
INTERPOLAZIONE E IL RELATIVO GRAFICO  
PREMERE UN TASTO PER PROSEGUIRE  
GRADO DEL POLINOMIO= 3

NUMERO DI DATI= 7

1  
X=-3  
Y=-4.5

2  
X=-2  
Y=2.7

3  
X=-1  
Y=3.1

4  
X=0  
Y=0.9

5  
X=1  
Y=2.6

6  
X=2  
Y=15.1

7  
X=3  
Y=42.6

SOMMA RESIDUA DEI QUADRATI= .303333339

VARIANZA DI XI= .101111113

ESTIMATORI DEI DEVIACIONI

PARAMETRI STANDARD

\*\*\*\*\*

B(0)=.842857139 .183585687

B(1)=-.930158738 162936657

B(2)=2.02142857 .0346944336

B(3)=.977777778 .0216357806

SE SI VUOLE LA MATRICE DELLE COVARIANZE

BATTERE UN TASTO ALTRIMENTI ESC

MATRICE DI COVARIANZA

COLONNA 0:

.0337037043

1.52689317E-11

-4.8148149E-03

-1.94105385E-12

PREMERE UN TASTO PER CONTINUARE

MATRICE DI COVARIANZA

COLONNA 1:

1.52689316E-11

.0265483543

-3.3645922E-12

-3.27674902E-03

PREMERE UN TASTO PER CONTINUARE

MATRICE DI COVARIANZA

COLONNA 2:

-4.8148149E-03

-3.3645922E-12

1.20370372E-03

4.46338587E-13

PREMERE UN TASTO PER CONTINUARE  
MATRICE DI COVARIANZA  
COLONNA 3:  
-1.94105385E-12  
-3.27674902E-03  
4.46338586E-13  
4.68107003E-04

PREMERE UN TASTO PER CONTINUARE

INTERPOLAZIONE? (S O N):S


X = 0.5  
Y=1.00535714

INTERPOLAZIONE? (S O N):S

X = -0.5  
Y=1.69107143

INTERPOLAZIONE? (S O N):N

REPRESENTAZIONE GRAFICA (S O N):S



tempo d'esecuzione : 10"2 (+ 26" per il grafico)

memoria richiesta : 5869 bytes (per poter eseguire il grafico occorrono invece più di 16K. Se non si dispone di tale capacità è necessario, come indicato in precedenza, eliminare dal testo del programma le frasi di commento).

# BIBLIOGRAFIA

- A1 — Handbook of Mathematical Functions  
Abramowitz M. and Segun I.A.  
(*Dover Publications Inc. New - York 1968*)
  
- A2 — Numerical Methods that Works  
Acton F.S.  
(*Harper and Row Publishers 1970*)
  
- B1 — Computing Methods (due volumi)  
Berezin and Zhidkov  
(*Pergamon Press 1965*)
  
- B2 — Numerical Methods  
Buckingham R.A.  
(*Sir Isaac Pitman & Sons Ltd London 1962*)
  
- B3 — An introduction to Numerical Methods  
Butler R. and Kerr E.  
(*Sir Isaac Pitman & Sons Ltd London*)
  
- C1 — Principles of Computation  
Calingaert P.  
(*Addison - Wesley 1965*)
  
- C2 — Analyse numérique  
Carasso C.  
(*Lidec Inc. Montréal 1971*)
  
- C3 — Introduction to Numerical Analysis  
Celia C.W.  
(*Mc Graw Hill 1969*)
  
- C4 — Numerical Analysis  
Cohen A.M.  
(*Mc Graw Hill 1973*)

- C5 — Fitting Equations to Data  
Cuthbert D. and Wood F.  
(*J. Wiley & Sons 1980*)
  
- D1 — Eléments de Calcul Numérique  
Demidovitch B. et Maron I.  
(*Editions de Moscou 1979*)
  
- G1 — Statistical Treatment of Experimental Data  
Green J.R. and Margerison D.  
(*Elsevier 1979*)
  
- H1 — Travaux Pratiques de Calcul Numérique  
Hennequin A.  
(*Vuibert 1966*)
  
- H2 — Lectures on Elementary Statistics and Probability  
Hudson D.J.  
(*CERN Report 63—29, 64—18*)
  
- J1 — A first Course in Computing and Numerical Methods  
Jacquez J.A.  
(*Addison - Wesley 1969*)
  
- J2 — Handbook of Statistical Distributions  
Jagdish K.P., Kapadia C.H. and Owen D.B.  
(*Marcel Dekker Inc., New York 1976*)
  
- L1 — Méthodes et Techniques de l'Analyse Numérique  
Legras J.  
(*Dunod 1971*)
  
- L2 — Précis d'Analyse Numérique  
Legras J.  
(*Dunod 1963*)
  
- L3 — Introduction à l'Analyse Numérique  
Louis F.  
(*Testo utilizzato nel corso CERN 1969/70*)
  
- M1 — Data Analysis for Scientists and Engineers  
Meyer S.L.  
(*J. Wiley & Sons New - York 1975*)



- S1 — Numerical Methods in Engineering  
Salvadori M.G. and Baron M.L.  
(*Prentice - Hall Inc. 1961*)
- S2 — Theory and Problems of Numerical Analysis  
Scheid F.  
(*Schaum's Outline Series, Mc Graw Hill 1968*)
- S3 — Introduction to Numerical Analysis  
Stoer J. and Bulirsch  
(*Springer - Verlag 1980*)
- W1 — Numerical Analysis, the Mathematics of Computing (due volumi)  
Watson W.A. Philipson T. and Oates P.J.  
(*Edward Arnold 1964*)









Leggendo questo libro il lettore potrà formarsi quella logica di base indispensabile per la risoluzione di problemi di matematica e statistica.

Potrà altresì ottenere quattro risultati parimenti importanti.

Comprendere a fondo la materia grazie all'approccio metodico di analisi del problema. Approfondire i concetti di programmazione conosciuti. Eliminare la risoluzione "a mano" un'inutile perdita di tempo. Avere alla fine, dei programmi che possono essere utilizzati sia singolarmente che in un ambito più vasto.

Ad ogni programma viene preposta un'esposizione schematica del metodo numerico e delle tecniche di programmazione utilizzate il diagramma a blocchi relativo all'algoritmo, il listato (direttamente da calcolatore) e un esempio pratico (anche esso ottenuto da calcolatore) in cui tra l'altro vengono specificati il tempo e la quantità di memoria impiegati.

Per la codifica dei programmi presentati (strutturati in due parti, una generale, assicura la gestione corretta su video dell'input/output, l'altra, costituita da una routine a sua volta suddivisibile in più subroutine, è la risoluzione del problema) si è utilizzato il BASIC Applesoft implementato sul sistema Apple II.

Le particolarità di questa versione, però, sono utilizzate solo nella gestione input/output, per cui risulta molto semplice adattare gli stessi programmi ad altre versioni BASIC.

71

# PROGRAMMI DI MATEMATICA E STATISTICA IN **BASIC**

Hervé Haut

GRUPPO  
EDITORIALE  
JACKSON

